

A general invariant manifold construction algorithm, including isochrons of slow manifolds

A. J. Roberts*

Nov 2013 – April 6, 2021

Abstract

This procedure constructs a specified invariant manifold for a specified system of ordinary differential equations or delay differential equations. The invariant manifold may be any of a centre manifold, a slow manifold, an un/stable manifold, a sub-centre manifold, a nonlinear normal form, any spectral submanifold, or indeed a normal form coordinate transform of the entire state space. Thus the procedure may be used to analyse pitchfork bifurcations, or oscillatory Hopf bifurcations, or any more complicated superposition. In the cases when the neglected spectral modes all decay, the constructed invariant manifold supplies a faithful large time model of the dynamics of the differential equations. Further, in the case of a slow manifold, this procedure now derives vectors defining the projection onto the invariant manifold along the isochrons: this projection is needed for initial conditions, forcing, system modifications, and uncertainty quantification.

Contents

1	Introduction	2
1.1	A simple example: <code>exampleslowman()</code>	4
1.2	Header of the procedure	5

*School of Mathematical Sciences, University of Adelaide, South Australia 5005, AUSTRALIA. <https://profajroberts.github.io/>

1	<i>Introduction</i>	2
1.3	Preamble to the procedure	6
1.4	Check the dimensionality of specified system	7
2	Dissect the linear part	8
2.1	Normalise the adjoint eigenvectors	9
2.2	Operator to represent delays	9
2.3	Linearise at the origin	10
2.4	Eigen-check	11
2.5	Ameliorate the nonlinearity	12
2.6	Store invariant manifold eigenvalues	13
2.7	Precompute matrices for updates	14
2.8	Define operators that invoke these inverses	16
3	Initialise LaTeX output	18
4	Linear approximation to the invariant manifold	21
5	Iteratively construct the invariant manifold	22
6	Output text version of results	24
7	Output LaTeX version of results	25
8	Fin	27
9	Override some system procedures	28

1 Introduction

Download and install the computer algebra package *Reduce* via <http://www.reduce-algebra.com> Download and unzip the folder <https://profajroberts.github.io/InvariantManifold.zip> Within the folder `InvariantManifold`, start-up *Reduce* and load the procedure by executing the command `in_tex "invariantManifold.tex"$`¹ Test your installation

¹This script changes a lot of internal settings of *Reduce*, so best only to do when needed.

by then executing `exampleslowman()`;

Thereafter, construct a specified invariant manifold of a specific dynamical system by executing the following command with specific values for the input parameters. See `allExamples.pdf` for many examples.

```
1 invariantmanifold(odefns, evals, evecs, adjvecs, toosmall);
```

Inputs As in the example of the next [Section 1.1](#), the input parameters to the procedure are the following:

- **odefns**, a comma separated list within `mat(...)`, the RHS expressions of the ODEs/DDEs of the system, a system expressed in terms of variables `u1, u2, ...`, for time derivatives $du_1/dt, du_2/dt, \dots$;
any time delayed variables in the RHS are coded by the time-delay in parenthesis after the variable, as in the example `u1(pi/2)` to represent $u_1(t - \pi/2)$ in the DDEs;
- **evals**, a comma separated list within `mat(...)`, the eigenvalues of the modes to be the basis for the invariant manifold;
- **evecs**, a comma separated list of vectors within `mat(...)`—each vector a comma separated list of components within `(...)`, the eigenvectors of the modes corresponding to the given eigenvalues of the invariant manifold basis;
- **adjvecs**, a comma separated list of vectors within `mat(...)`, usually the adjoint eigenvectors of the modes corresponding to the given eigenvalues of the invariant manifold basis;
- **toosmall**, an integer giving the desired order of error in the asymptotic approximation that is constructed. The procedure embeds the specified system in a family of systems parametrised by ε , and constructs an invariant manifold, and evolution thereon, of the embedding system to the asymptotic error $\mathcal{O}(\varepsilon^{\text{toosmall}})$ (as $\varepsilon \rightarrow 0$). Often the introduced artificial ε has a useful physical meaning, but strictly you should

evaluate the output at $\varepsilon = 1$ to recover results for the specified system, and then interpret the results in terms of actual ‘small’ parameters.

Outputs This procedure reports the specified system, the embedded system it actually analyses, the number of iterations taken, the invariant manifold approximation, the evolution on the invariant manifold, and optionally a basis for projecting onto the invariant manifold.

- A plain text report to the Terminal window in which `Reduce` is executing—the invariant manifold is parametrised by variables `s(1)`, `s(2)`, \dots , and the dynamics by their evolution in time.
- A \LaTeX source report written to the file `invarManReport.tex` (and `invarManReportSys.tex`)—the invariant manifold is parametrised by variables s_1, s_2, \dots , and the dynamics by their evolution in time.

One may change the appearance of the output somewhat. For example, it is often useful to execute `factor s`; before executing `invariantmanifold(...)` in order to group terms with the same powers of amplitudes/order-parameters/coarse-variables.

1.1 A simple example: `exampleslowman()`

Execute this example by invoking the command `exampleslowman()`; The example system to analyse is specified to be

$$\dot{u}_1 = -u_1 + u_2 - u_1^2, \quad \dot{u}_2 = u_1 - u_2 + u_2^2.$$

```
2 procedure exampleslowman;
3   invariantmanifold(
4     mat((-u1+u2-u1^2,u1-u2+u2^2)),
5     mat((0)),
6     mat((1,1)),
7     mat((1,1)),
8     5)$
```

We seek the slow manifold so specify the eigenvalue zero. From the linearisation matrix $\begin{bmatrix} -1 & 1 \\ 1 & -1 \end{bmatrix}$ a corresponding eigenvector is $\vec{e} = (1, 1)$, and a corresponding left-eigenvector is $\vec{z} = \vec{e} = (1, 1)$, as specified. The last parameter specifies to construct the slow manifold to errors $\mathcal{O}(\varepsilon^5)$.

The procedure actually analyses the embedding system, the family of problems,

$$\dot{u}_1 = -u_1 + u_2 - \varepsilon u_1^2, \quad \dot{u}_2 = u_1 - u_2 + \varepsilon u_2^2.$$

Here the artificial parameter ε has a physical interpretation in that it counts the nonlinearity: a term in ε^p will be a $(p + 1)$ th order term in $\vec{u} = (u_1, u_2)$. Hence the specified error $\mathcal{O}(\varepsilon^5)$ is here the same as error $\mathcal{O}(|\vec{s}|^6)$.

The constructed slow manifold is, in terms of the parameter s_1 (and reverse ordering!),

$$\begin{aligned} u_1 &= 3/8\varepsilon^3 s_1^4 - 1/2\varepsilon s_1^2 + s_1, \\ u_2 &= -3/8\varepsilon^3 s_1^4 + 1/2\varepsilon s_1^2 + s_1. \end{aligned}$$

On this slow manifold the evolution is

$$\dot{s}_1 = -3/4\varepsilon^4 s_1^5 + \varepsilon^2 s_1^3 :$$

here the leading term in s_1^3 indicates the origin is unstable. To project initial conditions onto the slow manifold, or non-autonomous forcing, or modifications of the original system, or to quantify uncertainty, use the projection defined by the derived vector

$$\vec{z}_1 = \begin{bmatrix} z_{11} \\ z_{12} \end{bmatrix} = \begin{bmatrix} 3/2\varepsilon^4 s_1^4 + 3/4\varepsilon^3 s_1^3 - 1/2\varepsilon^2 s_1^2 - 1/2\varepsilon s_1 + 1/2 \\ 3/2\varepsilon^4 s_1^4 - 3/4\varepsilon^3 s_1^3 - 1/2\varepsilon^2 s_1^2 + 1/2\varepsilon s_1 + 1/2 \end{bmatrix}.$$

Evaluate these at $\varepsilon = 1$ to apply to the original specified system, or here just interpret ε as a way to count the order of each term.

1.2 Header of the procedure

Need a couple of things established before defining the procedure: the `rlfi` package; and operator names for the variables of the dynamical system (in case they have delays)—currently code a max of nine variables.

```

9 load_package rlfi;
10 operator u1,u2,u3,u4,u5,u6,u7,u8,u9;

```

Now define the procedure as an operator so we can be flexible with its arguments.

```

11 operator invariantmanifold;
12 for all odefns, evals, evecs, adjvecs, toosmall let
13   invariantmanifold(odefns, evals, evecs, adjvecs, toosmall)
14   = begin

```

1.3 Preamble to the procedure

Operators and arrays are always global, but we can make variables and matrices local, except for matrices that need to be declared `matrix`. So, move to implement all arrays and operators to have underscores, and almost all scalars and most matrices to be declared local here.

```

15 scalar ff, evalm, ee, zz, maxiter, ff0, trace, ll, uvec,
16 reslin, ok, rhsjact, jacadj, resd, resde, resz, rhsfn, zs,
17 pp, est, eyem;

```

Transpose the defining matrices so that vectors are columns.

```

18 ff := tp odefns;
19 ee := tp evecs;
20 zz := tp adjvecs;

```

Define default parameters for the iteration: `maxiter` is the maximum number of allowed iterations. Specific problems may override these defaults.

```

21 maxiter:=29$
22 factor small;

```

For optional trace printing of test cases: comment out second line when not needed.

```

23 trace:=0$
24 %trace:=1; maxiter:=5;

```

The `rationalize` switch makes code much faster with complex numbers. The switch `gcd` seems to wreck convergence, so leave it off.

```
25 on div; off allfac; on revpri;
26 on rationalize;
```

Use `e_` as basis vector for matrices and vectors. Declare it non-commutative so that multiplication does not commute.

```
27 operator e_;
28 noncom e_;
29 factor e_;
30 let { e_(~j,~k)*e_(~l,~m)=>0 when k neq l
31      , e_(~j,~k)*e_(~l,~m)=>e_(j,m) when k=l
32      , e_(~j,~k)^2=>0 when j neq k
33      , e_(~j,j)^2=>e_(j,j) };
```

Also need (once) a transpose operator: do complex conjugation explicitly when needed.

```
34 operator tpe_; linear tpe_;
35 let tpe_(e_(~i,~j),e_)=>e_(j,i);
```

Empty the output LaTeX file in case of error.

```
36 out "invarManReport.tex";
37 write "This empty document indicates error.";
38 shut "invarManReport.tex";
```

1.4 Check the dimensionality of specified system

Extract dimension information from the parameters of the procedure: seek mD invariant manifold of an nD system.

```
39 write "total no. of variables ",
40 n:=part(length(ee),1);
41 write "no. of invariant modes ",
42 m:=part(length(ee),2);
```

```

43 if {length(evals),length(zz),length(ee),length(ff)}
44   ={{1,m},{n,m},{n,m},{n,1}}
45   then write "Input dimensions are OK"
46   else <<write "INCONSISTENT INPUT DIMENSIONS, I EXIT";
47       return>>;

```

For the moment limit to a maximum of nine components.

```

48 if n>9 then <<write "SORRY, MAX NUMBER ODEs IS 9, I EXIT";
49     return>>;

```

Need an $m \times m$ identity matrix for normalisation of the isochron projection.

```

50 eyem:=for j:=1:m sum e_(j,j)$

```

2 Dissect the linear part

Use the exponential $\exp(u) = e^u$, but not with the myriad of inbuilt properties so clear it! Do not (yet) invoke the simplification of $\exp(0)$ as I want it to label modes of no oscillation, zero eigenvalue.

```

51 clear exp; operator exp;
52 let { df(exp(~u),t) => df(u,t)*exp(u)
53       , exp(~u)*exp(~v) => exp(u+v)
54       , exp(~u)^^p => exp(p*u)
55     };

```

Need function `conj_` to do parsimonious complex conjugation.

```

56 operator exp__;
57 procedure conj_(a)$ sub(i=-i,a)$

```

Make an array of eigenvalues for simplicity (`evals` not used hereafter).

```

58 array eval_(m);
59 for j:=1:m do eval_(j):=evals(1,j);

```


2.1 Normalise the adjoint eigenvectors

When we include delay differential equations, then we need to account for the history of the eigenvector as well. Hence multiply each eigenvector by its oscillating factor, $e^{i\omega t}$, $e^{\lambda t}$, and then take the mean. This multiplication by its oscillating factor should not make any difference for non-delay equations by the natural orthogonality of left and right eigenvectors of different eigenvalues.

Note: the ‘left eigenvectors’ have to be the eigenvectors of the complex conjugate transpose, and for the complex conjugate eigenvalue. This seems best: for example, when the linear operator is $\begin{bmatrix} 0 & 1 \\ -1 & 0 \end{bmatrix}$ then the adjoint and the right eigenvectors are the same.

For oscillations and un/stable manifolds we have to cope with imaginary and with real eigenvalues. Seems to need `zz` to have negative complex conjugated frequency so store in `cexp_`—cannot remember why this appears to work!? It may only work for pure real and for pure imaginary eigenvalues??

```

60 matrix aa_(m,m),dexp_(m,m),cexp_(m,m);
61 for j:=1:m do dexp_(j,j):=exp(eval_(j)*t);
62 for j:=1:m do cexp_(j,j):=exp(-conj_(eval_(j))*t);
63 aa_:= (tp map(conj_(~b),ee*dexp_)*zz*cexp_ )$
64 if trace then write aa_:=aa_;
65 write "Normalising the left-eigenvectors:";
66 aa_:= (aa_ where {exp(0)=>1, exp(~a)=>0 when a neq 0})$
67 if trace then write aa_:=aa_;
68 if det(aa_)=0 then << write
69     "ORTHOGONALITY ERROR IN EIGENVECTORS; I EXIT";
70     return>>;
71 zz:=zz*aa_^(-1);

```

2.2 Operator to represent delays

Introduce an operator to represent delay factors more conveniently for analysis. The `exp` rule probably only works for pure imaginary modes!?

```

72 operator d_; linear d_;
73 let { d_(~a~^p,t,~dt)=>d_(a,t,dt)^p
74      , d_(~a*~b,t,~dt)=>d_(a,t,dt)*d_(b,t,dt)
75      , d_(exp(~a),t,~dt)=>exp(a)
76          *sub(t=-dt,cos(-i*a)+i*sin(-i*a))
77      , df(d_(~a,t,~dt),~b)=>d_(df(a,b),t,dt)
78      , d_(~a,t,0)=>a
79      , d_(d_(~a,t,~dta),t,~dtb)=>d_(a,t,dta+dtb)
80      };

```

Now rewrite the (delay) factors in terms of this operator. For the moment limit to a maximum of nine ODEs.

```

81 somerules:={}
82 depend u1,t; somerules:=(u1(~dt)=d_(u1,t,dt)).somerules$
83 depend u2,t; somerules:=(u2(~dt)=d_(u2,t,dt)).somerules$
84 depend u3,t; somerules:=(u3(~dt)=d_(u3,t,dt)).somerules$
85 depend u4,t; somerules:=(u4(~dt)=d_(u4,t,dt)).somerules$
86 depend u5,t; somerules:=(u5(~dt)=d_(u5,t,dt)).somerules$
87 depend u6,t; somerules:=(u6(~dt)=d_(u6,t,dt)).somerules$
88 depend u7,t; somerules:=(u7(~dt)=d_(u7,t,dt)).somerules$
89 depend u8,t; somerules:=(u8(~dt)=d_(u8,t,dt)).somerules$
90 depend u9,t; somerules:=(u9(~dt)=d_(u9,t,dt)).somerules$
91 ff:=(ff where somerules)$

```

2.3 Linearise at the origin

Assume the equilibrium is at the origin. Find the linear operator at the equilibrium. Include `small=0` as we notionally adjoin it in the list of variables. Do not need to here make any non-zero forcing small at the equilibrium as it gets multiplied by `small` later. (For some reason using `mkid(u,k)=>0` does not resolve the `mkid`, but `mkid(u,k)=0` does; however, not clear if it is a problem??)

```

92 ll:=ee*(tp ee)*0; %zero nxn matrix
93 uzero:=(for k:=1:n collect (mkid(u,k)=0))$

```

```

94 equilibrium:=(small=0).uzero$
95 for j:=1:n do for k:=1:n do begin
96   ll(j,k):=df(ff(j,1),mkid(u,k));
97   ll(j,k):=sub(equilibrium,ll(j,k));
98 end;
99 write "Find the linear operator is";
100 write ll:=ll;

```

We need a vector of unknowns for a little while.

```

101 uvec:=0*ff; %nx1 zero matrix
102 for j:=1:n do uvec(j,1):=mkid(u,j);

```

2.4 Eigen-check

Variable `aa_` appears here as the diagonal matrix of eigenvalues. Check that the eigenvalues and eigenvectors are specified correctly.

```

103 write "Check invariant subspace linearisation ";
104 for j:=1:m do for k:=1:m do aa_(j,k):=0;
105 for j:=1:m do aa_(j,j):=eval_(j);
106 % following maybe only for pure centre modes????
107 reslin:=(ll*(ee*dexp_)-(ee*dexp_)*aa_
108   where exp(~a)*d_(1,t,~dt)=>sub(t=-dt,cos(-i*a)+i*sin(-i*a))*
109 if trace then write reslin:=reslin;
110 ok:=1$
111 for j:=1:n do for k:=1:m do
112   ok:=if reslin(j,k)=0 then ok else 0$
113 if ok then write "Linearisation is OK";

```

Try to find a correction of the linear operator that is ‘close’. Multiply by the adjoint eigenvectors and then average over time: operator $\mathcal{L}_{\text{new}} := \mathcal{L} - \mathcal{L}_{\text{adj}}$ should now have zero residual. Lastly, correspondingly adjust the ODEs, since `lladj` does not involve delays we do not need delay operator transforms in the product.

```

114 if not ok then for iter:=1:2 do begin

```

```

115 write "WARNING: I NEED TO ADJUST LINEAR OPERATOR";
116 write
117 lladj:=reslin*tp map(conj_(~b),zz*cexp_);
118 write
119 lladj:=(lladj where {exp(0)=>1, exp(~a)=>0 when a neq 0});
120 write
121 ll:=ll-lladj;
122 % following maybe only for pure centre modes????
123 write
124 reslin:=(ll*(ee*dexp_)-(ee*dexp_)*aa_
125     where exp(~a)*d_(1,t,~dt)=>sub(t=-dt,cos(-i*a)+i*sin(-i*a))*
126 ok:=1$
127 for j:=1:n do for k:=1:m do
128     ok:=if reslin(j,k)=0 then ok else 0$
129 if ok then iter:=iter+1000;
130 end;
131 if not ok then << write
132     "OOPS, INCONSISTENT EIGENVALUES, EIGENVECTORS AND OPERATOR.
133     EMAIL ME; I EXIT";
134     return >>;

```

2.5 Ameliorate the nonlinearity

Anything not in the linear operator gets multiplied by `small` to be treated as small in the analysis. The feature of the second alternative is that when a user invokes `small` then the power of smallness is not then changed; however, causes issues in the relative scaling of some terms, so restore to the original version. This might need reconsidering?? The current `if` always chooses the first simple alternative.

```

135 somerules:=for j:=1:n collect
136     (d_(1,t,~dt)*mkid(u,j)=d_(mkid(u,j),t,dt))$
137 ll0:=(ll*uvec where somerules)$
138 ff:=(if 1 then small*ff
139     else ff-(1-small)*sub(small=0,ff))

```

```
140      +(1-small)*l10_$
```

Any constant term in the equations `ff` has to be multiplied by `exp(0)`.

```
141 ff0:=(ff where uzero)$
```

```
142 ff:=ff+(exp(0)-1)*ff0$
```

From the matrix versions of the equations, create algebraic form using the matrix basis.

```
143 rhsfn:=for i:=1:n sum e_(i,1)*ff(i,1)$
```

Also, create the algebraic form of the jacobian transpose using the matrix basis: take the conjugate later when used.

```
144 rhsjact:=for i:=1:n sum for j:=1:n sum
```

```
145      e_(j,i)*df(ff(i,1),mkid(u,j))$
```

2.6 Store invariant manifold eigenvalues

Extract all the eigenvalues in the invariant manifold, and the set of all the corresponding modes in the invariant manifold variables. The slow modes are accounted for as having zero eigenvalue. Remember the eigenvalue set is not in the ‘correct’ order. Array `modes_` stores the set of indices of all the modes of a given eigenvalue.

```
146 array eval_s(m),modes_(m);
```

```
147 neval:=0$ eval_set:={} $
```

```
148 for j:=1:m do if not(eval_(j) member eval_set) then begin
```

```
149   neval:=neval+1;
```

```
150   eval_s(neval):=eval_(j);
```

```
151   eval_set:=eval_(j).eval_set;
```

```
152   modes_(neval):=for k:=j:m join
```

```
153     if eval_(j)=eval_(k) then {k} else {};
```

```
154 end;
```

Set a flag for the case of a slow manifold when all eigenvalues are zero, as then we compute the isochron projection. The next challenge is to get this

isochron code working for the case of non-slow invariant manifolds.

```
155 itisSlowMan_:=if eval_set={0} then 1 else 0$
156 if trace then write itisSlowMan_:=itisSlowMan_;
```

Put in the non-singular general case as the zero entry of the arrays.

```
157 eval_s(0):=geneval$
158 modes_(0):={}$
```

2.7 Precompute matrices for updates

Precompute matrices to solve for updates for each of the critical eigenvalues, and the general case $k = 0$. The matrix

$$\text{llzz}_- = \begin{bmatrix} -\mathcal{L} + \partial_t & \mathcal{E}_0 \\ \mathcal{Z}_0^\dagger & 0 \end{bmatrix}$$

and then put its inverse in place. Subsequently, extract the blocks for the generalised inverses and solvability condition into `l_invs` and `g_invs`.

```
159 matrix llzz_(n+m,n+m);
160 array l_invs(neval),g_invs(neval);
161 array l1_invs(neval),g1_invs(neval),l2_invs(neval),g2_invs(neval);
162 operator sp_; linear sp_;
163 for k:=0:neval do begin
164   if trace then write "ITERATION ",k:=k;
```

Code the operator $\mathcal{L}\hat{v}$ where the delay is to only act on the oscillation part.

```
165   for ii:=1:n do for jj:=1:n do llzz_(ii,jj):=(
166     -sub(small=0,ll(ii,jj))
167     where d_(1,t,~dt)=>cos(i*eval_s(k)*dt)+i*sin(i*eval_s(k)*dt)
```

Code the operator $\partial\hat{v}/\partial t$ where it only acts on the oscillation part.

```
168   for j:=1:n do llzz_(j,j):=eval_s(k)+llzz_(j,j);
```

Now code the part leading to the solvability condition which arises from allowing the (oscillation) amplitude to evolve. Use operator `sp_` to extract the delay parts that subtly affect the updates of the evolution.

```

169   for j:=1:length(modes_(k)) do
170     for ii:=1:n do llzz_(ii,n+j):=ee(ii,part(modes_(k),j))
171       +(for jj:=1:n sum
172         sp_(ll(ii,jj)*ee(jj,part(modes_(k),j)),d_)
173         where { sp_(1,d_)=>0
174               , sp_(d_(1,t,~dt),d_)=>dt*(
175                 cos(i*eval_s(k)*dt)+i*sin(i*eval_s(k)*dt))
176               });

```

Force the updates to be orthogonal to the left-eigenvectors in the complex conjugate transpose adjoint.

```

177   for ii:=1:length(modes_(k)) do
178     for j:=1:n do llzz_(n+ii,j):=conj_(zz(j,part(modes_(k),ii)))
179   if trace then write "finished Force the updates to be orthogonal

```

Set the bottom-right corner of the matrix to zero.

```

180   for i:=1:length(modes_(k)) do
181     for j:=1:m do llzz_(n+i,n+j):=0;

```

Add some trivial rows and columns to make the matrix up to the same size for all eigenvalues.

```

182   for i:=length(modes_(k))+1:m do begin
183     for j:=1:n+i-1 do llzz_(n+i,j):=llzz_(j,n+i):=0;
184     llzz_(n+i,n+i):=1;
185   end;
186   if trace then write "finished Add some trivial rows and columns

```

Invert the matrix and unpack into arrays ready for use by the inversion operators.

```

187   if trace then write llzz_:=llzz_;
188   llzz_:=llzz_^(-1);

```

```

189   if trace then write llzz_:=llzz_;
190   l_invs(k):=for i:=1:n sum for j:=1:n sum e_(i,j)*llzz_(i,j);
191   g_invs(k):=for i:=1:length(modes_(k)) sum
192     for j:=1:n sum e_(part(modes_(k),i),j)*llzz_(i+n,j);
193 if trace then write "finished Invert the matrix and unpack";

```

Unpack the conjugate transpose for inverse operators used for the isochrons. A difference here is that the orthogonality condition is non-trivial (in the slow manifold we assumed amplitudes were exactly orthogonal to the left-eigenvectors), so we need to remember more parts of the inverse of the matrix.

```

194 %   realgeneval:={repart(geneval)=>geneval, impart(geneval)=>0}$
195   l1_invs(k):=for ii:=1:n sum for j:=1:n sum
196     e_(ii,j)*conj_(llzz_(j,ii));
197   l2_invs(k):=for ii:=1:n sum for j:=1:length(modes_(k)) sum
198     e_(ii,part(modes_(k),j))*conj_(llzz_(j+n,ii));
199   g1_invs(k):=for ii:=1:length(modes_(k)) sum for j:=1:n sum
200     e_(part(modes_(k),ii),j)*conj_(llzz_(j,ii+n));
201   g2_invs(k):=
202     for ii:=1:length(modes_(k)) sum for j:=1:length(modes_(k)) su
203     e_(part(modes_(k),ii),part(modes_(k),j))*conj_(llzz_(j+n,i
204   if trace then write "finished Unpack the conjugate transpose";
205 end;

```

2.8 Define operators that invoke these inverses

Decompose residuals into parts, and operate on each. First for the invariant manifold. But making $e_$ non-commutative means that it does not get factored out of these linear operators: must post-multiply by $e_$ because the linear inverse is a premultiply.

```

206 operator l_inv; linear l_inv;
207 let l_inv(e_(~j,~k)*exp(~a),exp)=>l_invproc(a/t)*e_(j,k);
208 procedure l_invproc(a);
209   if a member eval_set

```



```

210   then << k:=0;
211       repeat k:=k+1 until a=eval_s(k);
212       l_invs(k)*exp(a*t) >>
213   else sub(geneval=a,l_invs(0))*exp(a*t)$

```

Second for the evolution on the invariant manifold.

```

214 operator g_inv; linear g_inv;
215 let g_inv(e_(~j,~k)*exp(~a),exp)=>ginv_proc(a/t)*e_(j,k);
216 procedure ginv_proc(a);
217   if a member eval_set
218   then << k:=0;
219       repeat k:=k+1 until a=eval_s(k);
220       g_invs(k) >>
221   else sub(geneval=a,g_invs(0))$

```

Copy and adjust the above for the projection. But first define the generic procedure.

```

222 procedure inv_proc(a,invs);
223   if a member eval_set
224   then << k:=0;
225       repeat k:=k+1 until a=eval_s(k);
226       invs(k)*exp(a*t) >>
227   else sub(geneval=a,invs(0))*exp(a*t)$

```

Then define operators that we use to update the projection.

```

228 operator l1_inv; linear l1_inv;
229 operator l2_inv; linear l2_inv;
230 operator g1_inv; linear g1_inv;
231 operator g2_inv; linear g2_inv;
232 let { l1_inv(e_(~j,~k)*exp(~a),exp)=>inv_proc(a/t,l1_invs)*e_(j,l
233       , l2_inv(e_(~j,~k)*exp(~a),exp)=>inv_proc(a/t,l2_invs)*e_(j,l
234       , g1_inv(e_(~j,~k)*exp(~a),exp)=>inv_proc(a/t,g1_invs)*e_(j,l
235       , g2_inv(e_(~j,~k)*exp(~a),exp)=>inv_proc(a/t,g2_invs)*e_(j,l
236   };

```

3 Initialise LaTeX output

Set the default output to be inline mathematics.

```
237 \mathstyle math;
```

Define the Greek alphabet with `small` as well.

```
238 \defid small,name="\eps";%varepsilon;  
239 %\defid small,name=varepsilon;  
240 \defid alpha,name=alpha;  
241 \defid beta,name=beta;  
242 \defid gamma,name=gamma;  
243 \defid delta,name=delta;  
244 \defid epsilon,name=epsilon;  
245 \defid varepsilon,name=varepsilon;  
246 \defid zeta,name=zeta;  
247 \defid eta,name=eta;  
248 \defid theta,name=theta;  
249 \defid vartheta,name=vartheta;  
250 \defid iota,name=iota;  
251 \defid kappa,name=kappa;  
252 \defid lambda,name=lambda;  
253 \defid mu,name=mu;  
254 \defid nu,name=nu;  
255 \defid xi,name=xi;  
256 \defid pi,name=pi;  
257 \defid varpi,name=varpi;  
258 \defid rho,name=rho;  
259 \defid varrho,name=varrho;  
260 \defid sigma,name=sigma;  
261 \defid varsigma,name=varsigma;  
262 \defid tau,name=tau;  
263 \defid upsilon,name=upsilon;  
264 \defid phi,name=phi;  
265 \defid varphi,name=varphi;
```

```
266 defid chi,name=chi;
267 defid psi,name=psi;
268 defid omega,name=omega;
269 defid Gamma,name=Gamma;
270 defid Delta,name=Delta;
271 defid Theta,name=Theta;
272 defid Lambda,name=Lambda;
273 defid Xi,name=Xi;
274 defid Pi,name=Pi;
275 defid Sigma,name=Sigma;
276 defid Upsilon,name=Upsilon;
277 defid Phi,name=Phi;
278 defid Psi,name=Psi;
279 defid Omega,name=Omega;

280 defindex e_(down,down);
281 defid e_,name="e";
282 defindex d_(arg,down,down);
283 defid d_,name="D";
284 defindex u(down);
285 defid u1,name="u\sb1";
286 defid u2,name="u\sb2";
287 defid u3,name="u\sb3";
288 defid u4,name="u\sb4";
289 defid u5,name="u\sb5";
290 defid u6,name="u\sb6";
291 defid u7,name="u\sb7";
292 defid u8,name="u\sb8";
293 defid u9,name="u\sb9";
294 defindex s(down);
295 defid exp,name="\exp";
296 defindex exp(arg);
```

Can we write the system? Not in matrices apparently. So define a dummy array `tmp_` that we use to get the correct symbol typeset.

```

297 array tmp_(n),tmp_s(m),tmp_z(m);
298 defindex tmp_(down);
299 defindex tmp_s(down);
300 defindex tmp_z(down);
301 defid tmp_,name="\dot u";
302 defid tmp_s,name="\vec e";
303 defid tmp_z,name="\vec z";
304 rhs_:=rhsfn$
305 for k:=1:m do tmp_s(k):={for j:=1:n collect ee(j,k),exp(eval_(k))};
306 for k:=1:m do tmp_z(k):={for j:=1:n collect zz(j,k),exp(eval_(k))};

```

We have to be shifty here because `rlfi` does not work inside a loop: so write the commands to a file, and then input the file.

```

307 out "scratchfile.red";
308 write "off echo;$ % do not understand why needed in 2021??"
309 write "write ""\)"
310 \paragraph{The specified dynamical system}
311 \("";";
312 for j:=1:n do write "tmp_(",j,"):=coeffn(rhs_,e_(",j,",1),1);";
313 write "write ""\)"
314 \paragraph{Invariant subspace basis vectors}
315 \("";";
316 for j:=1:m do write "tmp_s(",j,"):=tmp_s(",j,")";";
317 for j:=1:m do write "tmp_z(",j,"):=tmp_z(",j,")";";
318 write "end;";
319 shut "scratchfile.red";

```

Now print the dynamical system to the LaTeX sub-file.

```

320 write "Ignore the following 15 lines of LaTeX"$
321 on latex$
322 out "invarManReportSys.tex"$
323 in "scratchfile.red"$
324 shut "invarManReportSys.tex"$
325 off latex$

```

4 Linear approximation to the invariant manifold

But first, write out the possibly adjusted nonlinear right-hand side function. According to the manual, this will append to the earlier output to the file.

```
326 write "Analyse ODE/DDE system du/dt = ",ff;
```

Parametrise the invariant manifold in terms of these amplitudes. For this substitution to work, `gg_` *cannot* be declared scalar as then it gets replaced by zero here and throughout.

```
327 clear gg_;
328 operator s; depend s,t;
329 let df(s(~j),t)=>coeffn(gg_,e_(j,1),1);
```

Invoke the following procedure to substitute whatever the current approximation is into (nonlinear) expressions.

```
330 procedure manifold_(uu);
331     for j:=1:n collect mkid(u,j)=coeffn(uu,e_(j,1),1)$
```

The linear approximation to the invariant manifold must be the following corresponding to the eigenvalues down the diagonal (even if zero). The amplitudes s_j are slowly evolving as they are either slow modes, or the complex amplitudes of oscillating modes.

```
332 uu_:=for j:=1:m sum s(j)*exp(eval_(j)*t)
333     *(for k:=1:n sum e_(k,1)*ee(k,j))$
334 gg_:=0$
335 if trace then write uu_:=uu_;
```

For some temporary trace printing, where for simplicity `small` is replaced by `s`.

```
336 procedure matify(a,m,n)$
337     begin matrix z(m,n);
338     for i:=1:m do for j:=1:n do z(i,j):=coeffn(a,e_(i,j),1);
339     return (z where {exp(0)=>1,small=>s});
340     end$
```

For the isochron may need to do something different with eigenvalues, but this should work as the inner product is complex conjugate transpose. The `pp` matrix is proposed to place the projection residuals in the range of the isochron.

```
341 zs:=for j:=1:m sum exp(eval_(j)*t)
342   *(for k:=1:n sum e_(k,j)*zz(k,j))$
343 pp:=0$
```

5 Iteratively construct the invariant manifold

But first establish the Taylor series in any delay factors of slow amplitudes.

```
344 let d_(s(~k),t,~dt)=>s(k)+(for n:=1:toosmall sum
345   (-dt)^n*df(s(k),t,n)/factorial(n));
```

Truncate expansions to specified order of error (via loop index trick).

```
346 for j:=toosmall:toosmall do let small^j=>0;
```

Iteratively construct the invariant manifold.

```
347 write "Start iterative construction of invariant manifold";
348 for iter:=1:maxiter do begin
349   if trace then write "
350   ITERATION = ",iter,"
351   -----";
```

Compute residual vector (matrix) of the dynamical system [Roberts \(1997\)](#).

```
352 resde:=-df(uu_,t)+sub(manifold_(uu_),rhsfn);
353 if trace then write "resde=",matify(resde,n,1);
```

Get the local directions of the coordinate system on the curving manifold:
store transpose as $m \times n$ matrix.

```
354 est:=tpe_(for j:=1:m sum df(uu_,s(j))*e_(1,j),e_);
355 est:=conj_(est);
356 if trace then write "est=",matify(est,m,n);
```

Compute residual matrix for the isochron projection [Roberts \(1989, 2000\)](#). But for the moment, only do it if the `eval_set` is for slow manifolds.

```
357 if itisSlowMan_ then begin
358     jacadj:=conj_(sub(manifold_(uu_),rhsjact));
359     if trace then write "jacadj=",matify(jacadj,n,n);
360     resd:=df(zs,t)+jacadj*zs+zs*pp;
361     if trace then write "resd=",matify(resd,n,m);
```

Compute residual of the normalisation of the projection.

```
362     resz:=est*zs-eyem*exp(0);
363     if trace then write "resz=",matify(resz,m,m);
364 end else resd:=resz:=0; % for when not slow manifold
```

Write lengths of residuals as a trace print (remember that the expression 0 has length one).

```
365 write lengthRes:=map(length(~a),{resde,resd,resz});
```

Solve for updates—all the hard work is already encoded in the operators.

```
366 uu_:=uu_+l_inv(resde,exp);
367 gg_:=gg_+g_inv(resde,exp);
368 if trace then write "gg=",matify(gg_,m,1);
369 if trace then write "uu=",matify(uu_,n,1);
```

Now update the isochron projection, with normalisation.

```
370 if itisSlowMan_ then begin
371     zs:=zs+l1_inv(resd,exp)-l2_inv(resz,exp);
372     pp:=pp-g1_inv(resd,exp)+youshouldnotseethis*g2_inv(resz,exp);
373     if trace then write "zs=",matify(zs,n,m);
374     if trace then write "pp=",matify(pp,m,m);
375 end;
```

Terminate the loop once residuals are zero.

```
376 showtime;
377 if {resde,resd,resz}={0,0,0} then write iter:=iter+10000;
```

```
378 end;
```

Only proceed to print if terminated successfully.

```
379 if {resde,resd,resz}={0,0,0}
380   then write "SUCCESS: converged to an expansion"
381   else <<write "FAILED TO CONVERGE; I EXIT";
382   return; >>;
```

6 Output text version of results

Once construction is finished, simplify `exp(0)`.

```
383 let exp(0)=>1;
```

Invoking `switch complex` improves some of the output of the complex numbers, but wrecks other parts of the output. Best left off.

Write text results.

```
384 write "The invariant manifold is (to one order lower)";
385 for j:=1:n do write "u",j," = ",
386   coeffn(small*uu_,e_(j,1),1)/small;
387 write "The evolution of the real/complex amplitudes";
388 for j:=1:m do write "ds(",j,")/dt = ",
389   coeffn(gg_,e_(j,1),1);
```

Optionally write the projection vectors.

```
390 if itisSlowMan_ then begin
391   write "The normals to the isochrons at the slow manifold.
392 Use these vectors: to project initial conditions
393 onto the slow manifold; to project non-autonomous
394 forcing onto the slow evolution; to predict the
395 consequences of modifying the original system; in
396 uncertainty quantification to quantify effects on
397 the model of uncertainties in the original system.";
398   for j:=1:m do write "z",j," = ",
```



```

399     for i:=1:n collect coeffn(zs,e_(i,j),1);
400 end;

```

Write text results numerically evaluated when expressions are long.

```

401 if length(gg_)>30 then begin
402   on rounded; print_precision 4$
403   write "Numerically, the invariant manifold is (to one order lower
404   for j:=1:n do write "u",j," = ",
405     coeffn(small*uu_,e_(j,1),1)/small;
406   write "Numerically, the evolution of the real/complex amplitudes
407   for j:=1:m do write "ds(",j,")/dt = ",
408     coeffn(gg_,e_(j,1),1);
409   if itisSlowMan_ then begin
410     write "Numerically, normals to isochrons at slow manifold.";
411     for j:=1:m do write "z",j," = ",
412       for i:=1:n collect coeffn(zs,e_(i,j),1);
413   end;
414   off rounded;
415 end;

```

7 Output LaTeX version of results

Change the printing of temporary arrays.

```

416 array tmp_zz(m,n);
417 defid tmp_,name="u";
418 defid tmp_s,name="\dot s";
419 defid tmp_z,name="\vec z";
420 defid tmp_zz,name="z";
421 defindex tmp_zz(down,down);

```

Gather complicated result

```

422 %for k:=1:m do tmp_z(k):=for j:=1:n collect (1*coeffn(zs,e_(j,k)
423 for k:=1:m do for j:=1:n do tmp_zz(k,j):=(1*coeffn(zs,e_(j,k),1))

```

Write to a file the commands needed to write the LaTeX expressions. Write the invariant manifold to one order lower than computed.

```

424 out "scratchfile.red";
425 write "off echo;"$ % do not understand why needed in 2021??
426 write "write ""\
427 \paragraph{The invariant manifold}
428 These give the location of the invariant manifold in
429 terms of parameters~\((s\sb j)\).
430 \("";";
431 for j:=1:n do write "tmp_(",j,
432   "):=coeffn(small*uu_,e_(",j,",1),1)/small;";
433 if length(gg_)>30 then begin
434 write "on rounded; print_precision 4$$$
435 for j:=1:n do write "tmp_(",j,
436   "):=coeffn(small*uu_,e_(",j,",1),1)/small;";
437 write "off rounded;"$
438 end;

```

Write the commands to write the ODEs on the invariant manifold.

```

439 write "write ""\
440 \paragraph{Invariant manifold ODEs}
441 The system evolves on the invariant manifold such
442 that the parameters evolve according to these ODEs.
443 \("";";
444 for j:=1:m do write "tmp_s(",j,"):=1*coeffn(gg_,e_(",j,",1),1);"
445 if length(gg_)>30 then begin
446 write "on rounded; print_precision 4$$$
447 for j:=1:m do write "tmp_s(",j,"):=1*coeffn(gg_,e_(",j,",1),1);"
448 write "off rounded;"$
449 end;

```

Optionally write the commands to write the projection vectors on the slow manifold.

```

450 if itisSlowMan_ then begin

```

```

451   write "write ""\
452 \paragraph{Normals to isochrons at the slow manifold}
453 Use these vectors: to project initial conditions
454 onto the slow manifold; to project non-autonomous
455 forcing onto the slow evolution; to predict the
456 consequences of modifying the original system; in
457 uncertainty quantification to quantify effects on
458 the model of uncertainties in the original system.
459 The normal vector \(\vec z_{\text{sb } j:=(z_{\text{sb }{j1}},\ldots,z_{\text{sb }{jn}})}\)
460 \("";";
461   for i:=1:m do for j:=1:n do
462     write "tmp_zz(",i,",",j,"):=tmp_zz(",i,",",j,");";
463   end;

```

Finish the scratchfile.

```

464 write "end;";
465 shut "scratchfile.red";

```

Execute the scratchfile with the required commands, with output to the main invariant manifold LaTeX file.

```

466 out "invarManReport.tex"$
467 on latex$
468 in "scratchfile.red"$
469 off latex$
470 shut "invarManReport.tex"$

```

8 *Fin*

That's all folks, so end the procedure.

```

471 return Finished_constructing_invariant_manifold_of_system$
472 end$

```

9 Override some system procedures

Bad luck if these interfere with anything else a user might try to do afterwards!

First define how various tokens get printed.

```
473 %load_package rlfi; %must be loaded early
474 deflist('(( ( !\b!i!g! ) ( ! ) !\b!i!g! )) ( !P!I !\p!i! )
475          (!p!i !\p!i! ) (!E !e) (!I !i) (e !e) (i !i)), 'name)$
```

Override the procedure that prints annoying messages about multicharacter symbols. It ends the output of one expression. This is just a copy from `rlfi.red` with the appropriate if-statement deleted. While interfering, hardcode that the mathematics is in inline mode.

```
476 %write "Ignore immediately following messages";
477 symbolic procedure prinlaend;
478 <<terpri();
479   prin2t "\)\par";
480   if !*verbatim then
481     <<prin2t "\begin{verbatim}";
482     prin2t "REDUCE Input:">>;
483   ncharspr!*:0;
484   if ofl!* then linelength(car linel!*)
485     else laline!*:cdr linel!*;
486   nochar!*:append(nochar!*,nochar1!*);
487   nochar1!*:nil >>$
488   %
```

Similarly, hardcode at the beginning of expression output that the mathematics is in inline mode.

```
489 symbolic procedure prinlabegin;
490 % Initializes the output
491 <<if !*verbatim then
492   <<terpri();
493   prin2t "\end{verbatim}">>;
494   linel!*:linelength nil . laline!*;
```

```

495   if ofl!* then linelength(laline!* + 2)
496       else laline!*: =car linel!* - 2;
497   prin2 "\(" >>$

```

Override the procedure that outputs the L^AT_EX preamble upon the command `on latex`. Presumably modified from that in `rlfi.red`. Use it to write a decent header that we use for one master file.

```

498 symbolic procedure latexon;
499 <<!*!*a2sfn:='texaeval;
500   !*raise:=nil;
501   prin2t "\documentclass[11pt,a5paper]{article}";
502   prin2t "\usepackage[a5paper,margin=13mm]{geometry}";
503   prin2t "\usepackage[parskip,time] \raggedright";
504   prin2t "\def\exp\big(#1\big){\, {\rm e}^{\#1}}";
505   prin2t "\def\eps{\varepsilon}";
506   prin2t "\title{Invariant manifold of your dynamical system}";
507   prin2t "\author{A. J. Roberts, University of Adelaide\\}";
508   prin2t "\texttt{http://orcid.org/0000-0001-8930-1552}}";
509   prin2t "\date{\now, \today}";
510   prin2t "\begin{document}";
511   prin2t "\maketitle";
512   prin2t "Throughout and generally: the lowest order, most";
513   prin2t "important, terms are near the end of each expression.";
514   prin2t "\input{invarManReportSys}";
515   if !*verbatim then
516       <<prin2t "\begin{verbatim}";
517       prin2t "REDUCE Input:">>;
518   put('tex,'rtypefn,'(lambda(x) 'tex)) >>$

```

End the file when read by Reduce

```

519 end;

```

References

- Roberts, A. J. (1989), ‘Appropriate initial conditions for asymptotic descriptions of the long term evolution of dynamical systems’, *J. Austral. Math. Soc. B* **31**, 48–75.
- Roberts, A. J. (1997), ‘Low-dimensional modelling of dynamics via computer algebra’, *Computer Phys. Comm.* **100**, 215–230.
- Roberts, A. J. (2000), ‘Computer algebra derives correct initial conditions for low-dimensional dynamical models’, *Computer Phys. Comm.* **126**(3), 187–206.