

A general invariant manifold construction algorithm, including isochrons of slow manifolds

A. J. Roberts*

Nov 2013 – April 5, 2021

Abstract

This procedure constructs a specified invariant manifold for a specified system of ordinary differential equations or delay differential equations. The invariant manifold may be any of a centre manifold, a slow manifold, an un/stable manifold, a sub-centre manifold, a nonlinear normal form, any spectral submanifold, or indeed a normal form coordinate transform of the entire state space. Thus the procedure may be used to analyse pitchfork bifurcations, or oscillatory Hopf bifurcations, or any more complicated superposition. In the cases when the neglected spectral modes all decay, the constructed invariant manifold supplies a faithful large time model of the dynamics of the differential equations. Further, in the case of a slow manifold, this procedure now derives vectors defining the projection onto the invariant manifold along the isochrons: this projection is needed for initial conditions, forcing, system modifications, and uncertainty quantification.

Contents

1	Introduction	2
1.1	A simple example: <code>exampleslowman()</code>	4
1.2	Header of the procedure	5

*School of Mathematical Sciences, University of Adelaide, South Australia 5005, AUSTRALIA. <https://profajroberts.github.io/>

1	<i>Introduction</i>	2
1.3	Preamble to the procedure	6
1.4	Check the dimensionality of specified system	7
2	Dissect the linear part	8
2.1	Normalise the adjoint eigenvectors	9
2.2	Operator to represent delays	9
2.3	Linearise at the origin	10
2.4	Eigen-check	11
2.5	Ameliorate the nonlinearity	12
2.6	Store invariant manifold frequencies	13
2.7	Precompute matrices for updates	14
2.8	Define operators that invoke these inverses	16
3	Initialise LaTeX output	18
4	Linear approximation to the invariant manifold	21
5	Iteratively construct the invariant manifold	22
6	Output text version of results	24
7	Output LaTeX version of results	26
8	Fin	28
9	Override some system procedures	28

1 Introduction

Download and install the computer algebra package *Reduce* via <http://www.reduce-algebra.com> Start-up *Reduce* and load the procedure by executing the command `in_tex "invariantManifold.tex"$`¹

Thereafter, construct a specified invariant manifold of a specific dynamical

¹This script changes a lot of internal settings of *Reduce*, so best only to do when needed.

system by executing the following command with specific values for the input parameters.

```
1 invariantmanifold(odefns, evals, evecs, adjvecs, toosmall);
```

Inputs As in the example of the next [Section 1.1](#), input parameters to the procedure are the following:

- **odefns**, a comma separated list within `mat(...)`, the RHS expressions of the ODEs/DDEs of the system, a system expressed in terms of variables `u1, u2, ...`, for time derivatives $du1/dt, du2/dt, \dots$;
any time delayed variables in the RHS are coded by the time-delay in parenthesis after the variable, as in the example `u1(pi/2)` to represent $u_1(t - \pi/2)$ in the DDEs;
- **evals**, a comma separated list within `mat(...)`, the eigenvalues of the modes to be the basis for the invariant manifold;
- **evecs**, a comma separated list of vectors within `mat(...)`—each vector a comma separated list of components within `(...)`, the eigenvectors of the modes corresponding to the given eigenvalues of the invariant manifold basis;
- **adjvecs**, a comma separated list of vectors within `mat(...)`, the adjoint eigenvectors of the modes corresponding to the given eigenvalues of the invariant manifold basis??;
- **toosmall**, an integer giving the desired order of error in the asymptotic approximation that is constructed. The procedure embeds the specified system in a family of systems parametrised by ε , and constructs an invariant manifold, and evolution thereon, of the embedding system to the asymptotic error $\mathcal{O}(\varepsilon^{\text{toosmall}})$ (as $\varepsilon \rightarrow 0$). Often the introduced artificial ε has a useful physical meaning, but strictly you should evaluate the output at $\varepsilon = 1$ to recover results for the specified system, and then interpret the results in terms of actual ‘small’ parameters.

Outputs This procedure reports the specified system, the embedded system it actually analyses, the number of iterations taken, the invariant manifold approximation, the evolution on the invariant manifold, and optionally a basis for projecting onto the invariant manifold.

- A plain text report to the Terminal window in which **Reduce** is executing—the invariant manifold is parametrised by variables $s(1)$, $s(2)$, \dots , and the dynamics by their evolution in time.
- A L^AT_EX source report written to the file `invarManReport.tex` (and `invarManReportSys.tex`)—the invariant manifold is parametrised by variables s_1, s_2, \dots , and the dynamics by their evolution in time.

One may change the appearance of the output somewhat. For example, it is often useful to execute `factor s`; before executing `invariantmanifold(...)` in order to group terms with the same powers of amplitudes/order-parameters/coarse-variables.

1.1 A simple example: `exampleslowman()`

Execute this example by invoking the command `exampleslowman()`; The example system to analyse is specified to be

$$\dot{u}_1 = -u_1 + u_2 - u_1^2, \quad \dot{u}_2 = u_1 - u_2 + u_2^2.$$

```
2 procedure exampleslowman;
3   invariantmanifold(
4     mat((-u1+u2-u1^2,u1-u2+u2^2)),
5     mat((0)),
6     mat((1,1)),
7     mat((1,1)),
8     5)$
```

We seek the slow manifold so specify the eigenvalue zero. From the linearisation matrix $\begin{bmatrix} -1 & 1 \\ 1 & -1 \end{bmatrix}$ a corresponding eigenvector is $\vec{e} = (1, 1)$, and corresponding left-eigenvector is $\vec{z} = \vec{e} = (1, 1)$, as specified. The last parameter specifies to construct the slow manifold to errors $\mathcal{O}(\varepsilon^5)$.

The procedure actually analyses the embedding system

$$\dot{u}_1 = -u_1 + u_2 - \varepsilon u_1^2, \quad \dot{u}_2 = u_1 - u_2 + \varepsilon u_2^2.$$

So here the artificial parameter ε has a physical interpretation in that it counts the nonlinearity: a term in ε^p will be a $(p + 1)$ th order term in $\vec{u} = (u_1, u_2)$. Hence the specified error $\mathcal{O}(\varepsilon^5)$ is here the same as error $\mathcal{O}(|\vec{u}|^6)$.

The constructed slow manifold is, in terms of the parameter s_1 (and reverse ordering!),

$$\begin{aligned} u_1 &= 3/8\varepsilon^3 s_1^4 - 1/2\varepsilon s_1^2 + s_1, \\ u_2 &= -3/8\varepsilon^3 s_1^4 + 1/2\varepsilon s_1^2 + s_1. \end{aligned}$$

On this slow manifold the evolution is

$$\dot{s}_1 = -3/4\varepsilon^4 s_1^5 + \varepsilon^2 s_1^3 :$$

here the leading term in s_1^3 indicates the origin is unstable. To project initial conditions onto the slow manifold, or non-autonomous forcing, or modifications of the original system, or to quantify uncertainty, use the projection defined by the derived vector

$$\vec{z}_1 = \begin{bmatrix} z_{11} \\ z_{12} \end{bmatrix} = \begin{bmatrix} 3/2\varepsilon^4 s_1^4 + 3/4\varepsilon^3 s_1^3 - 1/2\varepsilon^2 s_1^2 - 1/2\varepsilon s_1 + 1/2 \\ 3/2\varepsilon^4 s_1^4 - 3/4\varepsilon^3 s_1^3 - 1/2\varepsilon^2 s_1^2 + 1/2\varepsilon s_1 + 1/2 \end{bmatrix}.$$

Evaluate these at $\varepsilon = 1$ to apply to the original specified system, or here just interpret ε as a way to count the order of each term.

1.2 Header of the procedure

Need a couple of things established before defining the procedure: the `rlfi` package; and operator names for the variables of the dynamical system (in case they have delays), currently code a max of nine variables.

```
9 load_package rlfi;
10 operator u1,u2,u3,u4,u5,u6,u7,u8,u9;
```

Now define the procedure as an operator so we can be flexible with its arguments.

```
11 operator invariantmanifold;
12 for all odefns, evals, evecs, adjvecs, toosmall let
13   invariantmanifold(odefns, evals, evecs, adjvecs, toosmall)
14   = begin
```

1.3 Preamble to the procedure

Operators and arrays are always global, but we can make variables and matrices local, except for matrices that need to be declared `matrix`. So, move to implement all arrays and operators to have underscores, and almost all scalars and most matrices to be declared here.

```
15 scalar ff, freqm, ee, zz, maxiter, ff0, trace, ll, uvec,
16 reslin, ok, rhsjact, jacadj, resd, resde, resz, rhsfn, zs,
17 pp, est, eyem;
```

Transpose the defining matrices so that vectors are columns.

```
18 ff := tp odefns;
19 freqm := -i*evals;
20 ee := tp evecs;
21 zz := tp adjvecs;
```

Define default parameters for the iteration: `maxiter` is the maximum number of allowed iterations; Specific problems may override these defaults.

```
22 maxiter:=29$
23 factor small;
```

For optional trace printing of test cases: comment out second line when not needed.

```
24 trace:=0$
25 %trace:=1; maxiter:=5;
```

The `rationalize` switch makes code much faster with complex numbers. The switch `gcd` seems to wreck convergence, so leave it off.

```
26 on div; off allfac; on revpri;
27 on rationalize;
```

Propose to use `e_` as basis vector for matrices and vectors. Declare it non-commutative so that multiplication does not commute.

```
28 operator e_;
29 noncom e_;
30 factor e_;
31 let { e_(~j,~k)*e_(~l,~m)=>0 when k neq l
32      , e_(~j,~k)*e_(~l,~m)=>e_(j,m) when k=l
33      , e_(~j,~k)^2=>0 when j neq k
34      , e_(~j,j)^2=>e_(j,j) };
```

Also need a transpose operator: do complex conjugation explicitly when needed.

```
35 operator tpe_; linear tpe_;
36 let tpe_(e_(~i,~j),e_)=>e_(j,i);
```

Empty the output LaTeX file in case of error.

```
37 out "invarManReport.tex";
38 write "This empty document indicates error.";
39 shut "invarManReport.tex";
```

1.4 Check the dimensionality of specified system

Extract dimension information from the specification of the dynamical system: seek mD invariant manifold of an nD system.

```
40 write "total no. of variables ",
41 n:=part(length(ee),1);
42 write "no. of invariant modes ",
43 m:=part(length(ee),2);
```

```

44 if {length(freqm),length(zz),length(ee),length(ff)}
45   ={{1,m},{n,m},{n,m},{n,1}}
46   then write "Input dimensions are OK"
47   else <<write "INCONSISTENT INPUT DIMENSIONS, I EXIT";
48       return>>;

```

For the moment limit to a maximum of nine components.

```

49 if n>9 then <<write "SORRY, MAX NUMBER ODEs IS 9, I EXIT";
50     return>>;

```

Need an $m \times m$ identity matrix for normalisation of the isochron projection.

```

51 eyem:=for j:=1:m sum e_(j,j)$

```

2 Dissect the linear part

Define complex exponential $\text{cis}(u) = e^{iu}$. Do not (yet) invoke the simplification of $\text{cis}(0)$ as I want it to label modes of no oscillation, zero frequency.

```

52 operator cis;
53 let { df(cis(~u),t) => i*df(u,t)*cis(u)
54       , cis(~u)*cis(~v) => cis(u+v)
55       , cis(~u)^~p => cis(p*u)
56       };

```

Need function `conj_` to do parsimonious complex conjugation.

```

57 operator cis__;
58 procedure conj_(a)$ begin scalar aa;
59     aa:=(a where {i=>i__, cis(~b)=>cis__(b) when b neq 0})$
60     return (aa where {i__=>-i,cis__(~b)=>cis(-b)})$
61     end$

```

Make another array of frequencies for simplicity.

```

62 array freq_(m);
63 for j:=1:m do freq_(j):=freqm(1,j);

```


2.1 Normalise the adjoint eigenvectors

When we include delay differential equations, then we need to account for the history of the eigenvector as well. Hence multiply each eigenvector by its oscillating factor, $e^{i\omega t}$, and then take the mean. This multiplication by its oscillating factor should not make any difference for non-delay equations by the natural orthogonality of left and right eigenvectors of different eigenvalues. Reduce implements `conj` via `repart` and `impart`, so let `repart` do the conjugation of the cis factors.

Note: the ‘left eigenvectors’ have to be the eigenvectors of the complex conjugate transpose, and for the complex conjugate frequency. This seems best: for example, when the linear operator is $\begin{bmatrix} 0 & 1 \\ -1 & 0 \end{bmatrix}$ then the adjoint and the right eigenvectors are the same.

For un/stable manifolds we have to cope with complex frequencies. Seems to need `zz` to have complex conjugated frequency so store in `ccis_`—which is the same as `dcis_` for real frequencies.

```

64 matrix aa_(m,m),dcis_(m,m),ccis_(m,m);
65 for j:=1:m do dcis_(j,j):=cis(freq_(j)*t);
66 for j:=1:m do ccis_(j,j):=cis(conj_(freq_(j))*t);
67 aa_:= (tp map(conj_(~b),ee*dcis_)*zz*ccis_ )$
68 write "Normalising the left-eigenvectors:";
69 aa_:= (aa_ where {cis(0)=>1, cis(~a)=>0 when a neq 0})$
70 if det(aa_)=0 then << write
71     "ORTHOGONALITY ERROR IN EIGENVECTORS; I EXIT";
72     return>>;
73 zz:=zz*aa_^(-1);

```

2.2 Operator to represent delays

Introduce an operator to represent delay factors more conveniently for analysis.

```

74 operator d_; linear d_;

```

```

75 let { d_(~a~^p,t,~dt)=>d_(a,t,dt)^p
76      , d_(~a*~b,t,~dt)=>d_(a,t,dt)*d_(b,t,dt)
77      , d_(cis(~a),t,~dt)=>cis(a)
78          *sub(t=-dt,cos(a)+i*sin(a))
79      , df(d_(~a,t,~dt),~b)=>d_(df(a,b),t,dt)
80      , d_(~a,t,0)=>a
81      , d_(d_(~a,t,~dta),t,~dtb)=>d_(a,t,dta+dtb)
82      };

```

Now rewrite the (delay) factors in terms of this operator. Need to say that the symbol `u` depends upon time; later we write things into `u` and this dependence would be forgotten. For the moment limit to a maximum of nine ODEs.

```

83 somerules:={}$$
84 depend u1,t; somerules:=(u1(~dt)=d_(u1,t,dt)).somerules$
85 depend u2,t; somerules:=(u2(~dt)=d_(u2,t,dt)).somerules$
86 depend u3,t; somerules:=(u3(~dt)=d_(u3,t,dt)).somerules$
87 depend u4,t; somerules:=(u4(~dt)=d_(u4,t,dt)).somerules$
88 depend u5,t; somerules:=(u5(~dt)=d_(u5,t,dt)).somerules$
89 depend u6,t; somerules:=(u6(~dt)=d_(u6,t,dt)).somerules$
90 depend u7,t; somerules:=(u7(~dt)=d_(u7,t,dt)).somerules$
91 depend u8,t; somerules:=(u8(~dt)=d_(u8,t,dt)).somerules$
92 depend u9,t; somerules:=(u9(~dt)=d_(u9,t,dt)).somerules$
93 ff:=(ff where somerules)$

```

2.3 Linearise at the origin

Assume the equilibrium is at the origin. Find the linear operator at the equilibrium. Include `small=0` as we notionally adjoin it in the list of variables. Do not need to here make `small` any non-zero forcing at the equilibrium as it gets multiplied by `small` later?? For some reason using `mkid(u,k)=>0` does not resolve the `mkid`, but `mkid(u,k)=0` does; however, not clear if it is a problem??

```

94 ll:=ee*(tp ee)*0; %zero nxn matrix

```

```

95 uzero:=(for k:=1:n collect (mkid(u,k)=0))$
96 equilibrium:=(small=0).uzero$
97 for j:=1:n do for k:=1:n do begin
98   ll(j,k):=df(ff(j,1),mkid(u,k));
99   ll(j,k):=sub(equilibrium,ll(j,k));
100 end;
101 write "Find the linear operator is";
102 write ll:=ll;

```

We need a vector of unknowns for a little while.

```

103 uvec:=0*ff; %nx1 zero matrix
104 for j:=1:n do uvec(j,1):=mkid(u,j);

```

2.4 Eigen-check

Variable `aa_` appears here as the diagonal matrix of frequencies. Check that the frequencies and eigenvectors are specified correctly.

```

105 write "Check invariant subspace linearisation ";
106 for j:=1:m do for k:=1:m do aa_(j,k):=0;
107 for j:=1:m do aa_(j,j):=i*freq_(j);
108 reslin:=(ll*(ee*dcis_)-(ee*dcis_)*aa_
109   where cis(~a)*d_(1,t,~dt)=>sub(t=-dt,cos(a)+i*sin(a))*cis(a)
110 if trace then write reslin:=reslin;
111 ok:=1$
112 for j:=1:n do for k:=1:m do
113   ok:=if reslin(j,k)=0 then ok else 0$
114 if ok then write "Linearisation is OK";

```

Try to find a correction of the linear operator that is ‘close’. Multiply by the adjoint eigenvectors and then average over time: operator $\mathcal{L}_{\text{new}} := \mathcal{L} - \mathcal{L}_{\text{adj}}$ should now have zero residual. Lastly, correspondingly adjust the ODEs, since `lladj` does not involve delays we do not need delay operator transforms in the product.

```

115 if not ok then for iter:=1:2 do begin

```

```

116 write "WARNING: I NEED TO ADJUST LINEAR OPERATOR";
117 write
118 lladj:=reslin*tp map(conj_(~b),zz*ccis_);
119 write
120 lladj:=(lladj where {cis(0)=>1, cis(~a)=>0 when a neq 0});
121 write
122 ll:=ll-lladj;
123 write
124 reslin:=(ll*(ee*dcis_)-(ee*dcis_)*aa_
125     where cis(~a)*d_(1,t,~dt)=>sub(t=-dt,cos(a)+i*sin(a))*cis(a)
126 %for j:=1:n do for k:=1:m do
127 %    if reslin(j,k) neq 0 then << write
128 %    "OOPS, INCONSISTENT EIGENVALUES, EIGENVECTORS AND OPERATOR.
129 %    EMAIL ME; I QUIT"; write reslin:=reslin; rederr "aaaaah";qu
130 ok:=1$
131 for j:=1:n do for k:=1:m do
132     ok:=if reslin(j,k)=0 then ok else 0$
133 if ok then iter:=iter+1000;
134 end;
135 if not ok then << write
136     "OOPS, INCONSISTENT EIGENVALUES, EIGENVECTORS AND OPERATOR.
137     EMAIL ME; I EXIT";
138     return >>;

```

2.5 Ameliorate the nonlinearity

Anything not in the linear operator gets multiplied by `small` to be treated as small in the analysis. The feature of the second alternative is that when a user invokes `small` then the power of smallness is not then changed; however, causes issues in the relative scaling of some terms, so restore to the original version.

This might need reconsidering ?? but the current if always chooses the first simple alternative.

```

139 somerules:=for j:=1:n collect
140   (d_(1,t,~dt)*mkid(u,j)=d_(mkid(u,j),t,dt))$
141 ll0:=(ll*uvec where somerules)$
142 ff:=(if 1 then small*ff
143       else ff-(1-small)*sub(small=0,ff))
144   +(1-small)*ll0_$

```

Any constant term in the equations ff has to be multiplied by `cis(0)`.

```

145 ff0:=(ff where uzero)$
146 ff:=ff+(cis(0)-1)*ff0$

```

From the matrix versions of the equations, create algebraic form using the matrix basis.

```

147 rhsfn:=for i:=1:n sum e_(i,1)*ff(i,1)$

```

Also, create the algebraic form of the jacobian transpose using the matrix basis: take the conjugate later when used.

```

148 rhsjact:=for i:=1:n sum for j:=1:n sum
149   e_(j,i)*df(ff(i,1),mkid(u,j))$

```

2.6 Store invariant manifold frequencies

Extract all the frequencies in the invariant manifold, and the set of all the corresponding modes in the invariant manifold variables. The slow modes are accounted for as having zero frequency. Remember the frequency set is not in the ‘correct’ order. Array `modes_` stores the set of indices of all the modes of a given frequency.

```

150 array freq_s(m),modes_(m);
151 nfreq:=0$ freq_set:={}$
152 for j:=1:m do if not(freq_(j) member freq_set) then begin
153   nfreq:=nfreq+1;
154   freq_s(nfreq):=freq_(j);
155   freq_set:=freq_(j).freq_set;
156   modes_(nfreq):=for k:=j:m join

```

```

157     if freq_(j)=freq_(k) then {k} else {};
158 end;

```

Set a flag for the case of a slow manifold when all frequencies are zero, as then we compute the isochron projection. The next challenge is to get this isochron code working for the case of non-slow invariant manifolds.

```

159 itisSlowMan_:=if freq_set={0} then 1 else 0$
160 if trace then write itisSlowMan_:=itisSlowMan_;

```

Put in the non-singular general case as the zero entry of the arrays.

```

161 freq_s(0):=genfreq$
162 modes_(0):={}$

```

2.7 Precompute matrices for updates

Precompute matrices to solve for updates for each of the critical frequencies, and the general case $k = 0$. The matrix

$$\text{llzz}_- = \begin{bmatrix} -\mathcal{L} + \partial_t & \mathcal{E}_0 \\ \mathcal{Z}_0^\dagger & 0 \end{bmatrix}$$

and then put its inverse in place. Subsequently, extract the blocks for the generalised inverses and solvability condition into `l_invs` and `g_invs`.

```

163 matrix llzz_(n+m,n+m);
164 array l_invs(nfreq),g_invs(nfreq);
165 array l1_invs(nfreq),g1_invs(nfreq),l2_invs(nfreq),g2_invs(nfreq);
166 operator sp_; linear sp_;
167 for k:=0:nfreq do begin
168   if trace then write "ITERATION ",k:=k;

```

Code the operator $\mathcal{L}\hat{v}$ where the delay is to only act on the oscillation part.

```

169   for ii:=1:n do for jj:=1:n do llzz_(ii,jj):=(
170     -sub(small=0,ll(ii,jj))
171     where d_(1,t,~dt)=>cos(freq_s(k)*dt)-i*sin(freq_s(k)*dt));

```

Code the operator $\partial\hat{v}/\partial t$ where it only acts on the oscillation part.

```
172   for j:=1:n do llzz_(j,j):=i*freq_s(k)+llzz_(j,j);
```

Now code the part leading to the solvability condition which arises from allowing the (oscillation) amplitude to evolve. Use operator `sp_` to extract the delay parts that subtly affect the updates of the evolution.

```
173   for j:=1:length(modes_(k)) do
174     for ii:=1:n do llzz_(ii,n+j):=ee(ii,part(modes_(k),j))
175       +(for jj:=1:n sum
176         sp_(ll(ii,jj)*ee(jj,part(modes_(k),j)),d_)
177         where { sp_(1,d_)=>0
178               , sp_(d_(1,t,~dt),d_)=>dt*(
179                 cos(freq_s(k)*dt)-i*sin(freq_s(k)*dt))
180               });
```

Force the updates to be orthogonal to the left-eigenvectors in the complex conjugate transpose adjoint.

```
181   for ii:=1:length(modes_(k)) do
182     for j:=1:n do llzz_(n+ii,j):=conj_(zz(j,part(modes_(k),ii)))
183   if trace then write "finished Force the updates to be orthogonal"
```

Set the bottom-right corner of the matrix to zero.

```
184   for i:=1:length(modes_(k)) do
185     for j:=1:m do llzz_(n+i,n+j):=0;
```

Add some trivial rows and columns to make the matrix up to the same size for all frequencies.

```
186   for i:=length(modes_(k))+1:m do begin
187     for j:=1:n+i-1 do llzz_(n+i,j):=llzz_(j,n+i):=0;
188     llzz_(n+i,n+i):=1;
189   end;
190   if trace then write "finished Add some trivial rows and columns"
```

Invert the matrix and unpack into arrays ready for use by the inversion operators.

```

191   if trace then write llzz_:=llzz_;
192   llzz_:=llzz_^(-1);
193   if trace then write llzz_:=llzz_;
194   l_invs(k):=for i:=1:n sum for j:=1:n sum e_(i,j)*llzz_(i,j);
195   g_invs(k):=for i:=1:length(modes_(k)) sum
196     for j:=1:n sum e_(part(modes_(k),i),j)*llzz_(i+n,j);
197 if trace then write "finished Invert the matrix and unpack";

```

Unpack the conjugate transpose for inverse operators used for the isochrons. A difference here is that the orthogonality condition is non-trivial (in the slow manifold we assumed amplitudes were exactly orthogonal to the left-eigenvectors), so we need to remember more parts of the inverse of the matrix. Will it need to be more subtle for the un/stable manifolds case??

```

198 %   realgenfreq:={repart(genfreq)=>genfreq, impart(genfreq)=>0}$
199   l1_invs(k):=for ii:=1:n sum for j:=1:n sum
200     e_(ii,j)*conj_(llzz_(j,ii));
201   l2_invs(k):=for ii:=1:n sum for j:=1:length(modes_(k)) sum
202     e_(ii,part(modes_(k),j))*conj_(llzz_(j+n,ii));
203   g1_invs(k):=for ii:=1:length(modes_(k)) sum for j:=1:n sum
204     e_(part(modes_(k),ii),j)*conj_(llzz_(j,ii+n));
205   g2_invs(k):=
206     for ii:=1:length(modes_(k)) sum for j:=1:length(modes_(k)) sum
207       e_(part(modes_(k),ii),part(modes_(k),j))*conj_(llzz_(j+n,ii));
208   if trace then write "finished Unpack the conjugate transpose";
209 end;

```

2.8 Define operators that invoke these inverses

Decompose residuals into parts, and operate on each. First for the invariant manifold. But making $e_$ non-commutative means that it does not get factored out of these linear operators: must post-multiply by $e_$ because the linear inverse is a premultiply.

```

210 operator l_inv; linear l_inv;
211 let l_inv(e_(~j,~k)*cis(~a),cis)=>l_invproc(a/t)*e_(j,k);

```



```

212 procedure l_invproc(a);
213   if a member freq_set
214   then << k:=0;
215       repeat k:=k+1 until a=freq_s(k);
216       l_invs(k)*cis(a*t) >>
217   else sub(genfreq=a,l_invs(0))*cis(a*t)$

```

Second for the evolution on the invariant manifold.

```

218 operator g_inv; linear g_inv;
219 let g_inv(e_(~j,~k)*cis(~a),cis)=>ginv_proc(a/t)*e_(j,k);
220 procedure ginv_proc(a);
221   if a member freq_set
222   then << k:=0;
223       repeat k:=k+1 until a=freq_s(k);
224       g_invs(k) >>
225   else sub(genfreq=a,g_invs(0))$

```

Copy and adjust the above for the projection. But first define the generic procedure. Perhaps use conjugate/negative of the frequency when applying to the general case of oscillations—but it might already have been accounted for??

```

226 procedure inv_proc(a,invs);
227   if a member freq_set
228   then << k:=0;
229       repeat k:=k+1 until a=freq_s(k);
230       invs(k)*cis(a*t) >>
231   else sub(genfreq=a,invs(0))*cis(a*t)$

```

Then define operators that we use to update the projection.

```

232 operator l1_inv; linear l1_inv;
233 operator l2_inv; linear l2_inv;
234 operator g1_inv; linear g1_inv;
235 operator g2_inv; linear g2_inv;
236 let { l1_inv(e_(~j,~k)*cis(~a),cis)=>inv_proc(a/t,l1_invs)*e_(j,l
237       , l2_inv(e_(~j,~k)*cis(~a),cis)=>inv_proc(a/t,l2_invs)*e_(j,l

```

```

238     , g1_inv(e_(~j,~k)*cis(~a),cis)=>inv_proc(a/t,g1_invs)*e_(j,l)
239     , g2_inv(e_(~j,~k)*cis(~a),cis)=>inv_proc(a/t,g2_invs)*e_(j,l)
240   };

```

3 Initialise LaTeX output

Set the default output to be inline mathematics.

```
241 mathstyle math;
```

Define the Greek alphabet with small as well.

```

242 defid small,name="\eps";%varepsilon;
243 %defid small,name=varepsilon;
244 defid alpha,name=alpha;
245 defid beta,name=beta;
246 defid gamma,name=gamma;
247 defid delta,name=delta;
248 defid epsilon,name=epsilon;
249 defid varepsilon,name=varepsilon;
250 defid zeta,name=zeta;
251 defid eta,name=eta;
252 defid theta,name=theta;
253 defid vartheta,name=vartheta;
254 defid iota,name=iota;
255 defid kappa,name=kappa;
256 defid lambda,name=lambda;
257 defid mu,name=mu;
258 defid nu,name=nu;
259 defid xi,name=xi;
260 defid pi,name=pi;
261 defid varpi,name=varpi;
262 defid rho,name=rho;
263 defid varrho,name=varrho;
264 defid sigma,name=sigma;

```

```
265 defid varsigma,name=varsigma;
266 defid tau,name=tau;
267 defid upsilon,name=upsilon;
268 defid phi,name=phi;
269 defid varphi,name=varphi;
270 defid chi,name=chi;
271 defid psi,name=psi;
272 defid omega,name=omega;
273 defid Gamma,name=Gamma;
274 defid Delta,name=Delta;
275 defid Theta,name=Theta;
276 defid Lambda,name=Lambda;
277 defid Xi,name=Xi;
278 defid Pi,name=Pi;
279 defid Sigma,name=Sigma;
280 defid Upsilon,name=Upsilon;
281 defid Phi,name=Phi;
282 defid Psi,name=Psi;
283 defid Omega,name=Omega;

284 defindex e_(down,down);
285 defid e_,name="e";
286 defindex d_(arg,down,down);
287 defid d_,name="D";
288 defindex u(down);
289 defid u1,name="u\sb1";
290 defid u2,name="u\sb2";
291 defid u3,name="u\sb3";
292 defid u4,name="u\sb4";
293 defid u5,name="u\sb5";
294 defid u6,name="u\sb6";
295 defid u7,name="u\sb7";
296 defid u8,name="u\sb8";
297 defid u9,name="u\sb9";
298 defindex s(down);
```

```

299 defid cis,name="\cis";
300 defindex cis(arg);

```

Can we write the system? Not in matrices apparently. So define a dummy array `tmp_` that we use to get the correct symbol typeset.

```

301 array tmp_(n),tmp_s(m),tmp_z(m);
302 defindex tmp_(down);
303 defindex tmp_s(down);
304 defindex tmp_z(down);
305 defid tmp_,name="\dot u";
306 defid tmp_s,name="\vec e";
307 defid tmp_z,name="\vec z";
308 rhs_:=rhsfn$
309 for k:=1:m do tmp_s(k):={for j:=1:n collect ee(j,k),cis(freq_(k))};
310 for k:=1:m do tmp_z(k):={for j:=1:n collect zz(j,k),cis(freq_(k))};

```

We have to be shifty here because `rlfi` does not work inside a loop: so write the commands to a file, and then input the file.

```

311 out "scratchfile.red";
312 write "off echo;$ % do not understand why needed in 2021??"
313 write "write ""\"
314 \paragraph{The specified dynamical system}
315 \("";";
316 for j:=1:n do write "tmp_(",j,"):=coeffn(rhs_,e_(",j,",1),1);";
317 write "write ""\"
318 \paragraph{Invariant subspace basis vectors}
319 \("";";
320 for j:=1:m do write "tmp_s(",j,"):=tmp_s(",j,")";";
321 for j:=1:m do write "tmp_z(",j,"):=tmp_z(",j,")";";
322 write "end;";
323 shut "scratchfile.red";

```

Now print the dynamical system to the LaTeX sub-file.

```

324 write "Ignore the following 15 lines of LaTeX"$
325 on latex$

```

```

326 out "invarManReportSys.tex"$
327 in "scratchfile.red"$
328 shut "invarManReportSys.tex"$
329 off latex$

```

4 Linear approximation to the invariant manifold

But first, write out the possibly adjusted nonlinear right-hand side function. According to the manual, this will append to the earlier output to the file.

```

330 write "Analyse ODE/DDE system du/dt = ",ff;

```

Parametrise the invariant manifold in terms of these amplitudes. For this substitution to work, `gg_` *cannot* be declared scalar as then it gets replaced by zero here and throughout. What if we clear it??

```

331 clear gg_;
332 operator s; depend s,t;
333 let df(s(~j),t)=>coeffn(gg_,e_(j,1),1);

```

Invoke the following procedure to substitute whatever the current approximation is into (nonlinear) expressions.

```

334 procedure manifold_(uu);
335     for j:=1:n collect mkid(u,j)=coeffn(uu,e_(j,1),1)$

```

The linear approximation to the invariant manifold must be the following corresponding to the frequencies down the diagonal (even if zero). The amplitudes s_j are slowly evolving as they are either slow modes, or the complex amplitudes of oscillating modes.

```

336 uu_:=for j:=1:m sum s(j)*cis(freq_(j)*t)
337   *(for k:=1:n sum e_(k,1)*ee(k,j))$
338 gg_:=0$
339 if trace then write uu_:=uu_;

```

For some temporary trace printing??

```

340 procedure matify(a,m,n)$
341   begin matrix z(m,n);
342     for i:=1:m do for j:=1:n do z(i,j):=coeffn(a,e_(i,j),1);
343     return (z where {cis(0)=>1,small=>s});
344   end$

```

For the isochron may need to do something different with frequencies, but this should work as the inner product is complex conjugate transpose. The `pp` matrix is proposed to place the projection residuals in the range of the isochron.

```

345 zs:=for j:=1:m sum cis(freq_(j)*t)
346   *(for k:=1:n sum e_(k,j)*zz(k,j))$
347 pp:=0$

```

5 Iteratively construct the invariant manifold

But first establish the Taylor series in any delay factors of slow amplitudes.

```

348 let d_(s(~k),t,~dt)=>s(k)+(for n:=1:toosmall sum
349   (-dt)^n*df(s(k),t,n)/factorial(n));

```

Truncate expansions to specified order of error (via loop index trick).

```

350 for j:=toosmall:toosmall do let small^j=>0;

```

Iteratively construct the invariant manifold.

```

351 write "Start iterative construction of invariant manifold";
352 for iter:=1:maxiter do begin
353   if trace then write "
354   ITERATION = ",iter,"
355   -----";

```

Compute residual vector (matrix) of the dynamical system [Roberts \(1997\)](#).

```

356 %write rhsfn:=rhsfn;
357 %write manifolduu:=manifold_(uu_);

```

```

358 %write duudt:=df(uu_,t);
359 resde:=-df(uu_,t)+sub(manifold_(uu_),rhsfn);
360 if trace then write "resde=",matify(resde,n,1);

```

Get the local directions of the coordinate system on the curving manifold:
store transpose as $m \times n$ matrix.

```

361 est:=tpe_(for j:=1:m sum df(uu_,s(j))*e_(1,j),e_);
362 est:=conj_(est);
363 if trace then write "est=",matify(est,m,n);

```

Compute residual matrix for the isochron projection [Roberts \(1989, 2000\)](#).
But for the moment, only do it if the `freq_set` is for slow manifolds.

```

364 if itisSlowMan_ then begin
365     jacadj:=conj_(sub(manifold_(uu_),rhsjact));
366     if trace then write "jacadj=",matify(jacadj,n,n);
367     resd:=df(zs,t)+jacadj*zs+zs*pp;
368     if trace then write "resd=",matify(resd,n,m);

```

Compute residual of the normalisation of the projection.

```

369     resz:=est*zs-eyem*cis(0);
370     if trace then write "resz=",matify(resz,m,m);
371 end else resd:=resz:=0; % for when not slow manifold

```

Write lengths of residuals as a trace print (remember that the expression 0 has length one).

```

372 write lengthRes:=map(length(~a),{resde,resd,resz});

```

Solve for updates—all the hard work is already encoded in the operators.

```

373 uu_:=uu_+l_inv(resde,cis);
374 gg_:=gg_+g_inv(resde,cis);
375 if trace then write "gg=",matify(gg_,m,1);
376 if trace then write "uu=",matify(uu_,n,1);

```

Now update the isochron projection, with normalisation.

```

377 if itisSlowMan_ then begin

```

```

378 zs:=zs+l1_inv(resd,cis)-l2_inv(resz,cis);
379 pp:=pp-g1_inv(resd,cis)+youshouldnotseethis*g2_inv(resz,cis);
380 if trace then write "zs=",matify(zs,n,m);
381 if trace then write "pp=",matify(pp,m,m);
382 end;

```

Terminate the loop once residuals are zero.

```

383 showtime;
384 if {resde,resd,resz}={0,0,0} then write iter:=iter+10000;
385 end;

```

Only proceed to print if terminated successfully.

```

386 if {resde,resd,resz}={0,0,0}
387   then write "SUCCESS: converged to an expansion"
388   else <<write "FAILED TO CONVERGE; I EXIT";
389       return; >>;

```

6 Output text version of results

Once construction is finished, simplify `cis(0)`.

```

390 let cis(0)=>1;

```

Invoking `switch complex` improves some of the output of the complex numbers, but wrecks other parts of the output. Best left off.

Write text results.

```

391 write "The invariant manifold is (to one order lower)";
392 for j:=1:n do write "u",j," = ",
393   coeffn(small*uu_,e_(j,1),1)/small;
394 write "The evolution of the real/complex amplitudes";
395 for j:=1:m do write "ds(",j,")/dt = ",
396   coeffn(gg_,e_(j,1),1);

```

Optionally write the projection vectors.


```

397 if itisSlowMan_ then begin
398   write "The normals to the isochrons at the slow manifold.
399 Use these vectors: to project initial conditions
400 onto the slow manifold; to project non-autonomous
401 forcing onto the slow evolution; to predict the
402 consequences of modifying the original system; in
403 uncertainty quantification to quantify effects on
404 the model of uncertainties in the original system.";
405   for j:=1:m do write "z",j," = ",
406     for i:=1:n collect coeffn(zs,e_(i,j),1);
407 end;

```

Write text results numerically evaluated when expressions are long.

```

408 if length(gg_)>30 then begin
409   on rounded; print_precision 4$
410   write "Numerically, the invariant manifold is (to one order lower
411 for j:=1:n do write "u",j," = ",
412   coeffn(small*uu_,e_(j,1),1)/small;
413   write "Numerically, the evolution of the real/complex amplitudes
414 for j:=1:m do write "ds(",j,")/dt = ",
415   coeffn(gg_,e_(j,1),1);
416 if itisSlowMan_ then begin
417   write "Numerically, normals to isochrons at slow manifold.";
418   for j:=1:m do write "z",j," = ",
419     for i:=1:n collect coeffn(zs,e_(i,j),1);
420 end;
421 off rounded;
422 end;

```

There is an as yet unresolved problem in the typesetting when the argument of `cis` (frequency) is a rational number instead of integer: the numerator has an extra pair of parentheses which then makes the typesetting wrong; maybe we need a pre- \LaTeX filter??

7 Output LaTeX version of results

Change the printing of temporary arrays.

```
423 array tmp_zz(m,n);
424 defid tmp_,name="u";
425 defid tmp_s,name="\dot s";
426 defid tmp_z,name="\vec z";
427 defid tmp_zz,name="z";
428 defindex tmp_zz(down,down);
```

Gather complicated result

```
429 %for k:=1:m do tmp_z(k):=for j:=1:n collect (1*coeffn(zs,e_(j,k))
430 for k:=1:m do for j:=1:n do tmp_zz(k,j):=(1*coeffn(zs,e_(j,k)),1))
```

Write to a file the commands needed to write the LaTeX expressions. Write the invariant manifold to one order lower than computed.

```
431 out "scratchfile.red";
432 write "off echo;"$ % do not understand why needed in 2021??
433 write "write ""\
434 \paragraph{The invariant manifold}
435 These give the location of the invariant manifold in
436 terms of parameters~\((s\sb j)\).
437 \("";";
438 for j:=1:n do write "tmp_(",j,
439   "):=coeffn(small*uu_,e_(",j,",1),1)/small;";
440 if length(gg_)>30 then begin
441 write "on rounded; print_precision 4$"$
442 for j:=1:n do write "tmp_(",j,
443   "):=coeffn(small*uu_,e_(",j,",1),1)/small;";
444 write "off rounded;"$
445 end;
```

Write the commands to write the ODEs on the invariant manifold.

```
446 write "write ""\)
```

```

447 \paragraph{Invariant manifold ODEs}
448 The system evolves on the invariant manifold such
449 that the parameters evolve according to these ODEs.
450 \("");";
451 for j:=1:m do write "tmp_s(",j,"):=1*coeffn(gg_,e_(",j,",1),1);"
452 if length(gg_)>30 then begin
453 write "on rounded; print_precision 4$"$
454 for j:=1:m do write "tmp_s(",j,"):=1*coeffn(gg_,e_(",j,",1),1);"
455 write "off rounded;"$
456 end;

```

Optionally write the commands to write the projection vectors on the slow manifold.

```

457 if itisSlowMan_ then begin
458   write "write ""\)"
459 \paragraph{Normals to isochrons at the slow manifold}
460 Use these vectors: to project initial conditions
461 onto the slow manifold; to project non-autonomous
462 forcing onto the slow evolution; to predict the
463 consequences of modifying the original system; in
464 uncertainty quantification to quantify effects on
465 the model of uncertainties in the original system.
466 The normal vector \(\vec z\sb{j:=(z\sb{j1},\ldots,z\sb{jn})}\)
467 \("");";
468   for i:=1:m do for j:=1:n do
469     write "tmp_zz(",i,",",j,"):=tmp_zz(",i,",",j,")";
470 end;

```

Finish the scratchfile.

```

471 write "end;";
472 shut "scratchfile.red";

```

Execute the scratchfile with the required commands, with output to the main invariant manifold LaTeX file.

```

473 out "invarManReport.tex"$

```

```

474 on latex$
475 in "scratchfile.red"$
476 off latex$
477 shut "invarManReport.tex"$

```

8 Fin

That's all folks, so end the procedure.

```

478 return Finished_constructing_invariant_manifold_of_system$
479 end$

```

9 Override some system procedures

Bad luck if these interfere with anything else a user might try to do afterwards!

First define how various tokens get printed.

```

480 %load_package rlfi; %must be loaded early
481 deflist('(((! ( !\!b!i!g!() (!) !\!b!i!g!)) (!P!I !\!p!i! )
482          (!p!i !\!p!i! ) (!E !e) (!I !i) (e !e) (i !i)), 'name)$

```

Override the procedure that prints annoying messages about multicharacter symbols. It ends the output of one expression. This is just a copy from `rlfi.red` with the appropriate if-statement deleted. While interfering, hardcode that the mathematics is in inline mode.

```

483 %write "Ignore immediately following messages";
484 symbolic procedure prinlaend;
485 <<terpri();
486   prin2t "\)\par";
487   if !*verbatim then
488     <<prin2t "\begin{verbatim}";
489     prin2t "REDUCE Input:">>;
490   ncharspr!*:=0;

```

```

491   if ofl!* then linelength(car linel!*)
492       else laline!*:=(cdr linel!*);
493   nochar!*:=(append(nochar!*,nochar1!*);
494   nochar1!*:=(nil >>$
495   %

```

Similarly, hardcode at the beginning of expression output that the mathematics is in inline mode.

```

496 symbolic procedure prinlabegin;
497 % Initializes the output
498 <<if !*verbatim then
499     <<terpri();
500     prin2t "\end{verbatim}">>;
501   linel!*:=(linelength nil . laline!*);
502   if ofl!* then linelength(laline!* + 2)
503       else laline!*:=(car linel!* - 2);
504   prin2t "\" >>$

```

Override the procedure that outputs the L^AT_EX preamble upon the command `on latex`. Presumably modified from that in `rlfi.red`. Use it to write a decent header that we use for one master file.

```

505 symbolic procedure latexon;
506 <<!*!*a2sfn:='texaeval;
507   !*raise:=(nil;
508   prin2t "\documentclass[11pt,a5paper]{article}";
509   prin2t "\usepackage[a5paper,margin=13mm]{geometry}";
510   prin2t "\usepackage{parskip,time} \raggedright";
511   prin2t "\def\cis\big(#1\big){\,e^{\#1i}}";
512   prin2t "\def\eps{\varepsilon}";
513   prin2t "\title{Invariant manifold of your dynamical system}";
514   prin2t "\author{A. J. Roberts, University of Adelaide\\";
515   prin2t "\texttt{http://orcid.org/0000-0001-8930-1552}}";
516   prin2t "\date{\now, \today}";
517   prin2t "\begin{document}";
518   prin2t "\maketitle";

```

```

519  prin2t "Throughout and generally: the lowest order, most";
520  prin2t "important, terms are near the end of each expression.";
521  prin2t "\input{invarManReportSys}";
522  if !*verbatim then
523      <<prin2t "\begin{verbatim}";
524      prin2t "REDUCE Input:">>;
525  put('tex,'rtypefn,'(lambda(x) 'tex)) >>$

```

End the file when read by Reduce

```

526 end;

```

References

- Roberts, A. J. (1989), 'Appropriate initial conditions for asymptotic descriptions of the long term evolution of dynamical systems', *J. Austral. Math. Soc. B* **31**, 48–75.
- Roberts, A. J. (1997), 'Low-dimensional modelling of dynamics via computer algebra', *Computer Phys. Comm.* **100**, 215–230.
- Roberts, A. J. (2000), 'Computer algebra derives correct initial conditions for low-dimensional dynamical models', *Computer Phys. Comm.* **126**(3), 187–206.