

A general centre manifold construction algorithm for the web, including isochrons of slow manifolds

A. J. Roberts*

July 3, 2015

Abstract

This code is the heart and muscle of a web service. The web service derives a centre manifold of any specified system of ordinary differential equations or delay differential equations, when the system has fast and centre modes. The centre modes may be slow, as in a pitchfork bifurcation, or oscillatory, as in a Hopf bifurcation, or some more complicated superposition. In the case when the fast modes all decay, the centre manifold supplies a faithful large time model of the dynamics. Further, this code now derives vectors defining the projection onto the centre manifold along the isochrons: this projection is needed for initial conditions, forcing, system modifications, and uncertainty quantification.

Contents

1	Overall initialisation	3
2	Some example systems	5
2.1	Simple one variable delay differential equation	5
2.2	Another one variable delay differential equation	7

*School of Mathematical Sciences, University of Adelaide, South Australia 5005, AUSTRALIA. <http://www.maths.adelaide.edu.au/anthony.roberts/>

2.3	Separated delay differential equations	8
2.4	Linearly coupled 2D DDE	9
2.5	Double Hopf 2D DDE	10
2.5.1	Parameters $(\eta, \theta) = (3/5, 2)$	11
2.5.2	Parameters $(\eta, \theta) = (7/25, 4)$	12
2.6	Simple 2D ODE	14
2.6.1	The stable manifold	15
2.6.2	The slow-stable manifold	15
2.7	Four state Markov chain	16
2.8	Bifurcating 2D system	17
2.8.1	The stable manifold	18
2.8.2	The slow-stable manifold	19
2.9	Simple 3D system	20
2.9.1	Its 2D stable manifold with generalised eigenvectors	21
2.10	3D system with a generalised eigenvector	22
2.11	Separated system	23
2.12	Oscillatory centre manifold—separated form	24
2.13	Perturbed frequency oscillatory centre manifold—separated form	25
2.14	More general oscillatory centre manifold	26
2.15	Quasi-delay differential equation	27
2.16	Detuned version of quasi-delayed	28
2.17	Rossler-like system	29
2.18	Fudge a couple of these oscillations together	30
2.19	Fudge an oscillatory mode	32
2.20	Modulate Duffing oscillation	34
2.21	Modulate another oscillation	34
2.22	An example from Iulian Stoleriu	35
2.23	An second example from Iulian Stoleriu	36
2.24	Periodic chronic myelogenous leukemia	37
2.24.1	Delayed version	39
2.25	Nonlinear normal modes	40
2.26	Periodically forced van der Pol oscillator	42
2.27	Slow manifold of Lorenz 1986 model	43
2.27.1	Normal form shows drift from the fast waves	45

2.28 Check the dimensionality of specified system	45
3 Dissect the linear part	46
3.1 Normalise the adjoint eigenvectors	47
3.2 Operator to represent delays	48
3.3 Linearise at the origin	49
3.4 Eigen-check	49
3.5 Ameliorate the nonlinearity	51
3.6 Store invariant manifold frequencies	52
3.7 Precompute matrices for updates	52
3.8 Define operators that invoke these inverses	55
4 Initialise LaTeX output	56
5 Linear approximation to the invariant manifold	62
6 Iteratively construct the invariant manifold	63
7 Output text version of results	65
8 Output LaTeX version of results	67
9 Fin	69

1 Overall initialisation

In the following, assign `thecase:=myweb`; for the web service (or to read a system from file `cmsysb.red`), otherwise assign `thecase` to be any of the example dynamical systems in set `thecases`.

```

1 % see gcmafwFib.pdf for detailed explanation
2 % AJ Roberts, Nov 2013 -- Jul 2015
3 thecase:=substblem;
4 thecases:={onedde, anotherdde, twodde, dde2d, dde2d2ha,
5 dde2d2hb, simple2d, simple2ds, simple2dss, fourstatemarkov,
6 another2d, another2ds, simple3d, simple3ds, geneigenvec,
```

```

7 bifurcate2d, simpleosc, perturbfreq, nonseparatedosc,
8 quasidelayosc, quasidelayoscmod, rosslerlike, doubleosc,
9 oscmeanflow, modulateduffing, modulateoscillator,
10 StoleriuOne, StoleriuTwo, delayprolif, delayedprolif,
11 normalmodes, forcedvdp, lorenz86slow, lorenz86norm,
12 substablem }$

```

Define default parameters for the iteration: `maxiter_` is the maximum number of allowed iterations; `toosmall` is the order of errors in the analysis in terms of the parameter `small`. Specific problems may override these defaults.

```

13 maxiter_:=29$
14 factor small;
15 toosmall:=3$

```

For optional trace printing of test cases: comment out second line when not needed.

```

16 trace_:=0$
17 trace_:=1; maxiter_:=5;

```

The `rationalize` switch makes code much faster with complex numbers. The switch `gcd` seems to wreck convergence, so leave it off.

```

18 on div; off allfac; on revpri;
19 on rationalize;
20 linelength 60$

```

Propose to use `e_` as basis vector for matrices and vectors. Declare it non-commutative so that multiplication does not commute.

```

21 operator e_;
22 noncom e_;
23 factor e_;
24 let { e_(~j,~k)*e_(~l,~m)=>0 when k neq l
25      , e_(~j,~k)*e_(~l,~m)=>e_(j,m) when k=l
26      , e_(~j,~k)^2=>0 when j neq k
27      , e_(~j,j)^2=>e_(j,j) };

```

Also need a transpose operator: do complex conjugation explicitly when needed.

```
28 operator tpe_; linear tpe_;
29 let tpe_(e_(~i,~j),e_)=>e_(j,i);
```

Need to enter delayed factors in the ODEs, so use operators for the dependent variables in the dynamical system.

```
30 operator u;
```

Empty the output LaTeX file in case of error.

```
31 out "centreMan.tex";
32 write "This empty document indicates error.";
33 shut "centreMan.tex";
```

Automatically testing a set of examples does not yet work.

```
34 %foreach thecase in thecases do begin
```

2 Some example systems

Define the basic linear operator, centre manifold bases, and ‘nonlinear’ function. Note that Reduce’s matrix transpose does not take complex conjugate. Then the web service inputs the system from a file, otherwise get the system from one of the examples that follow.

```
35 if thecase=myweb then begin
36 in "cmsysb.red"$
37 end;
```

2.1 Simple one variable delay differential equation

Model a delayed ‘logistic’ system in one variable with

$$\frac{du}{dt} = -(1+a)[1+u(t)]u(t-\pi/2),$$

for small parameter a . We code the parameter a as ‘small’, and observe it is consequently considered as ‘small squared’ because all nonlinear terms and already ‘small’ terms are multiplied by `small`. The marginal modes are $e^{\pm it}$ so nominate the frequencies ± 1 . The eigenvectors are just $1 \cdot e^{\pm it}$. Because for delay differential equations the time dependence $e^{\pm i\omega t}$ is an integral part of the definition of the eigenvector; hence the coded eigenvectors can be the same, as here, because they are differentiated through the time dependence $e^{\pm i\omega t}$.

```

38 if thecase=onedde then begin
39 ff_:=tp mat((- (1+small*a)*(1+u1)*u1(pi/2)));
40 evalm_:=mat((i,-i));
41 ee_:=tp mat((1),(1));
42 zz_:=tp mat((1),(1));
43 toosmall:=3;
44 factor s,a,exp;
45 end;

```

The code works for orders higher than cubic, but is slow: takes about a minute per iteration.

The centre manifold

$$u_1 = e^{-2ti} s_2^2 \varepsilon (1/5i + 2/5) + e^{-ti} s_2 + e^{2ti} s_1^2 \varepsilon (-1/5i + 2/5) + e^{ti} s_1$$

Centre manifold ODEs

$$\dot{s}_1 = s_2 s_1^2 \varepsilon^2 (-2/5i\pi - 12/5i - 6/5\pi + 4/5) / (\pi^2 + 4) + s_1 a \varepsilon^2 (4i + 2\pi) / (\pi^2 + 4)$$

$$\dot{s}_2 = s_2^2 s_1 \varepsilon^2 (2/5i\pi + 12/5i - 6/5\pi + 4/5) / (\pi^2 + 4) + s_2 a \varepsilon^2 (-4i + 2\pi) / (\pi^2 + 4)$$

Observe that the real parts of these ODEs indicate linear growth for positive parameter a , limited by nonlinear saturation. A classic Hopf bifurcation (although I have not recorded here evidence for the attractiveness).

2.2 Another one variable delay differential equation

Model a delayed ‘logistic’ system in one variable with

$$\frac{du}{dt} = -u(t) - (\sqrt{2} + a)u(t - 3\pi/4) + \mu u(t - 3\pi/4)^2 + \nu u(t - 3\pi/4)^3,$$

for small parameter a and nonlinearity parameters μ and ν . Numerical computation of the spectrum indicates that the system has a Hopf bifurcation as parameter a crosses zero.¹

```
46 ac=-sqrt(2), tau=3*pi/4
47 ce=@(z) z+1-ac*exp(-tau*z)
48 lams=fsolve(ce,randn(100,2)*[2;2*i])
49 plot(real(lams),imag(lams),'o')
```

We code the parameter a as ‘small’, and observe it is consequently considered as ‘small squared’ because all nonlinear terms and already ‘small’ terms are multiplied by `small`. The marginal modes are $e^{\pm it}$ so nominate the frequencies ± 1 . The eigenvectors are just $1 \cdot e^{\pm it}$. Because for delay differential equations the time dependence $e^{\pm i\omega t}$ is an integral part of the definition of the eigenvector; hence the coded eigenvectors can be the same, as here, because they are differentiated through the time dependence $e^{\pm i\omega t}$.

```
50 if thecase=anotherdde then begin
51 ff_:=tp mat((-u1-(sqrt(2)+small*a)*u1(3*pi/4)
52      +mu*u1(3*pi/4)^2 +small*nu*u1(3*pi/4)^3));
53 evalm_:=mat((i,-i));
54 ee_:=tp mat((1),(1));
55 zz_:=tp mat((1),(1));
56 toosmall:=3;
57 factor s,a,mu,nu,cis;
58 end;
```

The modelling predicts a supercritical Hopf bifurcation as parameter a increases through zero, although if nonlinearity parameter ν is large enough

¹Replacing $-(\sqrt{2} + a)$ with $+(1 + a)$ leads to a pitchfork bifurcation with broken symmetry when $\mu \neq 0$.

negative, then the bifurcation will be subcritical.

2.3 Separated delay differential equations

Now consider the system

$$\dot{x} = -[1 + a - y(t)]x(t - \pi/2) \quad \text{and} \quad \dot{y} = -y + x^2.$$

Without the ‘fast’ variable y the x -ODE would be at marginal criticality when parameter $a = 0$. With the coupling, any oscillations in x should drive a positive y which then helps stabilise the oscillations. Let’s see this in analysis.

Code the system as follows with small parameter a . Because the system is linearly separated, the eigenvectors are simple: the eigenvectors of the marginal modes are $(1, 0)e^{\pm it}$, as are the adjoint’s eigenvectors.

```

59 if thecase=twodde then begin
60   ff_:=tp mat((
61     -(1+small*a-u2)*u1(pi/2),
62     -u2+u1^2
63   ));
64   evalm_:=mat((i,-i));
65   ee_:=tp mat((1,0),(1,0));
66   zz_:=tp mat((1,0),(1,0));
67   toosmall:=3;
68   factor s,a,exp;
69 end;
```

The centre manifold

$$u_1 = e^{-ti}s_2 + e^{ti}s_1$$

$$u_2 = e^{-2ti}s_2^2\varepsilon(2/5i + 1/5) + e^{2ti}s_1^2\varepsilon(-2/5i + 1/5) + 2s_2s_1\varepsilon$$

Centre manifold ODEs

$$\begin{aligned}\dot{s}_1 &= s_2 s_1^2 \varepsilon^2 (-4/5i\pi - 36/5i - 18/5\pi + 8/5) / (\pi^2 + 4) + s_1 a \varepsilon^2 (4i + 2\pi) / (\pi^2 + 4) \\ \dot{s}_2 &= s_2^2 s_1 \varepsilon^2 (4/5i\pi + 36/5i - 18/5\pi + 8/5) / (\pi^2 + 4) + s_2 a \varepsilon^2 (-4i + 2\pi) / (\pi^2 + 4)\end{aligned}$$

2.4 Linearly coupled 2D DDE

Here we explore a system where the centre modes involve both variables. Consider the system

$$\dot{u}_1 = u_2(t - \pi/2) - u_1^2 \quad \text{and} \quad \dot{u}_2 = u_1(t - \pi/2) + u_2^2.$$

We find the quadratic reaction does not stabilise oscillating growth.

Numerical solution of the characteristic equation indicate that there is one unstable mode, $\lambda = 0.4745$, two centre modes, $\lambda = \pm i$, and all the rest are stable modes with the gravest having eigenvalue $\lambda = -0.6846 \pm i2.8499$. The analysis gives the centre modes are nonlinearly unstable: $\dot{a} \approx (0.6758 \pm i1.8616)|a|^2 a$. The following Matlab/Octave code finds eigenvalues.

```
70 ce=@(z) z.^2-exp(-pi*z)
71 lams=fsolve(ce,randn(100,2)*[2;10*i])
72 plot(real(lams),imag(lams),'o')
```

Interestingly, the centre eigenvectors are $(1, -1)e^{\pm it}$ so that u_2 is in opposite phase to u_1 . The adjoint's eigenvectors are the same.

```
73 if thecase=dde2d then begin
74 ff_:=tp mat((+u2(pi/2)-u1^2,+u1(pi/2)+u2^2));
75 evalm_:=mat((i,-i));
76 ee_:=tp mat((1,-1),(1,-1));
77 zz_:=tp mat((1,-1),(1,-1));
78 toosmall:=3; factor s,small;
79 end;
```

The centre manifold

$$u_1 = s_2^2 \varepsilon (-2/5 e^{-2ti} i + 1/5 e^{-2ti}) - 2s_2 s_1 \varepsilon + s_2 e^{-ti} + s_1^2 \varepsilon (2/5 e^{2ti} i + 1/5 e^{2ti}) + s_1 e^{ti}$$

$$u_2 = s_2^2 \varepsilon (2/5 e^{-2ti} i - 1/5 e^{-2ti}) + 2s_2 s_1 \varepsilon - s_2 e^{-ti} + s_1^2 \varepsilon (-2/5 e^{2ti} i - 1/5 e^{2ti}) - s_1 e^{ti}$$

Centre manifold ODEs

$$\dot{s}_1 = s_2 s_1^2 \varepsilon^2 (-36/5 i \pi - 16/5 i - 8/5 \pi + 72/5) / (\pi^2 + 4)$$

$$\dot{s}_2 = s_2^2 s_1 \varepsilon^2 (36/5 i \pi + 16/5 i - 8/5 \pi + 72/5) / (\pi^2 + 4)$$

This model predicts nonlinear growth of the centre modes, in addition to the growth of the unstable mode.

2.5 Double Hopf 2D DDE

[Erneux \(2009\)](#) [§7.2] explored an example of a laser subject to optoelectronic feedback. For certain parameter values it has a two frequency Hopf bifurcation.

[Erneux \(2009\)](#) [eq. (7.42)] transformed the laser system to the non-dimensional

$$(1 + \eta) \frac{d^2 \log[1 + y]}{dt^2} = -\theta^2 [y(t) + \eta y(t - \pi)],$$

for parameters η and θ . [Erneux \(2009\)](#) identified double Hopf bifurcations from the origin at parameters (η, θ) of $(3/5, 2)$, $(7/25, 4)$, $(-5/13, 2)$ and $(-9/41, 4)$, among others. Here we work with a system of first order, DDEs, so transform the DDE to

$$\begin{aligned} \dot{x} &= -\theta^2 [y(t) + \eta y(t - \pi)] / (1 + \eta), \\ \dot{y} &= [1 + y(t)]x(t). \end{aligned}$$

The following Octave/Matlab code plots the spectrum for the equilibrium at the origin. The results indicate that in all four cases mentioned the centre manifold is attractive. The gravest eigenvalue being, respectively, $-0.69 \pm i3.87$, $-0.38 \pm i1.02$, -0.31 and $-0.41 \pm i2.03$.

```

80 eta=3/5, theta=2
81 ce=@(z) (1+eta)*z.^2+theta^2*(1+eta*exp(-pi*z))
82 lams=fsolve(ce,randn(100,2)*[2;10*i])
83 plot(real(lams),imag(lams),'o')

```

Ensure you interpret ‘left-eigenvectors’ as the eigenvectors of the adjoint operator (the complex conjugate transpose of the operator).

2.5.1 Parameters $(\eta, \theta) = (3/5, 2)$

I invoke a slightly different perturbation of the parameter η to that of [Erneux \(2009\)](#). The eigenvectors are $(1, \mp i/\omega)e^{\pm i\omega t}$ for frequencies $\omega = 1, 2$, while the eigenvectors of the adjoint are $(1, \mp i\omega)e^{\pm i\omega t}$.

```

84 if thecase=dde2d2ha then begin
85   eta:=3/5;
86   theta:=2*(1+small*delta);
87   ff_:=tp mat((
88     -theta^2*((1/(1+eta)-small*nu)*u2
89       +(eta/(1+eta)+small*nu)*u2(pi)),
90     +u1*(1+u2)
91   ));
92   evalm_:=mat((i,2*i,-i,-2*i));
93   ee_:=tp mat((1,-i),(1,-i/2),(1,+i),(1,+i/2));
94   zz_:=tp mat((1,-i),(1,-2*i),(1,+i),(1,+2*i));
95   toosmall:=3;
96   factor s,delta,nu,cis;
97 end;

```

The centre manifold is rather complicated.

$$\begin{aligned}
 u_1 = & 1/6 e^{-4ti} s_4^2 \varepsilon i + 3/16 e^{-3ti} s_4 s_2 \varepsilon i + e^{-2ti} s_4 + e^{-2ti} s_2^2 \varepsilon (-9/2i\pi^2 - 16i - 6\pi)/(9\pi^2 + 64) \\
 & + e^{-ti} s_4 s_1 \varepsilon (9/4i\pi^2 + 2i - 3/2\pi)/(9\pi^2 + 16) + e^{-ti} s_2 - 1/6 e^{4ti} s_3^2 \varepsilon i - 3/16 e^{3ti} s_3 s_1 \varepsilon i + e^{2ti} s_3 + e^{2ti} s_1^2 \varepsilon (9/2i\pi^2 + 16i - 6\pi)/(9\pi^2 + 64) \\
 & + e^{ti} s_3 s_2 \varepsilon (-9/4i\pi^2 - 2i - 3/2\pi)/(9\pi^2 + 16) + e^{ti} s_1
 \end{aligned}$$

$$u_2 = -1/6 e^{-4ti} s_4^2 \varepsilon - 9/16 e^{-3ti} s_4 s_2 \varepsilon + 1/2 e^{-2ti} s_4 i + e^{-2ti} s_2^2 \varepsilon (3i\pi - 9/4\pi^2 - 8)/(9\pi^2 + 64) + e^{-ti} s_4 s_1 \varepsilon (3/2i\pi + 9/4\pi^2 + 2)/(9\pi^2 + 16) + e^{-ti} s_2 i - 1/6 e^{4ti} s_3^2 \varepsilon - 9/16 e^{3ti} s_3 s_1 \varepsilon - 1/2 e^{2ti} s_3 i + e^{2ti} s_1^2 \varepsilon (-3i\pi - 9/4\pi^2 - 8)/(9\pi^2 + 64) + e^{ti} s_3 s_2 \varepsilon (-3/2i\pi + 9/4\pi^2 + 2)/(9\pi^2 + 16) - e^{ti} s_1 i$$

Centre manifold ODEs describe complicated interactions, but mainly it is the coefficients that are complicated functions of π .

$$\dot{s}_1 = s_4 s_3 s_1 \varepsilon^2 (-9963/4i\pi^6 - 38340i\pi^4 - 167424i\pi^2 - 147456i + 21141/16\pi^7 + 20007\pi^5 + 84096\pi^3 + 61440\pi)/(6561\pi^8 + 116640\pi^6 + 684288\pi^4 + 1474560\pi^2 + 1048576) + s_3 s_2 \varepsilon (-3i\pi - 4)/(9\pi^2 + 16) + s_2 s_1^2 \varepsilon^2 (-2916i\pi^6 - 17280i\pi^4 - 3072i\pi^2 - 196608i - 8019/2\pi^7 - 44064\pi^5 - 93312\pi^3 + 122880\pi)/(6561\pi^8 + 116640\pi^6 + 684288\pi^4 + 1474560\pi^2 + 1048576) + s_1 \delta \varepsilon^2 (16i - 12\pi)/(9\pi^2 + 16) + s_1 \nu \varepsilon^2 (-64i + 48\pi)/(9\pi^2 + 16)$$

$$\dot{s}_2 = s_4 s_3 s_2 \varepsilon^2 (9963/4i\pi^6 + 38340i\pi^4 + 167424i\pi^2 + 147456i + 21141/16\pi^7 + 20007\pi^5 + 84096\pi^3 + 61440\pi)/(6561\pi^8 + 116640\pi^6 + 684288\pi^4 + 1474560\pi^2 + 1048576) + s_4 s_1 \varepsilon (3i\pi - 4)/(9\pi^2 + 16) + s_2^2 s_1 \varepsilon^2 (2916i\pi^6 + 17280i\pi^4 + 3072i\pi^2 + 196608i - 8019/2\pi^7 - 44064\pi^5 - 93312\pi^3 + 122880\pi)/(6561\pi^8 + 116640\pi^6 + 684288\pi^4 + 1474560\pi^2 + 1048576) + s_2 \delta \varepsilon^2 (-16i - 12\pi)/(9\pi^2 + 16) + s_2 \nu \varepsilon^2 (64i + 48\pi)/(9\pi^2 + 16)$$

$$\dot{s}_3 = s_4 s_3^2 \varepsilon^2 (-16/3i - 2\pi)/(9\pi^2 + 64) + s_3 s_2 s_1 \varepsilon^2 (-34992i\pi^6 - 252288i\pi^4 - 559104i\pi^2 - 393216i - 10206\pi^7 - 64800\pi^5 - 138240\pi^3 - 98304\pi)/(6561\pi^8 + 116640\pi^6 + 684288\pi^4 + 1474560\pi^2 + 1048576) + s_3 \delta \varepsilon^2 (128i + 48\pi)/(9\pi^2 + 64) + s_1^2 \varepsilon (-24i\pi + 64)/(9\pi^2 + 64)$$

$$\dot{s}_4 = s_4^2 s_3 \varepsilon^2 (16/3i - 2\pi)/(9\pi^2 + 64) + s_4 s_2 s_1 \varepsilon^2 (34992i\pi^6 + 252288i\pi^4 + 559104i\pi^2 + 393216i - 10206\pi^7 - 64800\pi^5 - 138240\pi^3 - 98304\pi)/(6561\pi^8 + 116640\pi^6 + 684288\pi^4 + 1474560\pi^2 + 1048576) + s_4 \delta \varepsilon^2 (-128i + 48\pi)/(9\pi^2 + 64) + s_2^2 \varepsilon (24i\pi + 64)/(9\pi^2 + 64)$$

2.5.2 Parameters $(\eta, \theta) = (7/25, 4)$

The eigenvectors are $(1, \mp i/\omega)e^{\pm i\omega t}$ for frequencies $\omega = 3, 4$, while the eigenvectors of the adjoint are $(1, \mp i\omega)e^{\pm i\omega t}$.

```

98 if thecase=dde2d2hb then begin
99   eta:=7/25;
100  theta:=4*(1+small*delta);
101  ff_:=tp mat((
102    -theta^2*((1/(1+eta)-small*nu)*u2
103      +(eta/(1+eta)+small*nu)*u2(pi)),
104    +u1*(1+u2)
105  ));
106  evalm_:=mat((3*i,-3*i,4*i,-4*i));
107  ee_:=tp mat((1,-i/3),(1,+i/3),(1,-i/4),(1,+i/4));
108  zz_:=tp mat((1,-3*i),(1,+3*i),(1,-4*i),(1,+4*i));
109  toosmall:=3;
110  factor s,delta,nu,cis;
111 end;

```

The centre manifold

$$u_1 = 1/12 e^{-8ti} s_4^2 \varepsilon i + 21/160 e^{-7ti} s_4 s_2 \varepsilon i + 4/15 e^{-6ti} s_2^2 \varepsilon i + e^{-4ti} s_4 + e^{-3ti} s_2 + 3/32 e^{-ti} s_4 s_1 \varepsilon i - 1/12 e^{8ti} s_3^2 \varepsilon i - 21/160 e^{7ti} s_3 s_1 \varepsilon i - 4/15 e^{6ti} s_1^2 \varepsilon i + e^{4ti} s_3 + e^{3ti} s_1 - 3/32 e^{ti} s_3 s_2 \varepsilon i$$

$$u_2 = -1/24 e^{-8ti} s_4^2 \varepsilon - 49/480 e^{-7ti} s_4 s_2 \varepsilon - 1/10 e^{-6ti} s_2^2 \varepsilon + 1/4 e^{-4ti} s_4 i + 1/3 e^{-3ti} s_2 i - 1/96 e^{-ti} s_4 s_1 \varepsilon - 1/24 e^{8ti} s_3^2 \varepsilon - 49/480 e^{7ti} s_3 s_1 \varepsilon - 1/10 e^{6ti} s_1^2 \varepsilon - 1/4 e^{4ti} s_3 i - 1/3 e^{3ti} s_1 i - 1/96 e^{ti} s_3 s_2 \varepsilon$$

Centre manifold ODEs

$$\dot{s}_1 = s_4 s_3 s_1 \varepsilon^2 (-243/20i + 567/80\pi)/(49\pi^2 + 144) + s_2 s_1^2 \varepsilon^2 (-12/5i + 7/5\pi)/(49\pi^2 + 144) + s_1 \delta \varepsilon^2 (432i - 252\pi)/(49\pi^2 + 144) + s_1 \nu \varepsilon^2 (-768i + 448\pi)/(49\pi^2 + 144)$$

$$\dot{s}_2 = s_4 s_3 s_2 \varepsilon^2 (243/20i + 567/80\pi)/(49\pi^2 + 144) + s_2^2 s_1 \varepsilon^2 (12/5i + 7/5\pi)/(49\pi^2 + 144) + s_2 \delta \varepsilon^2 (-432i - 252\pi)/(49\pi^2 + 144) + s_2 \nu \varepsilon^2 (768i + 448\pi)/(49\pi^2 + 144)$$

$$\dot{s}_3 = s_4 s_3^2 \varepsilon^2 (-32/3i - 14/3\pi)/(49\pi^2 + 256) + s_3 s_2 s_1 \varepsilon^2 (-256/5i - 112/5\pi)/(49\pi^2 + 256) + s_3 \delta \varepsilon^2 (1024i + 448\pi)/(49\pi^2 + 256)$$

$$\dot{s}_4 = s_4^2 s_3 \varepsilon^2 (32/3i - 14/3\pi) / (49\pi^2 + 256) + s_4 s_2 s_1 \varepsilon^2 (256/5i - 112/5\pi) / (49\pi^2 + 256) + s_4 \delta \varepsilon^2 (-1024i + 448\pi) / (49\pi^2 + 256)$$

The interaction appears a lot simpler in this case. Presumably simpler because the frequencies are ‘more irrational’.

2.6 Simple 2D ODE

Consider the system $\dot{u}_1 = -\varepsilon u_1^2 + u_2 - u_1$ and $\dot{u}_2 = \varepsilon u_2^2 - u_2 + u_1$

```

112 if thecase=simple2d then begin
113 ff_:=tp mat((-u1+u2-u1^2,u1-u2+u2^2));
114 evalm_:=mat((0));
115 ee_:=tp mat((1,1));
116 zz_:=tp mat((1,1));
117 toosmall:=5;
118 end;
```

The centre manifold $u_1 = 3/8\varepsilon^3 s_1^4 - 1/2\varepsilon s_1^2 + s_1$

$$u_2 = -3/8\varepsilon^3 s_1^4 + 1/2\varepsilon s_1^2 + s_1$$

Centre manifold ODEs $\dot{s}_1 = -3/4\varepsilon^4 s_1^5 + \varepsilon^2 s_1^3$

Normals to isochrons at the slow manifold Use these vectors: to project initial conditions onto the slow manifold; to project non-autonomous forcing onto the slow evolution; to predict the consequences of modifying the original system; in uncertainty quantification to quantify effects on the model of uncertainties in the original system.

$$z_{11} = 3/2\varepsilon^4 s_1^4 + 3/4\varepsilon^3 s_1^3 - 1/2\varepsilon^2 s_1^2 - 1/2\varepsilon s_1 + 1/2$$

$$z_{12} = 3/2\varepsilon^4 s_1^4 - 3/4\varepsilon^3 s_1^3 - 1/2\varepsilon^2 s_1^2 + 1/2\varepsilon s_1 + 1/2$$

2.6.1 The stable manifold

Appears to get sensible answers even for the stable manifold! Just invoke this case to characterise the linear stable subspace.

```
119 if thecase=simple2ds then begin
120 ff_:=tp mat((-u1+u2-u1^2,u1-u2+u2^2));
121 evalm_:=mat((-2));
122 ee_:=tp mat((1,-1));
123 zz_:=tp mat((1,-1));
124 toosmall:=5;
125 end;
```

The stable manifold where the double factor of i in the exponentials give decaying modes of $e^{-2t}, e^{-6t}, e^{-8t}$.

$$u_1 = 1/8\epsilon^3 e^{8iti} s_1^4 + 1/4\epsilon^2 e^{6iti} s_1^3 + 1/2\epsilon e^{4iti} s_1^2 + e^{2iti} s_1$$

$$u_2 = -1/8\epsilon^3 e^{8iti} s_1^4 - 1/4\epsilon^2 e^{6iti} s_1^3 - 1/2\epsilon e^{4iti} s_1^2 - e^{2iti} s_1$$

Stable manifold ODEs is the trivial $\dot{s}_1 = 0$

2.6.2 The slow-stable manifold

Appears to get sensible answers even for the slow-stable manifold!! Which in this system is a coordinate transform that nonlinearly separates the dynamics. Amazing.

```
126 if thecase=simple2dss then begin
127 ff_:=tp mat((-u1+u2-u1^2,u1-u2+u2^2));
128 evalm_:=mat((0,-2));
129 ee_:=tp mat((1,1),(1,-1));
130 zz_:=tp mat((1,1),(1,-1));
131 toosmall:=3;
132 end;
```

The invariant manifold These give the location of the invariant manifold in terms of parameters s_j .

$$\begin{aligned} u_1 &= \varepsilon^3 (1/8 e^{-8t} s_2^4 + 1/2 e^{-6t} s_2^3 s_1 + 1/2 e^{-4t} s_2^2 s_1^2 - 1/2 e^{-2t} s_2 s_1^3 + 3/8 s_1^4) + \\ &\varepsilon^2 (1/4 e^{-6t} s_2^3 + 3/4 e^{-4t} s_2^2 s_1) + \varepsilon (1/2 e^{-4t} s_2^2 + e^{-2t} s_2 s_1 - 1/2 s_1^2) + e^{-2t} s_2 + s_1 \\ u_2 &= \varepsilon^3 (-1/8 e^{-8t} s_2^4 + 1/2 e^{-6t} s_2^3 s_1 - 1/2 e^{-4t} s_2^2 s_1^2 - 1/2 e^{-2t} s_2 s_1^3 - 3/8 s_1^4) + \\ &\varepsilon^2 (-1/4 e^{-6t} s_2^3 + 3/4 e^{-4t} s_2^2 s_1) + \varepsilon (-1/2 e^{-4t} s_2^2 + e^{-2t} s_2 s_1 + 1/2 s_1^2) - e^{-2t} s_2 + \\ &s_1 \end{aligned}$$

invariant manifold ODEs The system evolves on the invariant manifold such that the parameters evolve according to these ODEs.

$$\begin{aligned} \dot{s}_1 &= -3/4 \varepsilon^4 s_1^5 + \varepsilon^2 s_1^3 \\ \dot{s}_2 &= 1/4 \varepsilon^4 s_2 s_1^4 - \varepsilon^2 s_2 s_1^2 \end{aligned}$$

2.7 Four state Markov chain

Variable ε characterise the perturbation.

$$\begin{aligned} \dot{u}_1 &= -\varepsilon u_1 + u_2 \\ \dot{u}_2 &= \varepsilon (u_3 - u_2 + u_1) - u_2 \\ \dot{u}_3 &= \varepsilon (u_4 - u_3 + u_2) - u_3 \\ \dot{u}_4 &= -\varepsilon u_4 + u_3 \end{aligned}$$

The linear perturbation terms gets multiplied by **small** again, but I do not see how to avoid that without wrecking other desirable things: such as, it is useful to multiply some nonlinear terms by small to show they are of higher order than other nonlinear terms.

```
133 if thecase=fourstatemarkov then begin
134   factor epsilon;
135   ff_:=tp mat((u2,-u2,-u3,u3))
136   +small*tp mat((-u1,+u1-u2+u3,+u2-u3+u4,-u4));
137   evalm_:=mat((0,0));
```



```

138 ee_:=tp mat((0,0,0,1),(1,0,0,0));
139 zz_:=tp mat((0,0,1,1),(1,1,0,0));
140 toosmall:=7;
141 end;

```

The centre manifold $u_1 = \varepsilon^2(2s_2 - s_1) - \varepsilon s_2 + s_2$

$$u_2 = \varepsilon^2(-2s_2 + s_1) + \varepsilon s_2$$

$$u_3 = \varepsilon^2(s_2 - 2s_1) + \varepsilon s_1$$

$$u_4 = \varepsilon^2(-s_2 + 2s_1) - \varepsilon s_1 + s_1$$

Centre manifold ODEs $\dot{s}_1 = \varepsilon^3(-3s_2 + 3s_1) + \varepsilon^2(s_2 - s_1)$

$$\dot{s}_2 = \varepsilon^3(3s_2 - 3s_1) + \varepsilon^2(-s_2 + s_1)$$

Normals to isochrons at the slow manifold

$$z_{11} = 6\varepsilon^6 - \varepsilon^4$$

$$z_{12} = 19\varepsilon^6 - 4\varepsilon^4 + \varepsilon^2$$

$$z_{13} = -19\varepsilon^6 + 4\varepsilon^4 - \varepsilon^2 + 1$$

$$z_{14} = -6\varepsilon^6 + \varepsilon^4 + 1$$

$$z_{21} = -6\varepsilon^6 + \varepsilon^4 + 1$$

$$z_{22} = -19\varepsilon^6 + 4\varepsilon^4 - \varepsilon^2 + 1$$

$$z_{23} = 19\varepsilon^6 - 4\varepsilon^4 + \varepsilon^2$$

$$z_{24} = 6\varepsilon^6 - \varepsilon^4$$

2.8 Bifurcating 2D system

This example tests labelling a small parameter and having a cubic term labelled as smaller than a quadratic term.

$$\dot{u}_1 = -\varepsilon^2 u_2 u_1^2 - u_2 - 1/2 u_1$$

$$\dot{u}_2 = \varepsilon(-u_2^2 + u_2 \epsilon) - 2u_2 - u_1$$

```

142 if thecase=another2d then begin
143   ff_:=tp mat((
144     -u1/2-u2-small*u1^2*u2,
145     -u1-2*u2+small*epsilon*u2-u2^2
146   ));
147   evalm_:=mat((0));
148   ee_:=tp mat((1,-1/2));
149   zz_:=tp mat((1,-1/2));
150 end;
```

The centre manifold $u_1 = \varepsilon(-1/25s_1^2 - 2/25s_1\epsilon) + s_1$

$$u_2 = \varepsilon(-2/25s_1^2 - 4/25s_1\epsilon) - 1/2s_1$$

Centre manifold ODEs $\dot{s}_1 = \varepsilon^2(54/125s_1^3 + 12/125s_1^2\epsilon + 8/125s_1\epsilon^2) + \varepsilon(1/10s_1^2 + 1/5s_1\epsilon)$

Normals to isochrons at the slow manifold

$$z_{11} = \varepsilon^2(-352/3125s_1^2 - 8/125\epsilon) - 8/125\varepsilon s_1 + 4/5$$

$$z_{12} = \varepsilon^2(-544/3125s_1^2 - 16/125\epsilon) - 16/125\varepsilon s_1 - 2/5$$

2.8.1 The stable manifold

Appears to also get the stable manifold.

```

151 if thecase=another2ds then begin
152   ff_:=tp mat((
153     -u1/2-u2-small*u1^2*u2,
154     -u1-2*u2+small*epsilon*u2-u2^2
155   ));
```

```

156 evalm_:=mat((-5/2));
157 ee_:=tp mat((1,2));
158 zz_:=tp mat((1,2));
159 toosmall:=7;
160 end;

```

The stable manifold ignoring the as yet awful formatting of the exponential,

$$u_1 = \varepsilon^2 \left(838/1875 e^{(15iti/2)} s_1^3 + 8/25 e^{(5iti/2)} s_1 \epsilon \right) + 8/25 \varepsilon e^{5iti} s_1^2 + e^{(5iti/2)} s_1$$

$$u_2 = \varepsilon^2 \left(2116/1875 e^{(15iti/2)} s_1^3 - 4/25 e^{(5iti/2)} s_1 \epsilon \right) + 36/25 \varepsilon e^{5iti} s_1^2 + 2 e^{(5iti/2)} s_1$$

Stable manifold ODEs shows the change in rate due to parameter variation: $\dot{s}_1 = 4/5 \varepsilon^2 s_1 \epsilon$

2.8.2 The slow-stable manifold

Appears to also get the slow-stable manifold, namely a normal form coordinate transform of the 2D state space.

```

161 if thecase=another2dss then begin
162 ff_:=tp mat((
163     -u1/2-u2-small*u1^2*u2,
164     -u1-2*u2+small*epsilon*u2-u2^2
165 ));
166 evalm_:=mat((0,-5/2));
167 ee_:=tp mat((1,-1/2),(1,2));
168 zz_:=tp mat((1,-1/2),(1,2));
169 toosmall:=5;
170 end;

```

2.9 A sub-stable example shows zero divisors

Order three is OK, but order four gives zero divisor error.

```

171 if thecase=substblem then begin
172   ff_:=tp mat((
173     -u1/4-u1*u2,
174     -u2+u1^2 ));
175   evalm_:=mat((-1/4,-1));
176   ee_:=tp mat((1,0),(0,1));
177   zz_:=tp mat((1,0),(0,1));
178   toosmall:=6;
179 end;
```

2.10 Simple 3D system

This example is straightforward.

$$\dot{u}_1 = \varepsilon u_3 u_2 + 2u_3 + u_2 + 2u_1$$

$$\dot{u}_2 = -\varepsilon u_3 u_1 + u_3 - u_2 + u_1$$

$$\dot{u}_3 = -\varepsilon u_2 u_1 - 3u_3 - u_2 - 3u_1$$

```

180 if thecase=simple3d then begin
181   ff_:=tp mat((2*u1+u2+2*u3+u2*u3
182     ,u1-u2+u3-u1*u3
183     ,-3*u1-u2-3*u3-u1*u2));
184   evalm_:=mat((0));
185   ee_:=tp mat((1,0,-1));
186   zz_:=tp mat((4,1,3));
187 end;
```

The centre manifold $u_1 = -\varepsilon s_1^2 + s_1$

$$u_2 = \varepsilon s_1^2$$

$$u_3 = \varepsilon s_1^2 - s_1$$

Centre manifold ODEs $\dot{s}_1 = -9\varepsilon^2 s_1^3 + \varepsilon s_1^2$

Normals to isochrons at the slow manifold

$$z_{11} = 258\varepsilon^2 s_1^2 - 16\varepsilon s_1 + 4$$

$$z_{12} = 93\varepsilon^2 s_1^2 - 9\varepsilon s_1 + 1$$

$$z_{13} = 240\varepsilon^2 s_1^2 - 16\varepsilon s_1 + 3$$

2.10.1 Its 2D stable manifold with generalised eigenvectors

Despite the generalised eigenvectors, the following alternative appears to generate the stable manifold if you wish:

```

188 if thecase=simple3ds then begin
189   ff_:=tp mat((2*u1+u2+2*u3+u2*u3
190     ,u1-u2+u3-u1*u3
191     ,-3*u1-u2-3*u3-u1*u2));
192   evalm_:=mat((-1,-1));
193   ee_:=tp mat((1,-1,-1),(1,7/2,-5/2));
194   zz_:=tp mat((0,1,0),(1,0,1));
195 end;
```

The adjusted dynamical system Modified in order cater for the generalised eigenvector.

$$\dot{u}_1 = \varepsilon(u_3 u_2 - u_3 - u_1) + 3u_3 + u_2 + 3u_1$$

$$\dot{u}_2 = \varepsilon(-u_3 u_1 + u_3 + u_1) - u_2$$

$$\dot{u}_3 = \varepsilon(u_3 - u_2 u_1 + u_1) - 4u_3 - u_2 - 4u_1$$

The stable manifold noting the double i factors give decaying modes.

$$\begin{aligned} u_1 &= \varepsilon \left(-51/4 e^{2iti} s_2^2 - 3 e^{2iti} s_2 s_1 + 3 e^{2iti} s_1^2 \right) + e^{iti} s_2 + e^{iti} s_1 \\ u_2 &= \varepsilon \left(-5/2 e^{2iti} s_2^2 - 7/2 e^{2iti} s_2 s_1 - e^{2iti} s_1^2 \right) + 7/2 e^{iti} s_2 - e^{iti} s_1 \\ u_3 &= \varepsilon \left(25 e^{2iti} s_2^2 + 13/2 e^{2iti} s_2 s_1 - 5 e^{2iti} s_1^2 \right) - 5/2 e^{iti} s_2 - e^{iti} s_1 \end{aligned}$$

Stable manifold ODEs $\dot{s}_1 = 3/2\varepsilon s_2$ and $\dot{s}_2 = 0$

2.11 3D system with a generalised eigenvector

Took longer to converge, but converge it does. However, now I force the off-diagonal term to be small.

$$\begin{aligned} \dot{u}_1 &= \varepsilon (u_3 u_2 + u_3 + u_2 + u_1) + u_3 + u_1 \\ \dot{u}_2 &= -\varepsilon u_3 u_1 + u_3 + u_1 \\ \dot{u}_3 &= \varepsilon (-u_3 - u_2 u_1 - u_2 - u_1) - 2u_3 - 2u_1 \end{aligned}$$

```

196 if thecase=geneigenvec then begin
197   ff_:=tp mat((
198     2*u1+u2+2*u3+u2*u3,
199     u1+u3-u1*u3,
200     -3*u1-u2-3*u3-u1*u2
201   ));
202   evalm_:=mat((0,0));
203   ee_:=tp mat((1,0,-1),(0,1,0));
204   zz_:=tp mat((1,-1,0),(1,1,1));
205   toosmall:=3;
206 end;
```

The centre manifold $u_1 = 2\varepsilon s_2 s_1 + s_1$

$$u_2 = 2\varepsilon s_2 s_1 + s_2$$

$$u_3 = -4\varepsilon s_2 s_1 - s_1$$

Centre manifold ODEs $\dot{s}_1 = \varepsilon^2(-10s_2^2s_1 - 6s_2s_1^2) + \varepsilon(-3s_2s_1 + s_2)$
 $\dot{s}_2 = \varepsilon^2(-6s_2^2s_1 + 2s_2s_1^2) + \varepsilon(-2s_2s_1 + s_1^2)$

Normals to isochrons at the slow manifold

$$\begin{aligned} z_{11} &= \varepsilon^2(50s_2^2 + 60s_2s_1 + 14s_1^2 + s_1) + \varepsilon(5s_2 + 3s_1) + 2 \\ z_{12} &= \varepsilon^2(10s_2s_1 + 6s_1^2) \\ z_{13} &= \varepsilon^2(40s_2^2 + 54s_2s_1 + 14s_1^2 + s_1) + \varepsilon(5s_2 + 3s_1) + 1 \\ z_{21} &= \varepsilon^2(31s_2^2 + 8s_2s_1 - s_2 - 9s_1^2) + \varepsilon(3s_2 - s_1) + 1 \\ z_{22} &= \varepsilon^2(6s_2s_1 - 2s_1^2) + 1 \\ z_{23} &= \varepsilon^2(25s_2^2 + 10s_2s_1 - s_2 - 9s_1^2) + \varepsilon(3s_2 - s_1) + 1 \end{aligned}$$

2.12 Separated system

To see if small part in the slow variable ruins convergence. The answer is that it did—hence we include code to make anything non-oscillatory in the slow variables to be small. Also test a non-zero constant forcing.

```

i1 = ε( - u2u1 + u1α)
i2 = ε(β - 2u2^2 + u1^2) - u2

207 if thecase=bifurcate2d then begin
208   ff_:=tp mat((
209     alpha*u1-u1*u2,
210     -u2+u1^2-2*u2^2+beta
211   ));
212   evalm_:=mat((0));
213   ee_:=tp mat((1,0));
214   zz_:=tp mat((1,0));
215   toosmall:=4;
216 end;
```

The centre manifold $u_1 = s_1$

$$u_2 = \varepsilon(s_1^2 + \beta)$$

Centre manifold ODEs $\dot{s}_1 = -\varepsilon^2(s_1^3 - \beta s_1) + \varepsilon s_1 \alpha$

Normals to isochrons at the slow manifold

$$z_{11} = 2\varepsilon^2 s_1^2 + 1$$

$$z_{12} = -\varepsilon s_1$$

2.13 Oscillatory centre manifold—separated form

Let's try complex eigenvectors. Adjoint eigenvectors **zz_** must be the eigenvectors of the complex conjugate transpose matrix.

$$\dot{u}_1 = u_2$$

$$\dot{u}_2 = -\varepsilon u_3 u_1 - u_1$$

$$\dot{u}_3 = 5\varepsilon u_1^2 - u_3$$

```

217 if thecase=simpleosc then begin
218 ff_:=tp mat((u2,-u1-u1*u3,-u3+5*u1^2));
219 evalm_:=mat((i,-i));
220 ee_:=tp mat((1,+i,0),(1,-i,0));
221 %ee_:=tp mat((1+1/10,+i,0),(1+1/10,-i,0)); % causes fail, Jan 20
222 zz_:=tp mat((1,+i,0),(1,-i,0));
223 end;
```

The centre manifold $u_1 = e^{-ti} s_2 + e^{ti} s_1$

$$u_2 = -e^{-ti} s_2 i + e^{ti} s_1 i$$

$$u_3 = \varepsilon(2e^{-2ti} s_2^2 i + e^{-2ti} s_2^2 - 2e^{2ti} s_1^2 i + e^{2ti} s_1^2 + 10s_2 s_1)$$

Centre manifold ODEs $\dot{s}_1 = \varepsilon^2(11/2s_2s_1^2i + s_2s_1^2)$

$$\dot{s}_2 = \varepsilon^2(-11/2s_2^2s_1i + s_2^2s_1)$$

2.14 Perturbed frequency oscillatory centre manifold—separated form

Putting real parameters into the linear operator works here also.

$$\dot{u}_1 = \varepsilon(u_2b + u_1a) + u_2$$

$$\dot{u}_2 = \varepsilon(u_2d - u_1c) - u_1$$

$$\dot{u}_3 = -u_3$$

```

224 if thecase=perturbfreq then begin
225 ff_:=tp mat((a*u1+(1+b)*u2,d*u2-(1+c)*u1,-u3));
226 evalm_:=mat((i,-i));
227 ee_:=tp mat((1,+i,0),(1,-i,0));
228 zz_:=tp mat((1,+i,0),(1,-i,0));
229 b:=c:=0; d:=a;
230 toosmall:=2;
231 end;
```

The centre manifold $u_1 = \varepsilon(1/4e^{-ti}s_2ai + 1/4e^{-ti}s_2b - 1/4e^{-ti}s_2c - 1/4e^{-ti}s_2di - 1/4e^{ti}s_1ai + 1/4e^{ti}s_1b - 1/4e^{ti}s_1c + 1/4e^{ti}s_1di) + e^{-ti}s_2 + e^{ti}s_1$

$$u_2 = \varepsilon(-1/4e^{-ti}s_2a + 1/4e^{-ti}s_2bi - 1/4e^{-ti}s_2ci + 1/4e^{-ti}s_2d - 1/4e^{ti}s_1a - 1/4e^{ti}s_1bi + 1/4e^{ti}s_1ci + 1/4e^{ti}s_1d) - e^{-ti}s_2i + e^{ti}s_1i$$

$$u_3 = 0$$

Centre manifold ODEs

$$\dot{s}_1 = \varepsilon^2(-1/8s_1a^2i + 1/4s_1adi - 1/8s_1b^2i + 1/4s_1bci - 1/8s_1c^2i - 1/8s_1d^2i) + \varepsilon(1/2s_1a + 1/2s_1bi + 1/2s_1ci + 1/2s_1d)$$

$$\dot{s}_2 = \varepsilon^2(1/8s_2a^2i - 1/4s_2adi + 1/8s_2b^2i - 1/4s_2bci + 1/8s_2c^2i + 1/8s_2d^2i) + \varepsilon(1/2s_2a - 1/2s_2bi - 1/2s_2ci + 1/2s_2d)$$

2.15 More general oscillatory centre manifold

Consider the frequency two dynamics of the following system in non-separated form.

$$\dot{u}_1 = \varepsilon(u_2u_1 + u_1\epsilon) - 2u_3 - 2u_2$$

$$\dot{u}_2 = -2u_3 - 3u_2 + u_1$$

$$\dot{u}_3 = 2u_3 + 3u_2 + u_1$$

```

232 if thecase=nonseparatedosc then begin
233   ff_:=tp mat((
234     -2*u2-2*u3+epsilon*u1+u1*u2,
235     u1-3*u2-2*u3,
236     u1+3*u2+2*u3
237   ));
238   evalm_:=mat((+2*i,-2*i));
239   ee_:=tp mat((1,1,-1-i),(1,1,-1+i));
240   zz_:=tp mat((1,-i,-i),(1,+i,+i));
241 end;
```

The centre manifold $u_1 = \varepsilon(1/3 e^{-4ti} s_2^2 i + 1/8 e^{-2ti} s_2 \epsilon i - 1/3 e^{4ti} s_1^2 i - 1/8 e^{2ti} s_1 \epsilon i) + e^{-2ti} s_2 + e^{2ti} s_1$

$$u_2 = \varepsilon(5/51 e^{-4ti} s_2^2 i - 1/17 e^{-4ti} s_2^2 - 11/40 e^{-2ti} s_2 \epsilon i - 1/5 e^{-2ti} s_2 \epsilon - 5/51 e^{4ti} s_1^2 i - 1/17 e^{4ti} s_1^2 + 11/40 e^{2ti} s_1 \epsilon i - 1/5 e^{2ti} s_1 \epsilon - 2s_2 s_1) + e^{-2ti} s_2 + e^{2ti} s_1$$

$$u_3 = \varepsilon(-5/51 e^{-4ti} s_2^2 i - 11/102 e^{-4ti} s_2^2 + 11/40 e^{-2ti} s_2 \epsilon i + 13/40 e^{-2ti} s_2 \epsilon + 5/51 e^{4ti} s_1^2 i - 11/102 e^{4ti} s_1^2 - 11/40 e^{2ti} s_1 \epsilon i + 13/40 e^{2ti} s_1 \epsilon + 3s_2 s_1) + e^{-2ti} s_2 i - e^{-2ti} s_2 - e^{2ti} s_1 i - e^{2ti} s_1$$

Centre manifold ODEs $\dot{s}_1 = \varepsilon^2(-11/51 s_2 s_1^2 i - 35/34 s_2 s_1^2 - 1/16 s_1 \epsilon^2 i) + 1/2 \varepsilon s_1 \epsilon$

$$\dot{s}_2 = \varepsilon^2(11/51 s_2^2 s_1 i - 35/34 s_2^2 s_1 + 1/16 s_2 \epsilon^2 i) + 1/2 \varepsilon s_2 \epsilon$$

2.16 Quasi-delay differential equation

Shows Hopf bifurcation as parameter a crosses -4 to oscillations with base frequency two.

$$\dot{u}_1 = \varepsilon^2(-u_3 \alpha - u_1^3) - 2\varepsilon u_1^2 - 4u_3$$

$$\dot{u}_2 = -2u_2 + 2u_1$$

$$\dot{u}_3 = -2u_3 + 2u_2$$

```

242 if thecase=quasidelaysc then begin
243 ff_:=tp mat((
244     -4*u3-small*alpha*u3-2*u1^2-small*u1^3,
245     2*u1-2*u2,
246     2*u2-2*u3
247     ));
248 evalm_:=mat((2*i,-2*i));
249 ee_:=tp mat((1,1/2-i/2,-i/2),(1,1/2+i/2,+i/2));
250 zz_:=tp mat((1,-i,-1-i),(1,+i,-1+i));
251 end;
```

The centre manifold $u_1 = \varepsilon(-7/12 e^{-4ti} s_2^2 i + 1/12 e^{-4ti} s_2^2 + 7/12 e^{4ti} s_1^2 i + 1/12 e^{4ti} s_1^2 - s_2 s_1) + e^{-2ti} s_2 + e^{2ti} s_1$

$$u_2 = \varepsilon(-1/12 e^{-4ti} s_2^2 i + 1/4 e^{-4ti} s_2^2 + 1/12 e^{4ti} s_1^2 i + 1/4 e^{4ti} s_1^2 - s_2 s_1) + 1/2 e^{-2ti} s_2 i + 1/2 e^{-2ti} s_2 - 1/2 e^{2ti} s_1 i + 1/2 e^{2ti} s_1$$

$$u_3 = \varepsilon(1/12 e^{-4ti} s_2^2 i + 1/12 e^{-4ti} s_2^2 - 1/12 e^{4ti} s_1^2 i + 1/12 e^{4ti} s_1^2 - s_2 s_1) + 1/2 e^{-2ti} s_2 i - 1/2 e^{2ti} s_1 i$$

Centre manifold ODEs $\dot{s}_1 = \varepsilon^2(-16/15s_2s_1^2i - 1/5s_2s_1^2 + 1/5s_1\alpha i + 1/10s_1\alpha)$

$$\dot{s}_2 = \varepsilon^2(16/15s_2^2s_1i - 1/5s_2^2s_1 - 1/5s_2\alpha i + 1/10s_2\alpha)$$

2.17 Detuned version of quasi-delayed

The following modified version of the previous shows that we can ‘detune’ the linear operator and my ‘adjustment’ of the linear operator seems to work. Here the $1/2$ in $\mathcal{L}_{1,1}$ should be zero for these eigenvectors: my adjustment seems to fix it OK. But now, knowing the frequencies, my adjustment is different (and probably better).

$$\dot{u}_1 = \varepsilon^2(-u_3\alpha - u_1^3) + \varepsilon(-1/5u_3 + 1/5u_2 - 2u_1^2 + 2/5u_1) - 19/5u_3 - 1/5u_2 + 1/10u_1$$

$$\dot{u}_2 = -2u_2 + 2u_1$$

$$\dot{u}_3 = -2u_3 + 2u_2$$

```

252 if thecase=quasidelayscmod then begin
253   ff_:=tp mat((
254     u1/2-4*u3-small*alpha*u3-2*u1^2-small*u1^3,
255     2*u1-2*u2,
256     2*u2-2*u3
257   ));
258   evalm_:=mat((2*i,-2*i));
259   ee_:=tp mat((1,1/2-i/2,-i/2),(1,1/2+i/2,+i/2));
260   zz_:=tp mat((1,-i,-1-i),(1,+i,-1+i));
261   toosmall:=3;
262 end;
```

The centre manifold

$$u_1 = \varepsilon(-1840/3121 e^{-4ti} s_2^2 i + 860/9363 e^{-4ti} s_2^2 + 237/3842 e^{-2ti} s_2 i + 87/1921 e^{-2ti} s_2 + 1840/3121 e^{4ti} s_1^2 i + 860/9363 e^{4ti} s_1^2 - 237/3842 e^{2ti} s_1 i + 87/1921 e^{2ti} s_1 - 40/39 s_2 s_1) + e^{-2ti} s_2 + e^{2ti} s_1$$

$$u_2 = \varepsilon \left(-760/9363 e^{-4ti} s_2^2 i + 2380/9363 e^{-4ti} s_2^2 + 21/7684 e^{-2ti} s_2 i + 137/7684 e^{-2ti} s_2 + 760/9363 e^{4ti} s_1^2 i + 2380/9363 e^{4ti} s_1^2 - 21/7684 e^{2ti} s_1 i + 137/7684 e^{2ti} s_1 - 40/39 s_2 s_1 \right) + 1/2 e^{-2ti} s_2 i + 1/2 e^{-2ti} s_2 - 1/2 e^{2ti} s_1 i + 1/2 e^{2ti} s_1$$

$$u_3 = \varepsilon \left(800/9363 e^{-4ti} s_2^2 i + 260/3121 e^{-4ti} s_2^2 - 4/1921 e^{-2ti} s_2 i + 353/7684 e^{-2ti} s_2 - 800/9363 e^{4ti} s_1^2 i + 260/3121 e^{4ti} s_1^2 + 4/1921 e^{2ti} s_1 i + 353/7684 e^{2ti} s_1 - 40/39 s_2 s_1 \right) + 1/2 e^{-2ti} s_2 i - 1/2 e^{2ti} s_1 i$$

Centre manifold ODEs

$$\dot{s}_1 = \varepsilon^2 \left(-259684400/233822199 s_2 s_1^2 i - 1154340/5995441 s_2 s_1^2 + 390/1921 s_1 \alpha i + 200/1921 s_1 \alpha - 90446425/7088952961 s_1 i - 1300360/7088952961 s_1 \right) + \varepsilon \left(-200/1921 s_1 i + 390/1921 s_1 \right)$$

$$\dot{s}_2 = \varepsilon^2 \left(259684400/233822199 s_2^2 s_1 i - 1154340/5995441 s_2^2 s_1 - 390/1921 s_2 \alpha i + 200/1921 s_2 \alpha + 90446425/7088952961 s_2 i - 1300360/7088952961 s_2 \right) + \varepsilon \left(200/1921 s_2 i + 390/1921 s_2 \right)$$

Observe the terms linear in ε due to my fudging of the linear dynamics.

2.18 Rossler-like system

Has Hopf bifurcation as parameter crosses zero to oscillations of base frequency one.

$$\dot{u}_1 = -u_3 - u_2$$

$$\dot{u}_2 = \varepsilon u_2 a + u_1$$

$$\dot{u}_3 = \varepsilon (u_3 u_1 - 1/5 u_2 u_1) - 5 u_3$$

```

263 if thecase=rosslerlike then begin
264   ff_:=tp mat((
265     -u2-u3,
266     u1+small*a*u2,
267     -5*u3-u1*u2/5+u1*u3
268   ));
269   evalm_:=mat((i,-i));

```

```

270 ee_:=tp mat((1,-i,0),(1,i,0));
271 zz_:=tp mat((-5+i,1+5*i,1),(-5-i,1-5*i,1));
272 end;

```

The centre manifold $u_1 = \varepsilon(-4/435 e^{-2ti} s_2^2 i - 2/87 e^{-2ti} s_2^2 - 1/4 e^{-ti} s_2 a i + 4/435 e^{2ti} s_1^2 i - 2/87 e^{2ti} s_1^2 + 1/4 e^{ti} s_1 a i) + e^{-ti} s_2 + e^{ti} s_1$

$u_2 = \varepsilon(-1/87 e^{-2ti} s_2^2 i + 2/435 e^{-2ti} s_2^2 - 1/4 e^{-ti} s_2 a + 1/87 e^{2ti} s_1^2 i + 2/435 e^{2ti} s_1^2 - 1/4 e^{ti} s_1 a) + e^{-ti} s_2 i - e^{ti} s_1 i$

$u_3 = \varepsilon(-1/29 e^{-2ti} s_2^2 i + 2/145 e^{-2ti} s_2^2 + 1/29 e^{2ti} s_1^2 i + 2/145 e^{2ti} s_1^2)$

Centre manifold ODEs $\dot{s}_1 = \varepsilon^2(-92/28275 s_2 s_1^2 i - 4/1885 s_2 s_1^2 - 1/8 s_1 a^2 i) + 1/2 \varepsilon s_1 a$

$\dot{s}_2 = \varepsilon^2(92/28275 s_2^2 s_1 i - 4/1885 s_2^2 s_1 + 1/8 s_2 a^2 i) + 1/2 \varepsilon s_2 a$

2.19 Fudge a couple of these oscillations together

Use say different base frequencies of one and two. Put in a couple of coupling terms. It seems to work fine, although the computation time zooms up even for the basic third order errors.

$$\dot{u}_1 = \varepsilon u_4^2 - u_3 - u_2$$

$$\dot{u}_2 = \varepsilon u_2 a + u_1$$

$$\dot{u}_3 = \varepsilon(u_3 u_1 - 1/5 u_2 u_1) - 5 u_3$$

$$\dot{u}_4 = \varepsilon(u_6 u_5 + u_4 \varepsilon) - 2 u_6 - 2 u_5$$

$$\dot{u}_5 = \varepsilon u_1^2 - 2 u_6 - 3 u_5 + u_4$$

$$\dot{u}_6 = 2 u_6 + 3 u_5 + u_4$$

```

273 if thecase=doubleosc then begin

```

```

274 ff_:=tp mat((

```

```

275     -u2-u3+u4^2,

```

```

276    u1+a*u2,
277    -5*u3-u1*u2/5+u1*u3,
278    -2*u5-2*u6+small*epsilon*u4+u5*u6,
279    u4-3*u5-2*u6+u1^2,
280    u4+3*u5+2*u6
281    ));
282 evalm_:=mat((i,-i,2*i,-2*i));
283 ee_:=tp mat((1,-i,0,0,0,0),(1,i,0,0,0,0)
284    ,(0,0,0,1,1,-1-i),(0,0,0,1,1,-1+i));
285 zz_:=tp mat((-5+i,1+5*i,1,0,0,0),(-5-i,1-5*i,1,0,0,0)
286    ,(0,0,0,1,-i,-i),(0,0,0,1,+i,+i));
287 end;
```

The centre manifold $u_1 = \varepsilon(4/15 e^{-4ti} s_4^2 i - 4/435 e^{-2ti} s_2^2 i - 2/87 e^{-2ti} s_2^2 - 1/4 e^{-ti} s_2 a i - 4/15 e^{4ti} s_3^2 i + 4/435 e^{2ti} s_1^2 i - 2/87 e^{2ti} s_1^2 + 1/4 e^{ti} s_1 a i) + e^{-ti} s_2 + e^{ti} s_1$

$u_2 = \varepsilon(-1/15 e^{-4ti} s_4^2 i - 1/87 e^{-2ti} s_2^2 i + 2/435 e^{-2ti} s_2^2 - 1/4 e^{-ti} s_2 a - 1/15 e^{4ti} s_3^2 + 1/87 e^{2ti} s_1^2 i + 2/435 e^{2ti} s_1^2 - 1/4 e^{ti} s_1 a + 2s_4 s_3) + e^{-ti} s_2 i - e^{ti} s_1 i$

$u_3 = \varepsilon(-1/29 e^{-2ti} s_2^2 i + 2/145 e^{-2ti} s_2^2 + 1/29 e^{2ti} s_1^2 i + 2/145 e^{2ti} s_1^2)$

$u_4 = \varepsilon(-1/3 e^{-4ti} s_4^2 i - 1/3 e^{-4ti} s_4^2 + 1/8 e^{-2ti} s_4 \epsilon i - 1/8 e^{-2ti} s_2^2 + 1/3 e^{4ti} s_3^2 i - 1/3 e^{4ti} s_3^2 - 1/8 e^{2ti} s_3 \epsilon i - 1/8 e^{2ti} s_1^2 - s_2 s_1) + e^{-2ti} s_4 + e^{2ti} s_3$

$u_5 = \varepsilon(-8/51 e^{-4ti} s_4^2 i - 2/51 e^{-4ti} s_4^2 - 11/40 e^{-2ti} s_4 \epsilon i - 1/5 e^{-2ti} s_4 \epsilon + 2/5 e^{-2ti} s_2^2 i + 3/40 e^{-2ti} s_2^2 + 8/51 e^{4ti} s_3^2 i - 2/51 e^{4ti} s_3^2 + 11/40 e^{2ti} s_3 \epsilon i - 1/5 e^{2ti} s_3 \epsilon - 2/5 e^{2ti} s_1^2 i + 3/40 e^{2ti} s_1^2 + 2s_4 s_3 + s_2 s_1) + e^{-2ti} s_4 + e^{2ti} s_3$

$u_6 = \varepsilon(-1/102 e^{-4ti} s_4^2 i + 7/34 e^{-4ti} s_4^2 + 11/40 e^{-2ti} s_4 \epsilon i + 13/40 e^{-2ti} s_4 \epsilon - 11/40 e^{-2ti} s_2^2 i - 3/40 e^{-2ti} s_2^2 + 1/102 e^{4ti} s_3^2 i + 7/34 e^{4ti} s_3^2 - 11/40 e^{2ti} s_3 \epsilon i + 13/40 e^{2ti} s_3 \epsilon + 11/40 e^{2ti} s_1^2 i - 3/40 e^{2ti} s_1^2 - 3s_4 s_3 - s_2 s_1) + e^{-2ti} s_4 i - e^{-2ti} s_4 - e^{2ti} s_3 i - e^{2ti} s_3$

Centre manifold ODEs $\dot{s}_1 = \varepsilon^2(-1/130 s_4 s_3 s_1 i + 1/26 s_4 s_3 s_1 - 92/28275 s_2 s_1^2 i - 4/1885 s_2 s_1^2 - 1/8 s_1 a^2 i) + 1/2 \varepsilon s_1 a$

$$\dot{s}_2 = \varepsilon^2 (1/130 s_4 s_3 s_2 i + 1/26 s_4 s_3 s_2 + 92/28275 s_2^2 s_1 i - 4/1885 s_2^2 s_1 + 1/8 s_2 a^2 i) + 1/2 \varepsilon s_2 a$$

$$\dot{s}_3 = \varepsilon^2 (-223/204 s_4 s_3^2 i - 167/68 s_4 s_3^2 - 1/2 s_3 s_2 s_1 i - s_3 s_2 s_1 - 1/16 s_3 \varepsilon^2 i - 1/4 s_1^2 a - 1/16 s_1^2 \varepsilon) + \varepsilon (1/2 s_3 \varepsilon + 1/2 s_1^2 i)$$

$$\dot{s}_4 = \varepsilon^2 (223/204 s_4^2 s_3 i - 167/68 s_4^2 s_3 + 1/2 s_4 s_2 s_1 i - s_4 s_2 s_1 + 1/16 s_4 \varepsilon^2 i - 1/4 s_2^2 a - 1/16 s_2^2 \varepsilon) + \varepsilon (1/2 s_4 \varepsilon - 1/2 s_2^2 i)$$

2.20 Fudge an oscillatory mode

With frequency two, with a system with one slow mode. Couple them with something ad hoc.

$$\dot{u}_1 = \varepsilon (u_4 u_1 + u_2 u_1) - 2u_3 - 2u_2$$

$$\dot{u}_2 = -2u_3 - 3u_2 + u_1$$

$$\dot{u}_3 = 2u_3 + 3u_2 + u_1$$

$$\dot{u}_4 = \varepsilon (-u_4^2 - u_2 u_1) + u_5 - u_4$$

$$\dot{u}_5 = \varepsilon u_5^2 - u_5 + u_4$$

```

288 if thecase=oscmeanflow then begin
289   ff_:=tp mat((
290     -2*u2-2*u3+u4*u1+u1*u2,
291     u1-3*u2-2*u3,
292     u1+3*u2+2*u3,
293     -u4+u5-u4^2-u1*u2,
294     +u4-u5+u5^2
295   ));
296   evalm_:=mat((2*i,-2*i,0));
297   ee_:=tp mat((1,1,-1-i,0,0),(1,1,-1+i,0,0)
298     ,(0,0,0,1,1));
299   zz_:=tp mat((1,-i,-i,0,0),(1,+i,+i,0,0)
300     ,(0,0,0,1,1));
301 end;
```


The centre manifold $u_1 = \varepsilon(1/3 e^{-4ti} s_2^2 i + 1/8 e^{-2ti} s_3 s_2 i - 1/3 e^{4ti} s_1^2 i - 1/8 e^{2ti} s_3 s_1 i) + e^{-2ti} s_2 + e^{2ti} s_1$

$u_2 = \varepsilon(5/51 e^{-4ti} s_2^2 i - 1/17 e^{-4ti} s_2^2 - 11/40 e^{-2ti} s_3 s_2 i - 1/5 e^{-2ti} s_3 s_2 - 5/51 e^{4ti} s_1^2 i - 1/17 e^{4ti} s_1^2 + 11/40 e^{2ti} s_3 s_1 i - 1/5 e^{2ti} s_3 s_1 - 2s_2 s_1) + e^{-2ti} s_2 + e^{2ti} s_1$

$u_3 = \varepsilon(-5/51 e^{-4ti} s_2^2 i - 11/102 e^{-4ti} s_2^2 + 11/40 e^{-2ti} s_3 s_2 i + 13/40 e^{-2ti} s_3 s_2 + 5/51 e^{4ti} s_1^2 i - 11/102 e^{4ti} s_1^2 - 11/40 e^{2ti} s_3 s_1 i + 13/40 e^{2ti} s_3 s_1 + 3s_2 s_1) + e^{-2ti} s_2 i - e^{-2ti} s_2 - e^{2ti} s_1 i - e^{2ti} s_1$

$u_4 = \varepsilon(-9/40 e^{-4ti} s_2^2 i - 1/20 e^{-4ti} s_2^2 + 9/40 e^{4ti} s_1^2 i - 1/20 e^{4ti} s_1^2 - 1/2 s_2 s_1) + s_3$

$u_5 = \varepsilon(-1/40 e^{-4ti} s_2^2 i + 1/20 e^{-4ti} s_2^2 + 1/40 e^{4ti} s_1^2 i + 1/20 e^{4ti} s_1^2 + 1/2 s_3^2 + 1/2 s_2 s_1) + s_3$

Centre manifold ODEs $\dot{s}_1 = \varepsilon^2(-1/16 s_3^2 s_1 i - 1/4 s_3^2 s_1 - 421/4080 s_2 s_1^2 i - 887/680 s_2 s_1^2) + 1/2 \varepsilon s_3 s_1$

$\dot{s}_2 = \varepsilon^2(1/16 s_3^2 s_2 i - 1/4 s_3^2 s_2 + 421/4080 s_2^2 s_1 i - 887/680 s_2^2 s_1) + 1/2 \varepsilon s_3 s_2$

$\dot{s}_3 = \varepsilon^2(s_3^3 + 6/5 s_3 s_2 s_1) - \varepsilon s_2 s_1$

Used this system for a benchmark to compare several ways of handling matrices and vectors. This analysis using `e_` as basis for matrices and vectors takes about a second or two in the following five iterations.

```

302 lengthres := 10
303 Time: 20 ms
304 lengthres := 124
305 Time: 120 ms
306 lengthres := 289
307 Time: 420 ms
308 lengthres := 169
309 Time: 580 ms
310 lengthres := 1
311 Time: 420 ms
312 SUCCESS: converged to an expansion

```

2.21 Modulate Duffing oscillation

Tests that this code generates complex amplitude model for purely oscillating dynamics. Here model the frequency correction in the Duffing oscillator $\ddot{u} + u - u^3 = 0$. Code for $u_1 = u$ and $u_2 = \dot{u}$.

$$\dot{u}_1 = u_2$$

$$\dot{u}_2 = \varepsilon u_1^3 - u_1$$

```

313 if thecase=modulateduffing then begin
314 ff_:=tp mat((u2,-u1+u1^3-small*2*nu*u2));
315 evalm_:=mat((i,-i));
316 ee_:=tp mat((1,i),(1,-i));
317 zz_:=tp mat((1,i),(1,-i));
318 end;
```

Find the coordinate transform is $u_1 = \varepsilon(-1/8 e^{-3ti} s_2^3 + 3/4 e^{-ti} s_2^2 s_1 - 1/8 e^{3ti} s_1^3 + 3/4 e^{ti} s_2 s_1^2) + e^{-ti} s_2 + e^{ti} s_1$ where the amplitudes evolve according to $\dot{s}_1 = -51/16 \varepsilon^2 s_2^2 s_1^3 i - 3/2 \varepsilon s_2 s_1^2 i$ and its complex conjugate. This correctly predicts the frequency shift in the Duffing oscillator.

2.22 Modulate another oscillation

Retest that this code generates complex amplitude model for purely oscillating dynamics. Here model the frequency correction in the oscillator $\ddot{u} + u + \dot{u}^3 = 0$. Code for $u_1 = u$ and $u_2 = \dot{u}$.

$$\dot{u}_1 = u_2$$

$$\dot{u}_2 = -\varepsilon u_2^3 - u_1$$

```

319 if thecase=modulateoscillator then begin
320 ff_:=tp mat((u2,-u1-u2^3));
321 evalm_:=mat((i,-i));
322 ee_:=tp mat((1,i),(1,-i));
323 zz_:=tp mat((1,i),(1,-i));
324 end;
```

The coordinate transform $u_1 = e^{-ti}s_2 + e^{ti}s_1 + \varepsilon(1/8 e^{-3ti}s_2^3i + 3/4 e^{-ti}s_2^2s_1i - 1/8 e^{3ti}s_1^3i - 3/4 e^{ti}s_2s_1^2i)$ looks fine; although note that here higher orders do differ to other work due to the orthogonality I build in. The evolution seems appropriate: $\dot{s}_1 = -3/2\varepsilon s_2s_1^2 - 27/16\varepsilon^2s_2^2s_1^3i$

2.23 An example from Iulian Stoleriu

Consider the case [Stoleriu \(2012\)](#) calls $(3\pi/4, k^2/2)$. Use Taylor expansions for trigonometric functions in the ODEs. Eigenvalues are ± 1 and $\pm i$, so we find the centre manifold among stable and unstable modes. Sometimes we can have a parameter (here σ) in the linear operator, but may need to specify its real and imaginary parts.

```

325 if thecase=StoleriuOne then begin
326   let {repart(sigma)=>sigma, impart(sigma)=>0};
327   ff_:=tp mat((
328     u2,
329     sigma*u3+u1^2/2-small*u1^4/24,
330     u4,
331     u1/sigma+u3*u1+(u3+1/sigma)*(-small*u1^3/6)
332   ));
333   evalm_:=mat((i,-i));
334   ee_:=tp mat((sigma,i*sigma,-1,-i),(sigma,-i*sigma,-1,+i));
335   zz_:=tp mat((+i,-1,-i*sigma,sigma),(-i,-1,+i*sigma,sigma));
336 end;
```

A centre manifold is $x = u_1 = \varepsilon(-1/5 e^{-2ti}s_2^2\sigma^2 - 1/5 e^{2ti}s_1^2\sigma^2 + 2s_2s_1\sigma) + e^{-ti}s_2\sigma + e^{ti}s_1\sigma$ and $y = u_3 = \varepsilon(3/10 e^{-2ti}s_2^2\sigma + 3/10 e^{2ti}s_1^2\sigma - s_2s_1\sigma) - e^{-ti}s_2 - e^{ti}s_1$. On this centre manifold the oscillations have a frequency shift, but no amplitude evolution (to this order nor the next): $\dot{s}_1 = -6/5\varepsilon^2s_2s_1^2i\sigma^2$. Remember the system is unstable due to the unstable mode.

2.24 An second example from Iulian Stoleriu

Consider the case [Stoleriu \(2012\)](#) calls $(\pi/2, 0)$. Use Taylor expansions for trigonometric functions in the ODEs. Eigenvalues are $\pm i$, multiplicity two, so we find modulation equations for coupled oscillators.

The system is

- $\dot{u}_1 = u_2$
- $\dot{u}_2 = -1/120\varepsilon^2 u_1^5 + 1/6\varepsilon u_1^3 + u_3\sigma - u_1$
- $\dot{u}_3 = u_4$
- $\dot{u}_4 = -1/24\varepsilon^2 u_3 u_1^4 + 1/2\varepsilon u_3 u_1^2 - u_3$

```

337 if thecase=StoleriuTwo then begin
338   ff_:=tp mat((
339     u2,
340     -u1+u1^3/6-small*u1^5/120+sigma*u3,
341     u4,
342     -u3+u3*(u1^2/2-small*u1^4/24)
343   ));
344   evalm_:=mat((i,-i,i,-i));
345   ee_:=tp mat((1,i,0,0),(1,-i,0,0),(0,0,1,i),(0,0,1,-i));
346   zz_:=tp mat((1,i,0,0),(1,-i,0,0),(0,0,1,i),(0,0,1,-i));
347   toosmall:=3;
348 end;
```

This used to take five iterates to construct the coordinate transform and modulation equations, but now less as the off-diagonal term is made small by the linear adjustment. The original variables are approximately

- $x = u_1 = 1/4 e^{-ti} s_4 \sigma + e^{-ti} s_2 + 1/4 e^{ti} s_3 \sigma + e^{ti} s_1$
- $y = u_3 = e^{-ti} s_4 + e^{ti} s_3$

The modulation equations are the following, and their complex conjugates:

- $\dot{s}_1 = \varepsilon \left(-1/64 s_4 s_3^2 i \sigma^3 - 3/32 s_4 s_3 s_1 i \sigma^2 - 1/8 s_4 s_1^2 i \sigma - 5/64 s_3^2 s_2 i \sigma^2 - 1/4 s_3 s_2 s_1 i \sigma - 1/4 s_2 s_1^2 i \right) - 1/2 s_3 i \sigma;$
- $\dot{s}_3 = \varepsilon \left(-3/64 s_4 s_3^2 i \sigma^2 - 1/4 s_4 s_3 s_1 i \sigma - 1/4 s_4 s_1^2 i - 1/8 s_3^2 s_2 i \sigma - 1/2 s_3 s_2 s_1 i \right).$

Since every term is multiplied by i one expects there to be just frequency shifts, but there are oscillator interaction terms as well. These should be equivalent to the averaging method, but more easily extended to higher order (just change parameter `toosmall`).

2.25 Periodic chronic myelogenous leukemia

Ion & Georgescu (2013) explored Hopf bifurcations in a delay differential equation modelling leukaemia:²

$$\dot{x} = -\frac{x(t)}{1+x(t)^n} - \delta x(t) + \frac{kx(t-r)}{1+x(t-r)^n}$$

For simplicity we fix upon parameters $n = 2$, $\delta \approx 1/8$, $k = 3/2$ and time delay $r = 64/3$; that is,

$$\dot{x} = -\frac{x(t)}{1+x(t)^2} - \left(\frac{1}{8} + \delta'\right)x(t) + \frac{\frac{3}{2}x(t-r)}{1+x(t-r)^2}$$

Near these parameters the equilibrium $x = X = \sqrt{3}$ perhaps undergoes a Hopf bifurcation. ‘Perhaps’ because instead of a precise time delay, we model $x(t-r)$ via two intermediaries in the system, after defining $x(t) = X + u_1(t)$,

$$\begin{aligned}\dot{u}_1 &= -\frac{(X+u_1)}{1+(X+u_1)^2} - \left(\frac{1}{8} + \delta'\right)(X+u_1) + \frac{\frac{3}{2}(X+u_3)}{1+(X+u_3)^2}, \\ \dot{u}_2 &= \frac{3}{32}(u_1 - u_2), \\ \dot{u}_3 &= \frac{3}{32}(u_2 - u_3).\end{aligned}$$

²Their parameter β_0 is absorbed in a time scaling.

This system does undergo a Hopf bifurcation as δ' decreases through zero. My code only analyses multinomial forms, so Taylor expand the rational function:

$$\begin{aligned}\frac{X+u}{1+(X+u)^2} &= \frac{X}{1+X^2} + \frac{1-X^2}{(1+X^2)^2}u + \frac{X(X^2-3)}{(1+X^2)^3}u^2 + \frac{-1+6X^2-X^4}{(1+X^2)^4}u^3 + \dots \\ &= \frac{\sqrt{3}}{4} - \frac{1}{8}u + 0u^2 + \frac{1}{32}u^3 + \dots \quad \text{at } X = \sqrt{3}.\end{aligned}$$

```

349 if thecase=delayprolif then begin
350 ff_:=tp mat((
351     -3/16*u3-u1^3/32-small*delta*(sqrt(3)+u1)+3/64*u3^3,
352     3/32*u1-3/32*u2,
353     3/32*u2-3/32*u3
354 ));
355 evalm_:=mat((3/32*i,-3/32*i));
356 ee_:=tp mat((1,1/2-i/2,-i/2),(1,1/2+i/2,+i/2));
357 zz_:=tp mat((1,-i,-1-i),(1,+i,-1+i));
358 toosmall:=2;
359 factor delta,s;
360 end;

```

The specified dynamical system

$$\dot{u}_1 = \varepsilon \left(-\sqrt{3}\delta + 3/64u_3^3 - 1/32u_1^3 - u_1\delta \right) - 3/16u_3$$

$$\dot{u}_2 = -3/32u_2 + 3/32u_1$$

$$\dot{u}_3 = -3/32u_3 + 3/32u_2$$

The centre manifold

$$u_1 = e^{-3t/32i}s_2 + e^{3t/32i}s_1$$

$$u_2 = 1/2 e^{-3t/32i}s_2i + 1/2 e^{-3t/32i}s_2 - 1/2 e^{3t/32i}s_1i + 1/2 e^{3t/32i}s_1$$

$$u_3 = 1/2 e^{-3t/32i}s_2i - 1/2 e^{3t/32i}s_1i$$

Centre manifold ODEs

$$\dot{s}_1 = \varepsilon(3/256s_2s_1^2i - 21/512s_2s_1^2 + 1/5s_1\delta i - 2/5s_1\delta)$$

$$\dot{s}_2 = \varepsilon(-3/256s_2^2s_1i - 21/512s_2^2s_1 - 1/5s_2\delta i - 2/5s_2\delta)$$

These indicate that $\vec{s} = \vec{0}$ is stable for $\delta' \geq 0$. For parameter $\delta' < 0$ there is a stable limit cycle of amplitude $|s_j| = 16\sqrt{\frac{-2\delta'}{105}}$.

2.25.1 Delayed version

Return to the original system linearised about $x = \sqrt{3}$, the following finds the spectrum and identifies a Hopf bifurcation of frequency 3/16.

```
361 % linearised about x=sqrt(3), freq is 3/16
362 delta=1/8, k=1+4*delta, r=8/3*pi
363 ce=@(z) -z+1/8-delta-k/8*exp(-r*z)
364 lams=fsolve(ce,randn(100,2)*[1;3*i]/2)
365 plot(real(lams),imag(lams),'o')
```

The following works only by careful use of smallness.

```
366 if thecase=delayedprolif then begin
367   r3:=sqrt(3);
368   delta:=1/8; k:=1+4*delta; r:=8/3*pi;
369   ff_:=tp mat((
370     -r3*(1/4-3/8/r3*u1+1/8*u1^2-3/32/r3*u1^3*small)
371     -u1*(1/4-3/8/r3*u1+1/8*u1^2*small)
372 %    -(r3+u1)*(1/4-3/8/r3*u1+1/8*u1^2-3/32/r3*small^2*u1^3)
373     -delta*(r3+u1)
374     +k*r3*(1/4-3/8/r3*u1(r)+1/8*u1(r)^2-3/32/r3*u1(r)^3*small)
375     +k*u1(r)*(1/4-3/8/r3*u1(r)+1/8*u1(r)^2*small)
376 %    +k*(r3+u1(r))*(1/4-3/8/r3*u1(r)+1/8*u1(r)^2-small^2*3/32/r3
377     ));
378   evalm_:=mat((3/16*i,-3/16*i));
379   ee_:=tp mat((1),(1));
380   zz_:=tp mat((1),(1));
```

```

381 toosmall:=4;
382 factor s;
383 end;

```

The specified dynamical system

$$\dot{u}_1 = \varepsilon^2 (3/64 D_{t, (8\pi)/3} (u_1)^3 - 1/32 u_1^3) - 3/16 D_{t, (8\pi)/3} (u_1)$$

The centre manifold

$$u_1 = s_2^3 \varepsilon^2 (-1/24 e^{(-9ti/16)} i + 1/16 e^{(-9ti/16)}) + s_2 e^{(-3ti/16)} + s_1^3 \varepsilon^2 (1/24 e^{(9ti/16)} i + 1/16 e^{(9ti/16)}) + s_1 e^{(3ti/16)}$$

Centre manifold ODEs

$$\begin{aligned} \dot{s}_1 &= s_2 s_1^2 \varepsilon^2 (3/16 i \pi - 9/16 i - 9/32 \pi - 3/8) / (\pi^2 + 4) \\ \dot{s}_2 &= s_2^2 s_1 \varepsilon^2 (-3/16 i \pi + 9/16 i - 9/32 \pi - 3/8) / (\pi^2 + 4) \end{aligned}$$

2.26 Nonlinear normal modes

[Renson et al. \(2012\)](#) explored finite element construction of the nonlinear normal modes of a pair of coupled oscillators. Defining two new variables one of their example systems is

$$\begin{aligned} \dot{x}_1 &= x_3, \\ \dot{x}_2 &= x_4, \\ \dot{x}_3 &= -2x_1 + x_2 - \frac{1}{2}x_1^3 + \frac{3}{10}(-x_3 + x_4), \\ \dot{x}_4 &= x_1 - 2x_2 + \frac{3}{10}(x_3 - 2x_4). \end{aligned}$$

In the following code, force the linear damping to be effectively small (which then makes it small squared); consequently scale the smallness of the cubic nonlinearity.


```

384 if thecase=normalmodes then begin
385   r3:=sqrt(3);
386   ff_:=tp mat((
387     u3,
388     u4,
389     -2*u1+u2-small*u1^3/2+small*3/10*(-u3+u4),
390     u1-2*u2+small*3/10*(u3-2*u4)
391   ));
392   evalm_:=mat((i,-i,r3*i,-r3*i));
393   ee_:=tp mat((1,1,+i,+i),(1,1,-i,-i)
394     ,(1,-1,i*r3,-i*r3),(1,-1,-i*r3,i*r3));
395   zz_:=tp mat((1,1,+i,+i),(1,1,-i,-i)
396     ,(-i*r3,+i*r3,1,-1),(+i*r3,-i*r3,1,-1));
397   toosmall:=3;
398 end;

```

The square root frequencies do not cause any trouble (although may need to reformat the LaTeX of the cis operator). In the model, observe that $s_1 = s_2 = 0$ is invariant, as is $s_3 = s_4 = 0$. These are the nonlinear normal modes.

The centre manifold

$$\begin{aligned}
 u_1 &= e^{-\sqrt{3}ti} s_4 + e^{-ti} s_2 + e^{\sqrt{3}ti} s_3 + e^{ti} s_1 \\
 u_2 &= -e^{-\sqrt{3}ti} s_4 + e^{-ti} s_2 - e^{\sqrt{3}ti} s_3 + e^{ti} s_1 \\
 u_3 &= -\sqrt{3} e^{-\sqrt{3}ti} s_4 i - e^{-ti} s_2 i + \sqrt{3} e^{\sqrt{3}ti} s_3 i + e^{ti} s_1 i \\
 u_4 &= \sqrt{3} e^{-\sqrt{3}ti} s_4 i - e^{-ti} s_2 i - \sqrt{3} e^{\sqrt{3}ti} s_3 i + e^{ti} s_1 i
 \end{aligned}$$

Centre manifold ODEs

$$\begin{aligned}
 \dot{s}_1 &= \varepsilon (3/4 s_4 s_3 s_1 i + 3/8 s_2 s_1^2 i - 3/40 s_1) \\
 \dot{s}_2 &= \varepsilon (-3/4 s_4 s_3 s_2 i - 3/8 s_2^2 s_1 i - 3/40 s_2) \\
 \dot{s}_3 &= \varepsilon (1/8 \sqrt{3} s_4 s_3^2 i + 1/4 \sqrt{3} s_3 s_2 s_1 i - 3/8 s_3)
 \end{aligned}$$

$$\dot{s}_4 = \varepsilon \left(-1/8\sqrt{3}s_4^2s_3i - 1/4\sqrt{3}s_4s_2s_1i - 3/8s_4 \right)$$

2.27 Periodically forced van der Pol oscillator

Hinvi et al. (2013) used renormalisation group to explore periodically forced van der Pol oscillator

$$\ddot{x} + x - \epsilon(1 - ax^2 - bx^2)\dot{x} = \epsilon c \sin \Omega t.$$

Introducing $u_1 = x$, rewrite as the system

$$\begin{aligned}\dot{u}_1 &= u_2, \\ \dot{u}_2 &= -u_1 + \epsilon(1 - au_1^2 - bu_2^2)u_2 + \epsilon cu_3, \\ \dot{u}_3 &= \Omega u_4, \\ \dot{u}_4 &= -\Omega u_3.\end{aligned}$$

This system has eigenvalues $\pm i$ and $\pm i\Omega$ so we seek the modulation equations of the oscillations.

Only the directly resonant case appears to be interesting, so set $\Omega = 1$, and then perturb it in the equations.

```

399 if thecase=forcedvdp then begin
400   om:=1;
401   ff_:=tp mat((
402     +u2,
403     -u1+small*(1-a*u1^2-b*u2^2)*u2+small*c*u3,
404     +om*u4*(1+small*omega),
405     -om*u3*(1+small*omega)
406   ));
407   evalm_:=mat((i,-i,om*i,-om*i));
408   ee_:=tp mat((1,+i,0,0),(1,-i,0,0)
409     ,(0,0,1,+i),(0,0,1,-i));
410   zz_:=tp mat((1,+i,0,0),(1,-i,0,0)
411     ,(0,0,1,+i),(0,0,1,-i));
412   toosmall:=4;
413 end;
```

2.28 Slow manifold of Lorenz 1986 model

In this case we construct the slow sub-centre manifold, analogous to quasi-geostrophy, in order to disentangle the slow dynamics from fast oscillations, analogous to gravity waves. The normals to the isochrons determine ‘balancing’ onto the slow manifold.

```

414 if thecase=lorenz86slow then begin
415   factor b;
416   ff_:=tp mat((-u2*u3+b*u2*u5
417     ,u1*u3-b*u1*u5
418     ,-u1*u2
419     ,-u5
420     ,+u4+b*u1*u2));
421   evalm_:=mat((0,0,0));
422   ee_:=zz_:=tp mat((1,0,0,0,0),(0,1,0,0,0),(0,0,1,0,0));
423   toosmall:=4;
424 end;
```

The centre manifold These give the location of the centre manifold in terms of parameters s_j .

$$u_1 = s_1$$

$$u_2 = s_2$$

$$u_3 = s_3$$

$$u_4 = -b\varepsilon s_2 s_1$$

$$u_5 = b\varepsilon^2(-s_3 s_2^2 + s_3 s_1^2)$$

Centre manifold ODEs The system evolves on the centre manifold such that the parameters evolve according to these ODEs.

$$\dot{s}_1 = b^2\varepsilon^3(-s_3 s_2^2 + s_3 s_1^2) - \varepsilon s_3 s_2$$

$$\dot{s}_2 = b^2 \varepsilon^3 (s_3 s_2^2 s_1 - s_3 s_1^3) + \varepsilon s_3 s_1$$

$$\dot{s}_3 = -\varepsilon s_2 s_1$$

Normals to isochrons at the slow manifold The normal vector $\vec{z}_j := (z_{j1}, \dots, z_{jn})$

$$z_{11} = b^2 \varepsilon^2 s_2^2 + 1$$

$$z_{12} = b^2 \varepsilon^2 s_2 s_1$$

$$z_{13} = 0$$

$$z_{14} = b^3 \varepsilon^3 (s_2^3 - s_2 s_1^2) + b \varepsilon^3 (-s_2^3 + s_2 s_1^2) + b \varepsilon s_2$$

$$z_{15} = 0$$

$$z_{21} = -b^2 \varepsilon^2 s_2 s_1$$

$$z_{22} = -b^2 \varepsilon^2 s_1^2 + 1$$

$$z_{23} = 0$$

$$z_{24} = b^3 \varepsilon^3 (-s_2^2 s_1 + s_1^3) + b \varepsilon^3 (s_2^2 s_1 - s_1^3) - b \varepsilon s_1$$

$$z_{25} = 0$$

$$z_{31} = 0$$

$$z_{32} = 0$$

$$z_{33} = 1$$

$$z_{34} = -4b \varepsilon^3 s_3 s_2 s_1$$

$$z_{35} = b \varepsilon^2 (-s_2^2 + s_1^2)$$

2.28.1 Normal form shows drift from the fast waves

Finds that any fast waves will generate a mean drift effect on the slow dynamics (in the $s_3 \approx u_3$ equation), an effect quadratic in amplitude of the fast waves.

```

425 if thecase=lorenz86norm then begin
426   factor b;
427   ff_:=tp mat((-u2*u3+b*u2*u5
428     ,u1*u3-b*u1*u5
429     ,-u1*u2
430     ,-u5
431     ,+u4+b*u1*u2));
432   evalm_:=mat((0,0,0,i,-i));
433   ee_:=zz_:=tp mat((1,0,0,0,0),(0,1,0,0,0),(0,0,1,0,0)
434     ,(0,0,0,1,-i),(0,0,0,1,+i));
435   toosmall:=4;
436 end;
```

2.29 Check the dimensionality of specified system

Extract dimension information from the specification of the dynamical system: seek mD centre manifold of an nD system.

```

437 if thecase=myweb then begin
438   % out "cmsyso.txt"$
439   ODE_function:=ff_;
440   subspace_eigenvalues:=evalm_;
441   subspace_eigenvectors:=ee_;
442   adjoint_eigenvectors:=zz_;
443 end;

444 write "total no. of modes ",
445 n:=part(length(ee_),1);
446 write "no. of manifold modes ",
447 m:=part(length(ee_),2);
```

```

448 if {length(evalm_),length(zz_),length(ee_),length(ff_)}
449     ={{1,m},{n,m},{n,m},{n,1}}
450     then write "Input dimensions are OK"
451     else <<write "INCONSISTENT INPUT DIMENSIONS, I QUIT";
452         quit>>;

```

Need an $m \times m$ identity matrix for normalisation of the isochron projection.

```

453 eyem_:=for j:=1:m sum e_(j,j)$

```

3 Dissect the linear part

Define exponential $\exp(u) = e^u$. Do not (yet) invoke the simplification of $\exp(0)$ as I want it to label modes of no oscillation/growth, zero frequency.

```

454 clear exp;
455 operator exp;
456 let { df(exp(~u),t) => df(u,t)*exp(u)
457     , exp(~u)*exp(~v) => exp(u+v)
458     , exp(~u)^~p => exp(p*u)
459     };

```

Need function `conj_` to do parsimonious complex conjugation.

```

460 procedure conj_(a)$
461     ((a where {i=>i__}) where {i__=>-i})$

```

Make an array of eigenvalues for simplicity (instead of a matrix).

```

462 array evl_(m);
463 for j:=1:m do evl_(j):=evalm_(1,j);

```

Decide the presumed nature of the invariant manifold from an “or” of the eigenvalues.

```

464 slowM_:=centreM_:=stableM_:=unstabM_:=0$
465 for j:=1:m do begin
466     slowM_:=if evl_(j)=0 then 1 else slowM_;

```

```

467     centreM_:=if repart(evl_(j))==0 and evl_(j) neq 0
468     then 1 else centreM_;
469     stableM_:=if repart(evl_(j))<0 then 1 else stableM_;
470     unstabM_:=if repart(evl_(j))>0 then 1 else unstabM_;
471 end;
472 natureMan_:=part({"EMPTY","Slow","Centre","Centre"
473     ,"Stable","Slow-stable","Centre-stable","Centre-stable"
474     ,"Unstable","Slow-unstable","Centre-unstable","Centre-unstab
475     ,"Invariant","Invariant","Fast","Invariant"
476     },1+slowM_+2*(centreM_+2*(stableM_+2*unstabM_)));

```

3.1 Normalise the adjoint eigenvectors

When we include delay differential equations, then we need to account for the history of the eigenvector as well. Hence multiply each eigenvector by its oscillating factor, $e^{i\omega t}$, and then take the mean. This multiplication by its oscillating factor should not make any difference for non-delay equations by the natural orthogonality of left and right eigenvectors of different eigenvalues. Reduce implements `conj` via `repart` and `impart`, so let `repart` do the conjugation of the cis factors.

Note: the ‘left eigenvectors’ have to be the eigenvectors of the complex conjugate transpose, and for the complex conjugate frequency. This seems best: for example, when the linear operator is $\begin{bmatrix} 0 & 1 \\ -1 & 0 \end{bmatrix}$ then the adjoint and the right eigenvectors are the same.

For un/stable manifolds we have to cope with complex frequencies. Seems to need `zz_` to have minus?? complex conjugated frequency so store in `cexp_`—which is the same as `dexp_` for real frequencies?? Need to decide on the inner product, especially to cater for the case of DDEs??

```

477 matrix aa_(m,m),dexp_(m,m),cexp_(m,m);
478 for j:=1:m do dexp_(j,j):=exp(evl_(j)*t);
479 for j:=1:m do cexp_(j,j):=exp(-conj_(evl_(j))*t);
480 aa_:=(tp map(conj_(~b),ee_*dexp_)*zz_*cexp_ );

```

```

481 write "Normalising the left-eigenvectors:";
482 aa_:=(aa_ where {exp(0)=>1, exp(~a)=>0 when a neq 0});
483 if det(aa_)=0 then << write
484     "ORTHOGONALITY ERROR IN EIGENVECTORS; I QUIT"; quit>>;
485 zz_:=zz_*aa_^(-1);

```

3.2 Operator to represent delays

Introduce an operator to represent delay factors more conveniently for analysis.

What do we do about $\cos(a)$ and $\sin(a)$ in the following?? for general eigenvalues??

```

486 operator d_; linear d_;
487 let { d_(~a~p,t,~dt)=>d_(a,t,dt)^p
488     , d_(~a*~b,t,~dt)=>d_(a,t,dt)*d_(b,t,dt)
489     , d_(cis(~a),t,~dt)=>cis(a)
490         *sub(t=-dt,cos(a)+i*sin(a))
491     , df(d_(~a,t,~dt),~b)=>d_(df(a,b),t,dt)
492     , d_(~a,t,0)=>a
493     , d_(d_(~a,t,~dta),t,~dtb)=>d_(a,t,dta+dtb)
494 };

```

Now rewrite the (delay) factors in terms of this operator. Need to say that the symbol u depends upon time; later we write things into u and this dependence would be forgotten. For the moment limit to a maximum of nine ODEs.

Create synonyms.

```

495 for k:=1:n do set(mkid(u,k),u(k));

496 ff:=(ff_ where {u(~k,~dt)=>d_(u(k),t,dt)})$
497 %somerules:={}$
498 %depend u1,t;somerules:=(u1(~dt)=d_(u1,t,dt)).somerules$
499 %depend u2,t;somerules:=(u2(~dt)=d_(u2,t,dt)).somerules$

```



```

500 %depend u3,t;somerules:=(u3(~dt)=d_(u3,t,dt)).somerules$
501 %depend u4,t;somerules:=(u4(~dt)=d_(u4,t,dt)).somerules$
502 %depend u5,t;somerules:=(u5(~dt)=d_(u5,t,dt)).somerules$
503 %depend u6,t;somerules:=(u6(~dt)=d_(u6,t,dt)).somerules$
504 %depend u7,t;somerules:=(u7(~dt)=d_(u7,t,dt)).somerules$
505 %depend u8,t;somerules:=(u8(~dt)=d_(u8,t,dt)).somerules$
506 %depend u9,t;somerules:=(u9(~dt)=d_(u9,t,dt)).somerules$
507 %ff_:=(ff_ where somerules)$

```

3.3 Linearise at the origin

Assume the equilibrium is at the origin. Find the linear operator at the equilibrium. Include `small=0` as we notionally adjoin it in the list of variables. Do not need to here make small any non-zero forcing at the equilibrium as it gets multiplied by `small` later??

```

508 matrix ll_(n,n);
509 for j:=1:n do for k:=1:n do begin
510   ll_(j,k):=df(ff_(j,1),u(k));
511   ll_(j,k):=(ll_(j,k) where {small=>0,u(~1)=>0});
512 end;
513 write "Find the linear operator is";
514 ll_:=ll_;

```

We need a vector of unknowns for a little while: only used once.

```

515 matrix uvec(n,1);
516 for j:=1:n do uvec(j,1):=u(j);

```

3.4 Eigen-check

Variable `aa_` appears here as the diagonal matrix of frequencies. Check that the frequencies and eigenvectors are specified correctly.

Again need to worry about delays??

```

517 write "Check ",natureMan_," subspace linearisation ";
518 for j:=1:m do for k:=1:m do aa_(j,k):=0;
519 for j:=1:m do aa_(j,j):=evl_(j);
520 write %temporary write
521 reslin:=(ll_*(ee_*dexp_)-(ee_*dexp_)*aa_
522     where cis(~a)*d_(1,t,~dt)=>sub(t=-dt,cos(a)+i*sin(a))*cis(a)
523 ok_:=1$
524 for j:=1:n do for k:=1:m do
525     ok_:=if reslin(j,k)=0 then ok_ else 0$
526 if ok_ then write "Linearisation is OK";

```

Try to find a correction of the linear operator that is ‘close’. Multiply by the adjoint eigenvectors and then average over time: operator $\mathcal{L}_{\text{new}} := \mathcal{L} - \mathcal{L}_{\text{adj}}$ should now have zero residual. Lastly, correspondingly adjust the ODEs, since `lladj` does not involve delays we do not need delay operator transforms in the product.

Again delays??

```

527 if not ok_ then for iter:=1:2 do begin
528 write "WARNING: I NEED TO ADJUST LINEAR OPERATOR";
529 write
530 lladj:=reslin*tp map(conj_(~b),zz_*cexp_);
531 write
532 lladj:=(lladj where {exp(0)=>1, exp(~a)=>0 when a neq 0});
533 write
534 ll_:=ll_-lladj;
535 write
536 reslin:=(ll_*(ee_*dexp_)-(ee_*dexp_)*aa_
537     where exp(~a)*d_(1,t,~dt)=>sub(t=-dt,cos(a)+i*sin(a))*cis(a)
538 %for j:=1:n do for k:=1:m do
539 %     if reslin(j,k) neq 0 then << write
540 %         "OOPS, INCONSISTENT EIGENVALUES, EIGENVECTORS AND OPERATOR.
541 %         EMAIL ME; I QUIT"; write reslin:=reslin; rederr "aaaaah";qu
542 ok_:=1$
543 for j:=1:n do for k:=1:m do

```

```

544     ok_:=if reslin(j,k)=0 then ok_ else 0$
545 if ok_ then iter:=iter+1000;
546 end;
547 if not ok_ then << write
548     "OOPS, INCONSISTENT EIGENVALUES, EIGENVECTORS AND OPERATOR.
549     EMAIL ME; I QUIT"; rederr "aaaaah";quit >>;

```

3.5 Ameliorate the nonlinearity

Anything not in the linear operator gets multiplied by `small` to be treated as small in the analysis. The feature of the second alternative is that when a user invokes `small` then the power of smallness is not then changed; however, causes issues in the relative scaling of some terms, so restore to the original version.

This might need reconsidering ?? but the if always chooses the first simple alternative.

```

550 somerules:=for j:=1:n collect
551     (d_(1,t,~dt)*u(j)=d_(u(j),t,dt))$
552 ff_:=if 1 then small*ff_
553     else ff_-(1-small)*sub(small=0,ff_) +(1-small)
554     *(ll_*uvec where somerules)$

```

Any constant term in the equations `ff_` has to be multiplied by `exp(0)`.

```

555 ff_:=ff_+(exp(0)-1)*(ff_ where {small=>0,u(~1)=>0})$

```

From the matrix versions of the equations, create algebraic form using the matrix basis.

```

556 rhsfn_:=for i:=1:n sum e_(i,1)*ff_(i,1)$

```

Also, create the algebraic form of the jacobian transpose using the matrix basis: take the conjugate later when used.

```

557 rhsjact_:=for i:=1:n sum for j:=1:n sum
558     e_(j,i)*df(ff_(i,1),u(j))$

```

3.6 Store invariant manifold frequencies

Extract all the frequencies in the invariant manifold, and the set of all the corresponding modes in the invariant manifold variables. The slow modes have zero frequency. Remember the frequency set is not in the ‘correct’ order. Array `modes` stores the set of indices of all the modes of a given frequency.

```

559 array evals(m),modes(m);
560 neval:=0$ evalset:={} $
561 for j:=1:m do if not(evl_(j) member evalset) then begin
562   neval:=neval+1;
563   evals(neval):=evl_(j);
564   evalset:=evl_(j).evalset;
565   modes(neval):=for k:=j:m join
566     if evl_(j)=evl_(k) then {k} else {};
567 end;
```

Set a flag for the case of a slow manifold when all frequencies are zero, as then we compute the isochron projection. The next challenge is to get this isochron code working for the case of non-slow invariant manifolds.

```

568 itisSlowMan_:=if evalset={0} then 1 else 0$
569 if trace_ then write itisSlowMan_:=itisSlowMan_;
```

Put in the non-singular general case as the zero entry of the arrays.

```

570 evals(0):=geneval$
571 modes(0):={} $
```

3.7 Precompute matrices for updates

Precompute matrices to solve for updates for each of the critical frequencies, and the general case $\mathbf{k} = 0$. The matrix

$$\mathbf{llzz} = \begin{bmatrix} -\mathcal{L} + \partial_t & \mathcal{E}_0 \\ \mathcal{Z}_0^\dagger & 0 \end{bmatrix}$$

and then put its inverse in place. Subsequently, extract the blocks for the generalised inverses and solvability condition into `linvs` and `ginvs`.

```
572 matrix llzz(n+m,n+m);
573 array linvs(neval),ginvs(neval);
574 array l1invs(neval),g1invs(neval),l2invs(neval),g2invs(neval);
575 operator sp_; linear sp_;
576 for k:=0:neval do begin
```

Code the operator $\mathcal{L}\hat{v}$ where the delay is to only act on the oscillation part.

Again, what do we do about `cos()` and `sin()` of delays??

```
577   for ii:=1:n do for jj:=1:n do llzz(ii,jj):=(
578       -sub(small=0,ll_(ii,jj))
579       where d_(1,t,~dt)=>cos(freqs(k)*dt)-i*sin(freqs(k)*dt));
```

Code the operator $\partial\hat{v}/\partial t$ where it only acts on the oscillation part.

```
580   for j:=1:n do llzz(j,j):=evals(k)+llzz(j,j);
```

Now code the part leading to the solvability condition which arises from allowing the (oscillation) amplitude to evolve. Use operator `sp_` to extract the delay parts that subtly affect the updates of the evolution.

Again `cos()` and `sin()` here??

```
581   for j:=1:length(modes(k)) do
582     for ii:=1:n do llzz(ii,n+j):=ee_(ii,part(modes(k),j))
583       +(for jj:=1:n sum
584         sp_(ll_(ii,jj)*ee_(jj,part(modes(k),j)),d_)
585         where { sp_(1,d_)=>0
586             , sp_(d_(1,t,~dt),d_)=>dt*(
587                 cos(freqs(k)*dt)-i*sin(freqs(k)*dt))
588             });
```

Force the updates to be orthogonal to the left-eigenvectors in the complex conjugate transpose adjoint.,

```
589   for i:=1:length(modes(k)) do
```

```
590     for j:=1:n do llzz(n+i,j):=conj_(zz_(j,part(modes(k),i)));
```

Set the bottom-right corner of the matrix to zero.

```
591   for i:=1:length(modes(k)) do
592     for j:=1:m do llzz(n+i,n+j):=0;
```

Add some trivial rows and columns to make the matrix up to the same size for all frequencies.

```
593   for i:=length(modes(k))+1:m do begin
594     for j:=1:n+i-1 do llzz(n+i,j):=llzz(j,n+i):=0;
595     llzz(n+i,n+i):=1;
596   end;
```

Invert the matrix and unpack into arrays ready for use by the inversion operators.

```
597 if trace_ then write llzz:=llzz;
598   llzz:=llzz^(-1);
599 if trace_ then write llzz:=llzz;
600   linsv(k):=for i:=1:n sum for j:=1:n sum e_(i,j)*llzz(i,j);
601   ginsv(k):=for i:=1:length(modes(k)) sum
602     for j:=1:n sum e_(part(modes(k),i),j)*llzz(i+n,j);
```

Unpack the conjugate transpose for inverse operators used for the isochrons. A difference here is that the orthogonality condition is non-trivial (in the slow manifold we assumed amplitudes were exactly orthogonal to the left-eigenvectors), so we need to remember more parts of the inverse of the matrix. Will it need to be more subtle for the un/stable manifolds case??

```
603 %   realgeneval:={repart(geneval)=>geneval, impart(geneval)=>0}$
604   l1insv(k):=for i:=1:n sum for j:=1:n sum
605     e_(i,j)*conj_(llzz(j,i));
606   l2insv(k):=for i:=1:n sum for j:=1:length(modes(k)) sum
607     e_(i,part(modes(k),j))*conj_(llzz(j+n,i));
608   g1insv(k):=for i:=1:length(modes(k)) sum for j:=1:n sum
609     e_(part(modes(k),i),j)*(llzz(j,i+n)); %conj_??
610   g2insv(k):=
```

```

611     for i:=1:length(modes(k)) sum for j:=1:length(modes(k)) sum
612         e_(part(modes(k),i),part(modes(k),j))*conj_(llzz(j+n,i+n))
613 end;

```

3.8 Define operators that invoke these inverses

Decompose residuals into parts, and operate on each. First for the invariant manifold. But making `e_` non-commutative means that it does not get factored out of these linear operators: must post-multiply by `e_` because the linear inverse is a premultiply.

```

614 operator linv; linear linv;
615 let linv(e_(~j,~k)*exp(~a),exp)=>linvproc(a/t)*e_(j,k);
616 procedure linvproc(a);
617     if a member evalset
618     then << k:=0;
619         repeat k:=k+1 until a=evals(k);
620         linvs(k)*exp(a*t) >>
621     else sub(geneval=a,linvs(0))*exp(a*t)$

```

Second for the evolution on the invariant manifold.

```

622 operator ginv; linear ginv;
623 let ginv(e_(~j,~k)*exp(~a),exp)=>ginvproc(a/t)*e_(j,k);
624 procedure ginvproc(a);
625     if a member evalset
626     then << k:=0;
627         repeat k:=k+1 until a=evals(k);
628         ginvs(k) >>
629     else sub(geneval=a,ginvs(0))$

```

Copy and adjust the above for the projection. But first define the generic procedure. Perhaps use conjugate/negative of the frequency when applying to the general case of oscillations—but it might already have been accounted for??

```

630 procedure invproc(a,inv);

```

```

631   if a member evalset
632   then << k:=0;
633       repeat k:=k+1 until a=evals(k);
634       invs(k)*exp(a*t) >>
635   else sub(geneval=a, invs(0))*exp(a*t)$

```

Then define operators that we use to update the projection.

```

636 operator l1inv; linear l1inv;
637 operator l2inv; linear l2inv;
638 operator g1inv; linear g1inv;
639 operator g2inv; linear g2inv;
640 let { l1inv(e_(~j,~k)*exp(~a),exp)=>invproc(a/t,l1invs)*e_(j,k)
641      , l2inv(e_(~j,~k)*exp(~a),exp)=>invproc(a/t,l2invs)*e_(j,k)
642      , g1inv(e_(~j,~k)*exp(~a),exp)=>invproc(a/t,g1invs)*e_(j,k)
643      , g2inv(e_(~j,~k)*exp(~a),exp)=>invproc(a/t,g2invs)*e_(j,k)
644      };

```

This section writes to various files so the output to `cmsyso.txt` must be redone afterwards.

4 Initialise LaTeX output

This section writes to various files so the output to `cmsyso.txt` must be redone afterwards.

First define how various tokens get printed.

```

645 load_package rlf;
646 %deflist('(( ( !{\!b!i!g!() (!) !\!b!i!g!(!) (!P!I !\!p!i! )
647 %      (!p!i !\!p!i! ) (!E !e) (!I !i) (e !e) (i !i)), 'name)$
648 deflist('(( ( !\!l!P!a!r! ) (!) !\!r!P!a!r)
649      (!P!I !\!p!i! ) (!p!i !\!p!i! ) (!E !e) (!I !i)
650      (e !e) (i !i)), 'name)$

```

Force all fractions (coded in Reduce as `quotient`) to use `\frac` command so we can change how it appears.


```
651 put('quotient','laprifn','prinfrac');
```

Override the procedure that prints annoying messages about multicharacter symbols. It ends the output of one expression. This is just a copy from `rlfi.red` with the appropriate if-statement deleted. While interfering, hardcode that the mathematics is in inline mode.

```
652 %write "Ignore immediately following messages";
653 symbolic procedure prinlaend;
654 <<terpri();
655   prin2t "\\)\par";
656   if !*verbatim then
657     <<prin2t "\\begin{verbatim}";
658     prin2t "REDUCE Input:">>;
659   ncharspr!*::=0;
660   if ofl!* then linelength(car linel!*)
661     else laline!*::=cdr linel!*;
662   nochar!*::=append(nochar!*,nochar1!*);
663   nochar1!*::=nil >>$
664   %
```

Similarly, hardcode at the beginning of expression output that the mathematics is in inline mode.

```
665 symbolic procedure prinlabegin;
666 % Initializes the output
667 <<if !*verbatim then
668   <<terpri();
669   prin2t "\\end{verbatim}">>;
670   linel!*::=linelength nil . laline!*;
671   if ofl!* then linelength(laline!* + 2)
672     else laline!*::=car linel!* - 2;
673   prin2 "\\(" >>$
```

Override the procedure that outputs the \LaTeX preamble upon the command `on latex`. Presumably modified from that in `rlfi.red`. Use it to write a decent header that we use for one master file.

In the following, not clear that we should simply omit parentheses with the `exp` function. Could do something cleverer with `\lPar` and `\rPar` such as have a counter and cycle through the alternatives depending upon the counter.

```

674 symbolic procedure latexon;
675 <<!*!*a2sfn:='texaeval;
676   !*raise:=nil;
677   prin2t "\documentclass[11pt,a5paper]{article}";
678   prin2t "\usepackage[a5paper,margin=13mm]{geometry}";
679   prin2t "\usepackage{parskip,time} \raggedright";
680   prin2t "\def\lPar{\mathchoice{\big()}{\big()}{()}{()}}";
681   prin2t "\def\rPar{\mathchoice{\big)}{\big)}{)}{)}}";
682   prin2t "\let\FRaC\frac";
683   prin2t "\renewcommand{\frac}[2]{\mathchoice%";
684   prin2t "    {\FRaC{#1}{#2}}{\FRaC{#1}{#2}}{#1/#2}{#1/#2}}";
685   prin2t "\def\exp{\,e}";
686   prin2t "\def\eps{\varepsilon}";
687   prin2t "\title{Invariant manifold of your dynamical system}";
688   prin2t "\author{A. J. Roberts, University of Adelaide\\}";
689   prin2t "\texttt{http://www.maths.adelaide.edu.au/anthony.roberts}";
690   prin2t "\date{\now, \today}";
691   prin2t "\begin{document}";
692   prin2t "\maketitle";
693   prin2t "Throughout and generally: the lowest order, most";
694   prin2t "important, terms are near the end of each expression.";
695   prin2t "\input{centreManSys}";
696   if !*verbatim then
697       <<prin2t "\begin{verbatim}";
698       prin2t "REDUCE Input:>>";
699   put('tex','rtypefn',(lambda(x) 'tex)) >>$

```

Set the default output to be inline mathematics.

```
700 mathstyle math;
```

Define the Greek alphabet with `small` as well.

```
701 defid small,name="\eps";%varepsilon;
702 defid alpha,name=alpha;
703 defid beta,name=beta;
704 defid gamma,name=gamma;
705 defid delta,name=delta;
706 defid epsilon,name=epsilon;
707 defid varepsilon,name=varepsilon;
708 defid zeta,name=zeta;
709 defid eta,name=eta;
710 defid theta,name=theta;
711 defid vartheta,name=vartheta;
712 defid iota,name=iota;
713 defid kappa,name=kappa;
714 defid lambda,name=lambda;
715 defid mu,name=mu;
716 defid nu,name=nu;
717 defid xi,name=xi;
718 defid pi,name=pi;
719 defid varpi,name=varpi;
720 defid rho,name=rho;
721 defid varrho,name=varrho;
722 defid sigma,name=sigma;
723 defid varsigma,name=varsigma;
724 defid tau,name=tau;
725 defid upsilon,name=upsilon;
726 defid phi,name=phi;
727 defid varphi,name=varphi;
728 defid chi,name=chi;
729 defid psi,name=psi;
730 defid omega,name=omega;
731 defid Gamma,name=Gamma;
732 defid Delta,name=Delta;
733 defid Theta,name=Theta;
734 defid Lambda,name=Lambda;
735 defid Xi,name=Xi;
```

```

736 defid Pi,name=Pi;
737 defid Sigma,name=Sigma;
738 defid Upsilon,name=Upsilon;
739 defid Phi,name=Phi;
740 defid Psi,name=Psi;
741 defid Omega,name=Omega;

742 defindex e_(down,down);
743 defid e_,name="e";
744 defindex d_(arg,down,down);
745 defid d_,name="D";
746 defindex u(down);
747 %defid u1,name="u\sb1";
748 %defid u2,name="u\sb2";
749 %defid u3,name="u\sb3";
750 %defid u4,name="u\sb4";
751 %defid u5,name="u\sb5";
752 %defid u6,name="u\sb6";
753 %defid u7,name="u\sb7";
754 %defid u8,name="u\sb8";
755 %defid u9,name="u\sb9";
756 defindex s(down);
757 defindex exp(up);
758 defid exp,name="e"; %does not work??

```

Can we write the system? Not in matrices apparently. So define a dummy array `tmp` that we use to get the correct symbol typeset.

```

759 array tmp(n),tmps(m),tmpz(m);
760 defindex tmp(down);
761 defindex tmps(down);
762 defindex tmpz(down);
763 defid tmp,name="\dot u";
764 defid tmps,name="\vec e";
765 defid tmpz,name="\vec z";
766 rhs_:=rhsfn_$

```

```

767 for k:=1:m do tmps(k):={for j:=1:n collect ee_(j,k),exp(evl_(k)*
768 for k:=1:m do tmpz(k):={for j:=1:n collect zz_(j,k),exp(evl_(k)*

```

We have to be shifty here because `rlfi` does not work inside a loop: so write the commands to a file, and then input the file. The output line length of each ‘write’ statement must be short enough as otherwise Reduce puts in a line break.

```

769 out "scratchfile.red";
770 write "write ""\
771 \paragraph{The specified dynamical system}
772 \("";";
773 for j:=1:n do write "tmp(",j,"):=coeffn(rhs_,e_(",j,",1),1);";
774 write "write ""\
775 \paragraph{"",natureMan_,"
776 subspace basis vectors}",""
777 \("";";
778 for j:=1:m do write "tmps(",j,"):=tmps(",j,")";";
779 for j:=1:m do write "tmpz(",j,"):=tmpz(",j,")";";
780 write "end;";
781 shut "scratchfile.red";

```

Now print the dynamical system to the LaTeX sub-file.

```

782 on latex$
783 out "centreManSys.tex"$
784 in "scratchfile.red"$
785 shut "centreManSys.tex"$
786 off latex$

```

Finish the input.

```

787 end;
788 in_tex "latexinit2.tex"$

```

5 Linear approximation to the invariant manifold

But first, and if for the web, open the output file and write out the possibly adjusted nonlinear right-hand side function. According to the manual, this will append to the earlier output to the file.

```
789 %if thecase=myweb then out "cmsyso.txt"$
790 write "Analyse ODE/DDE system du/dt = ",ff_;
```

Parametrise the invariant manifold in terms of these amplitudes.

```
791 operator s; depend s,t;
792 let df(s(~j),t)=>coeffn(gg_,e_(j,1),1);
```

Invoke the following procedure to substitute whatever the current approximation is into (nonlinear) expressions. ??

```
793 procedure manifold_;
794     for j:=1:n collect u(j)=coeffn(uu_,e_(j,1),1)$
```

The linear approximation to the invariant manifold must be the following corresponding to the frequencies down the diagonal (even if zero). The amplitudes s_j are slowly evolving as they are either slow modes, or the complex amplitudes of oscillating modes.

```
795 uu_:=for j:=1:m sum s(j)*exp(ev1_(j)*t)
796   *(for k:=1:n sum e_(k,1)*ee_(k,j))$
797 gg_:=0$
```

For some temporary trace printing??

```
798 procedure matify(a,m,n)$
799     begin matrix z(m,n);
800     for i:=1:m do for j:=1:n do z(i,j):=coeffn(a,e_(i,j),1);
801     return (z where {exp(0)=>1,small=>s});
802     end$
```

For the isochron may need to do something different with frequencies, but this should work as the inner product is complex conjugate transpose. The

`pp_` matrix is proposed to place the projection residuals in the range of the isochron.

```
803 zs_:=for j:=1:m sum exp(evl_(j)*t)
804   *(for k:=1:n sum e_(k,j)*zz_(k,j))$
805 pp_:=0$
```

6 Iteratively construct the invariant manifold

But first establish the Taylor series in any delay factors of slow amplitudes.

```
806 let d_(s(~k),t,~dt)=>s(k)+(for n:=1:toosmall sum
807   (-dt)^n*df(s(k),t,n)/factorial(n));
```

Truncate expansions to specified order of error, and start the iteration.

```
808 for j:=toosmall:toosmall do let small^j=>0;
809 write "Start iterative construction of ",natureMan_," manifold";
810 for iter:=1:maxiter_ do begin
811 if trace_ then write "
812 ITERATION = ",iter,"
813 -----";
```

Compute residual vector (matrix) of the dynamical system [Roberts \(1997\)](#).

```
814 resde_:=df(uu_,t)+sub(manifold_(),rhsfn_);
815 if trace_ then write "resde_=",matify(resde_,n,1);
```

Get the local directions of the coordinate system on the curving manifold: store transpose as $m \times n$ matrix.

```
816 est_:=tpe_(for j:=1:m sum df(uu_,s(j))*e_(1,j),e_);
817 est_:=conj_(est_);
818 if trace_ then write "est_=",matify(est_,m,n);
```

Compute residual matrix for the isochron projection [Roberts \(1989, 2000\)](#). But only when the `evalset` is for slow manifolds: the reason is that there is no sensible concept of isochron for un/stable modes when in the presence of

centre modes.³ For example, consider the normal form system $\dot{X} = 0$ and $\dot{Y} = G(Y)Y$: it has solutions $Y(t) = Y_0 e^{G(X_0)t}$ and so for general G there are no curves $Y(X)$ which have the same rate of decay to the slow manifold; that is, there are no curves that ‘collapse together’.

```

819 if itisSlowMan_ then begin
820     jacadj_:=conj_(sub(manifold_(),rhsjact_));
821 if trace_ then write "jacadj_=",matify(jacadj_,n,n);
822     resd_:=df(zs_,t)+jacadj_*zs_+zs_*pp_;
823 if trace_ then write "resd_=",matify(resd_,n,m);

```

Compute residual of the normalisation of the projection.

```

824     resz_:=est_*zs_-eyem_*exp(0);
825 if trace_ then write "resz_=",matify(resz_,m,m);
826 end else resd_:=resz_:=0; % for when not slow manifold

```

Write lengths of residuals as a trace print (remember that the expression 0 has length one).

```

827 write lengthRes:=map(length(~a),{resde_,resd_,resz_});

```

Solve for updates all the hard work is already encoded in the operators.

```

828 uu_:=uu_+linv(resde_,exp);
829 gg_:=gg_+ginv(resde_,exp);
830 if trace_ then write "gg_=",matify(gg_,m,1);
831 if trace_ then write "uu_=",matify(uu_,n,1);

```

Now update the isochron projection, with normalisation.

```

832 if itisSlowMan_ then begin
833     zs_:=zs_+l1inv(resd_,exp)-l2inv(resz_,exp);
834     pp_:=pp_-g1inv(resd_,exp)+youshouldnotseethis*g2inv(resz_,exp);
835 if trace_ then write "zs_=",matify(zs_,n,m);

```

³Although there is a sensible concept of ‘isochron’ if there are no centre modes—justified by the Hartman–Grossman theorem which asserts topological equivalence to the local linearisation.


```

836 if trace_ then write "pp_=",matify(pp_,m,m);
837 end;

```

Terminate the loop once residuals are zero.

```

838 showtime;
839 if {resde_,resd_,resz_}={0,0,0} then write iter:=iter+10000;
840 end;

```

Only proceed to print if terminated successfully.

```

841 if {resde_,resd_,resz_}={0,0,0}
842   then write "SUCCESS: converged to an expansion"
843   else <<write "FAILED TO CONVERGE; I QUIT";
844       if thecase=myweb then <<shut "cmsyso.txt";
845       quit >> >>;
846 %write "Temporarily halt here";end;

```

7 Output text version of results

Once construction is finished, simplify `exp(0)`.

```

847 let exp(0)=>1;

```

Invoking switch `complex` improves some of the output of the complex numbers, but wrecks other parts of the output. Best left off.

Write text results.

```

848 write "The ",natureMan_," manifold is (to one order lower)";
849 for j:=1:n do write "u",j," = ",
850   coeffn(small*uu_,e_(j,1),1)/small;
851 write "The evolution of the real/complex amplitudes";
852 for j:=1:m do write "ds(",j,")/dt = ",
853   coeffn(gg_,e_(j,1),1);

```

Optionally write the projection vectors.

```

854 if itisSlowMan_ then begin
855   write "The normals to the isochrons at the slow manifold.
856 Use these vectors: to project initial conditions
857 onto the slow manifold; to project non-autonomous
858 forcing onto the slow evolution; to predict the
859 consequences of modifying the original system; in
860 uncertainty quantification to quantify effects on
861 the model of uncertainties in the original system.";
862   for j:=1:m do write "z",j," = ",
863     for i:=1:n collect coeffn(zs_,e_(i,j),1);
864 end;

```

Write text results numerically evaluated when expressions are long.

```

865 if length(gg_)>30 then begin
866 on rounded; print_precision 4;
867 write "Numerically, the ",natureMan_," manifold is (to one order
868 for j:=1:n do write "u",j," = ",
869   coeffn(small*uu_,e_(j,1),1)/small;
870 write "Numerically, the evolution of the real/complex amplitudes
871 for j:=1:m do write "ds(",j,")/dt = ",
872   coeffn(gg_,e_(j,1),1);
873 if itisSlowMan_ then begin
874   write "Numerically, normals to isochrons at slow manifold.";
875   for j:=1:m do write "z",j," = ",
876     for i:=1:n collect coeffn(zs_,e_(i,j),1);
877 end;
878 off rounded;
879 end;

880 if thecase=myweb then shut "cmsyso.txt"$

```

There is an as yet unresolved problem in the typesetting when the argument of **exp** (eigenvalue) is a rational number instead of integer ??: the numerator has an extra pair of parentheses which then makes the typesetting wrong; maybe we need a pre-L^AT_EX filter??

8 Output LaTeX version of results

Change the printing of temporary arrays.

```
881 array tmpzz(m,n);
882 defid tmp,name="u";
883 defid tmps,name="\dot s";
884 defid tmpz,name="\vec z";
885 defid tmpzz,name="z";
886 defindex tmpzz(down,down);
```

Gather complicated result

```
887 %for k:=1:m do tmpz(k):=for j:=1:n collect (1*coeffn(zs_,e_(j,k))
888 for k:=1:m do for j:=1:n do tmpzz(k,j):=(1*coeffn(zs_,e_(j,k),1))
```

Write to a file the commands needed to write the LaTeX expressions. Write the invariant manifold to one order lower than computed. The output line length of each ‘write’ statement must be short enough as otherwise Reduce puts in a line break—its counting is a bit mysterious!

```
889 out "scratchfile.red";
890 write "write ""\
891 \paragraph{The ",natureMan_,"
892 manifold}";
893 write "These give the location of the invariant manifold in
894 terms of parameters~\((s\sb j)\).
895 \("";";
896 for j:=1:n do write "tmp(",j,
897   "):=coeffn(small*uu_,e_(",j,",1),1)/small;";
```

Write the commands to write the ODEs on the centre manifold.

```
898 write "write ""\
899 \paragraph{"",natureMan_,"
900 manifold ODEs}";
901 write "The system evolves on the invariant manifold such
902 that the parameters evolve according to these ODEs.
```

```

903 \("";";
904 for j:=1:m do write "tmps(",j,"):=1*coeffn(gg_,e_(",j,",1),1);";

```

Optionally write the commands to write the projection vectors on the slow manifold.

```

905 if itisSlowMan_ then begin
906   write "write ""\)"
907   \paragraph{Normals to isochrons at the slow manifold}
908   Use these vectors: to project initial conditions
909   onto the slow manifold; to project non-autonomous
910   forcing onto the slow evolution; to predict the
911   consequences of modifying the original system; in
912   uncertainty quantification to quantify effects on
913   the model of uncertainties in the original system.
914   The normal vector \(\vec z\sb j:=(z\sb{j1},\ldots,z\sb{jn})\)
915   \("";";
916   for i:=1:m do for j:=1:n do
917     write "tmpzz(",i,",",j,"):=tmpzz(",i,",",j,")";
918 end;

```

Finish the scratchfile.

```

919 write "end;";
920 shut "scratchfile.red";

```

Execute the file with the required commands, with output to the main centre manifold LaTeX file.

```

921 out "centreMan.tex"$
922 on latex$
923 in "scratchfile.red"$
924 off latex$
925 shut "centreMan.tex"$

926 end;

927 in_tex "latexout2.tex"$

```

9 Fin

That's all folks.

```

928 write "Finished constructing ",natureMan_," manifold of ODE/DDE"
929 if thecase=myweb then begin
930 quit;
931 end;

932 %end;%loop over cases--not working
933 end;

```

References

- Erneux, T. (2009), *Applied Delay Differential equations*, Vol. 3 of *Surveys and Tutorials in the Applied Mathematical Sciences*, Springer.
- Ion, A. V. & Georgescu, R. M. (2013), ‘Bautin bifurcation in a delay differential equation modeling leukemia’, *Nonlinear Analysis, Theory, Methods & Applications* **82**, 142–157.
<http://arXiv.org/abs/1208.1707>
- Renson, L., Deliege, G. & Kerschen, G. (2012), Finite element computation of nonlinear normal modes of nonconservative systems, *in* ‘Proceedings of the ISMA 2012 conference’.
<http://hdl.handle.net/2268/129189>
- Roberts, A. J. (1989), ‘Appropriate initial conditions for asymptotic descriptions of the long term evolution of dynamical systems’, *J. Austral. Math. Soc. B* **31**, 48–75.
- Roberts, A. J. (1997), ‘Low-dimensional modelling of dynamics via computer algebra’, *Computer Phys. Comm.* **100**, 215–230. doi:[10.1016/S0010-4655\(96\)00162-2](https://doi.org/10.1016/S0010-4655(96)00162-2).

- Roberts, A. J. (2000), ‘Computer algebra derives correct initial conditions for low-dimensional dynamical models’, *Computer Phys. Comm.* **126**(3), 187–206.
- Stoleriu, I. (2012), Periodic orbits of a pair of coupled oscillators near resonance, Technical report, University of Iasi.
<http://www.math.uaic.ro/~ITN2012/files/talk/Stoleriu.pdf>