

Computer algebra coordinate-free construction of slow manifold in stochastic or non-autonomous multiscale differential equations

A. J. Roberts*

31 May 2023 – September 9, 2023

Abstract

Provides a coded procedure to construct a slow manifold of a wide class of systems of non-autonomous or stochastic differential equations (herein abbreviated by s/ODEs). The method is coordinate-free. The methodology is based upon earlier research (Coullet & Spiegel 1983, Cox & Roberts 1991, Roberts 1997, Chao & Roberts 1996, Roberts 2008). Interpret all s/ODEs in the Stratonovich sense so the analysis applies to deterministic differential equations, both non-autonomous and autonomous. Cater for deterministic autonomous systems by simply omitting the ‘noise’. For generality, this coded procedure caters for unstable modes, and for differential equation systems with a rational right-hand side.

Contents

1	Introduction	2
1.1	A simple example: <code>examplesystem()</code>	4

*School of Mathematical Sciences, University of Adelaide, South Australia 5005, AUSTRALIA. <https://profajroberts.github.io/>

1	<i>Introduction</i>	2
2	Header of the procedure	6
2.1	Preamble to the procedure	7
2.2	Extract the differential equations	8
3	Setup LaTeX output using rlfi	10
4	Delayed write of text info	12
5	Represent the noise	13
6	Operators to solve noisy homological equation	15
7	Initialise approximate transform	17
8	Iterative updates	18
8.1	Fast modes	18
8.2	Slow modes	18
9	Output results	19
9.1	Plain text version	19
9.2	L ^A T _E X version	20
10	Fin	22
11	Override some rlfi procedures	22

1 Introduction

Inputs Write your s/ODE system in terms of variables $u_j(t)$. You must denote the slow modes by $\mathbf{u}(1)$ through to $\mathbf{x}(\mathbf{nu})$, or the aliases $\mathbf{u}1, \dots$. Each non-autonomous factor must be denoted by $\mathbf{w}(\cdot)$ where the dot denotes almost any label you care to choose: simple numbers such as $\mathbf{w}(1)$ and/or $\mathbf{w}(2)$ are the usual choices (no aliases); but other labels may be used. Often $\mathbf{w}(\cdot)$ is a Stratonovich white noise, a derivative of a Stratonovich Wiener process. Analyse deterministic, autonomous, systems by omitting any noise term $\mathbf{w}(\cdot)$

in the differential equations. Then, as in the example of the next [Section 1.1](#), the input parameters to the procedure are the following:

- **dudt**, a comma separated list within `mat(...)` of the right-hand sides of the S/ODEs for the variables $u_j(t)$;
- **slowbasis**, a comma separated list within `mat(...)` of the (desired) basis vectors for the slow subspace, corresponding to eigenvalues (near) zero;
- **fasteval**, a comma separated list within `mat(...)` of the fast eigenvalues of the system—best if real numbers (not complex), but maybe OK otherwise, especially by specifying `sign(??)=>-1` or whatever;
- **fasteval**, a comma separated list within `mat(...)` of the corresponding (generalised) eigenvectors that span the fast subspace;
- **toosmall**, an integer giving the desired order of error in the asymptotic approximation that is constructed. The procedure embeds the specified system in a family of systems parametrised by ε , and constructs a slow manifold, and evolution thereon, of the embedding system to the asymptotic error $\mathcal{O}(\varepsilon^{\text{toosmall}})$ (as $\varepsilon \rightarrow 0$). Often the introduced artificial ε has a useful physical meaning, but strictly you should evaluate the output at $\varepsilon = 1$ to recover results for the specified system, and then reinterpret the results, and errors, in terms of your system’s actual ‘small’ parameters.

The code *also* truncates to errors $\mathcal{O}(\sigma^3)$ where σ characterises the magnitude of the non-autonomous/stochastic effects $\mathbf{w}(j)$.

The right-hand side expressions for the time-derivatives must be multinomial in variables u_i and w_i . To cater for rational function right-hand sides, one also may include some \dot{u}_i factors in the right-hand sides.

Outputs This procedure reports the embedded system it actually analyses, the number of iterations taken, the constructed time dependent coordinate transform (the original variables \vec{u} as a function of the new variables \vec{s}), and the corresponding evolution in the new variables in terms of S/ODEs for \vec{s} .

- A plain report to the Terminal window in which *Reduce* is executing.
- A L^AT_EX source report written to the file `slowReport.tex` (with `slowReportHdr.tex` and `slowReportSys.tex`). Generate a pdf version by executing `pdflatex slowReport`.
- Global array `u_(i)` gives the parametrisation of the slow manifold $u_i(\vec{s}, \sigma, \varepsilon)$.
- Global array `ff(i)` gives the corresponding evolution $\dot{U}_i = \mathbf{ff}(i)$ as functions of $(\vec{s}, \sigma, \varepsilon)$.

One may change the appearance of the output somewhat. For example, in order to group terms in output expressions, execute a `factor` command before executing the procedure `slownonauto(...)`.

Background The theoretical support for the results of the analysis of this procedure is centre/stable/unstable manifold theory (e.g., Carr 1981, Knobloch & Aulbach 1982, Haragus & Iooss 2011, Roberts 2015), and an embryonic backwards theory (Roberts 2022). This particular procedure is developed from that for human-efficient computer algebra (Roberts 1997), and extended to stochastic/non-autonomous systems (Chao & Roberts 1996, Roberts 2008).

We use the computer algebra package *Reduce* [<http://reduce-algebra.com/>] because it is both free and perhaps the fastest general purpose computer algebra system (Fateman 2003, e.g.).

1.1 A simple example: `examplesystem()`

Execute this example by loading the code with command `in_tex "slowNonauto.tex"$` and then invoking the command `examplesystem()`; The code here constructs the non-autonomous slow manifold for the system

$$\dot{u}_1 = -u_1 + u_2 - u_1^2 + \sigma w_1(t), \quad \dot{u}_2 = +u_1 - u_2 + u_2^2 - \sigma w_1(t). \quad (1)$$

The lowercase $w_j(t)$, called a *noise* within this document, may denote the formal derivative dW_j/dt of a Stratonovich Wiener process $W_j(t, \omega)$. Alternatively, $w_j(t)$ represent an arbitrary deterministic time-dependent forcing, or some control, or some ‘coloured’ random process, or some other extrinsic input to the system. Parameter σ controls the strength of the so-called noise.

Use variables `u1,u2` to denote u_1, u_2 , and use `w(j)` to denote the non-autonomous effect/noise w_j . Hence the system (1) is analysed for ‘small’ (u_1, u_2, σ) by executing the following defined procedure:

```

1 procedure examplesystem(); begin
2   factor small,sigma;
3   slownonauto(% first give RHS of S/ODEs
4   mat(( -u1+u2-u1^2+w(1), u1-u2+u2^2-w(1) )),
5   mat((1,1)), % 2nd list basis vectors for slow subspace
6   mat((-2)), % 3rd list fast eigenvalues
7   mat((-1,1)),% 4th list corresponding fast eigenvectors
8   3 )$          % last give order of error in small eps
9   end;
```

The procedure `slownonauto` automatically multiplies the noise/non-autonomous factors by a parameter `sigma` so there is no need include the parameter σ in the specification of the problem—it will be done for you.

Further, the procedure uses the parameter `small`, often denoted by ε , to control truncation in nonlinearity. The fourth parameter in the above specifies to construct the manifold to errors $\mathcal{O}(\varepsilon^3)$.

Consequently, the procedure embeds the given system as the $\varepsilon = 1$ version of the following system that it actually analyses:

$$\dot{u}_1 = -u_1 + u_2 - \varepsilon u_1^2 + \sigma w_1(t), \quad \dot{u}_2 = +u_1 - u_2 + \varepsilon u_2^2 - \sigma w_1(t). \quad (2)$$

The construction is actually based about the subspace of equilibria $u_1 - u_2 = \varepsilon = \sigma = 0$. The constructed time-dependent slow manifold and corresponding S/ODEs are the following.

Time-dependent slow manifold (for simplicity reported to one order lower in both ε and σ than actually constructed)

$$u_1 = s_1 - 1/2\varepsilon s_1^2 + \sigma e^{-2t} \star w_1 + \sigma \varepsilon e^{-2t} \star w_1 s_1 + O(\varepsilon^2, \sigma^2)$$

$$u_2 = s_1 + 1/2\varepsilon s_1^2 - \sigma e^{-2t} \star w_1 + \sigma \varepsilon e^{-2t} \star w_1 s_1 + O(\varepsilon^2, \sigma^2)$$

On this slow manifold the evolution is governed by the s/ODE

$$\dot{s}_1 = +\varepsilon^2 s_1^3 - \sigma \varepsilon s_1 w_1 + 1/2\sigma^2 \varepsilon^2 e^{-2t} \star w_1 s_1 w_1 + O(\varepsilon^3, \sigma^3)$$

Since this construction is based upon the subspace of equilibria $u_1 - u_2 = \varepsilon = \sigma = 0$, these constructed expressions for the slow manifold are global in parameter s_1 and asymptotic as $\varepsilon, \sigma \rightarrow 0$. However, evaluation at $\varepsilon = 1$ typically means that the slow manifold expressions should be reinterpreted as asymptotic as $s_1, \sigma \rightarrow 0$.

2 Header of the procedure

For testing only, the following procedure is for reporting some errors during development

```

10 procedure testy(a,b,ind); begin
11   write "TESTY";
12   write a:=a;
13   write b:=b;
14   write indi:=ind;
15   rederr "procedure testy caught some failure";
16   end$

```

Need a couple of things established before defining the procedure: the `rlfi` package for a nicer version of the output via `LATEX`; and operator names for the variables of the s/ODEs.

```

17 load_package rlfi;
18 operator u;
19 operator w;

```

Cater for rational function s/ODEs by allowing time dependence in `u` at specification. For rational function s/ODEs, users must multiply each s/ODE

by a common denominator, and put on the right-hand side the nonlinear terms involving the time derivative.

```
20 depend u,t;
```

Now define the procedure as an operator so we can define procedures internally, and may be flexible with its arguments.

```
21 operator slownonauto;
22 for all dudt, slowbasis, fasteval, fastevec, toosmall let
23     slownonauto(dudt, slowbasis, fasteval, fastevec, toosmall)
24     = begin
```

2.1 Preamble to the procedure

Operators and arrays are always global, but we can make variables and matrices local, except for matrices that need to be declared `matrix`. So, move to implement all arrays and operators to have underscores, and almost all scalars and most matrices to be declared local here (for some reason `x/y/zrhs` must be global).

```
25 scalar maxiter, trace, nu, offdiag, jac, ok, res,
26 res0, res1, lengthresu;
```

Parameter `maxiter` is the maximum number of allowed iterations; this default may be changed.

```
27 maxiter:=9$
```

For optional trace printing of test cases: comment out second line when not needed.

```
28 trace:=0$
29 %trace:=1; maxiter:=5;
```

Write an intro message.

```
30 write "Construct non-autonomous slow manifold (version 9 Sep 2023)"$
```

Transpose the defining matrices so that vectors and lists are columns.

```
31 urhs_ := tp dudt;
```

```

32 if trace then write urhs_:=urhs_;
33 sbase_ := tp slowbasis;
34 if trace then write sbase_:=sbase_;
35 fevec_ := tp fastevec;
36 if trace then write fevec_:=fevec_;
37 feval_ := tp fasteval;
38 if trace then write feval_:=feval_;

```

In the printed expressions, may factor `small` (ε) and `sigma` (σ).

```

39 if trace then factor small,sigma;

```

The code cannot yet handle any cubic or higher order in noise amplitude `sigma`.

```

40 let sigma^3=>0;

```

The `rationalize` switch may make code much faster with complex numbers. The switch `gcd` seems to wreck convergence, so leave it off.

```

41 on div; off allfac; on revpri;
42 on rationalize; %on ezgcd;

```

2.2 Extract the differential equations

The number of differential equations is the number of rows in the matrix `dudt`.

```

43 write "no. of DEs ",nu:=part(length(urhs_),1);
44 write "no. of slow variables ",ns:=part(length(sbase_),2);
45 write "no. of fast variables ",nf:=part(length(fevec_),2);
46 if nu=ns+nf then else rederr "Abort: incompatible dimensions";
47 if nf=part(length(feval_),1) then
48 else rederr "different number of eigenvalues and eigenvectors";
49 if nu=part(length(sbase_),1) then
50 else rederr "dimension of slow basis vectors do not match the DEs";
51 if nu=part(length(fevec_),1) then
52 else rederr "dimension of fast eigenvectors do not match the DEs";
53 %for i:=1:nf do if numberp(feval_(i,1)) then
54 %     else rederr "Need all fast eigenvalues to be simple numbers";

```


Define aliases `ui` map to `u(i)`, except you *must* use the `u(i)`-form *within* any `df(,t)` on the right-hand side.

```
55 for i:=1:nu do set(mkid(u,i),u(i));
56 if trace then write "finished aliases";
```

Define a zero matrix and zero vectors.

```
57 matrix o_(nu,nu),ov_(nu,1),ov_s(ns,1);
```

Check on the basis matrix: putting the slow vectors in the first n_s columns; and the fast vectors into the remaining n_f columns.

```
58 pp_=o_;
59 for i:=1:nu do for j:=1:ns do pp_(i,j):=sbase_(i,j);
60 for i:=1:nu do for j:=1:nf do pp_(i,j+ns):=fevec_(i,j);
61 pp_inv:=1/pp_;
62 if trace then write "pp_inv := ",pp_inv;
```

Adjust system to suit the linear operator implicitly defined by the basis vectors and eigenvalues. Multiply all other terms, linear or nonlinear, by `small`.

```
63 ll_=o_;
64 for i:=1:nf do ll_(i+ns,i+ns):=feval_(i,1);
65 ll_:=pp_*ll_*pp_inv;
66 if trace then write "ll_ :=",ll_;
67 u_=ov_;
68 for i:=1:nu do u_(i,1):=u(i);
69 urhs_:=ll_*u_+small*(urhs_-ll_*u_);
```

Adjust the non-autonomous terms. Remove the `small` multiplication of noise terms, and instead multiply by `sigma` to empower independent control of the truncation in noise amplitude.

```
70 urhs_=(urhs_ where w(~j)=>sigma*w(j,xyzy)/small)$
71 urhs_=(urhs_ where w(~j,xyzy)=>w(j))$
```

Section 4 writes the resulting differential equations for information.

```
72 if trace then for i:=1:nu do
73     write "du(",i,")/dt = ",urhs_(i,1);
```

3 Setup LaTeX output using rlf

Use inline math environment so that long lines, the norm, get line breaks. The command `\raggedright` in the \LaTeX preamble appears the best option for the line breaking, but `\sloppy` would also work reasonably.

```
74 mathstyle math;
```

Define names for \LaTeX formatting Define some names I use, so that rlf translates them to Greek characters in the \LaTeX .

```
75 defid small,name="\eps";%varepsilon;
76 defid alpha,name=alpha;
77 defid beta,name=beta;
78 defid gamma,name=gamma;
79 defid delta,name=delta;
80 defid epsilon,name=epsilon;
81 defid varepsilon,name=varepsilon;
82 defid zeta,name=zeta;
83 defid eta,name=eta;
84 defid theta,name=theta;
85 defid vartheta,name=vartheta;
86 defid iota,name=iota;
87 defid kappa,name=kappa;
88 defid lambda,name=lambda;
89 defid mu,name=mu;
90 defid nu,name=nu;
91 defid xi,name=xi;
92 defid pi,name=pi;
93 defid varpi,name=varpi;
94 defid rho,name=rho;
95 defid varrho,name=varrho;
96 defid sigma,name=sigma;
97 defid varsigma,name=varsigma;
98 defid tau,name=tau;
99 defid upsilon,name=upsilon;
100 defid phi,name=phi;
101 defid varphi,name=varphi;
102 defid chi,name=chi;
```

```

103 defid psi,name=psi;
104 defid omega,name=omega;
105 defid Gamma,name=Gamma;
106 defid Delta,name=Delta;
107 defid Theta,name=Theta;
108 defid Lambda,name=Lambda;
109 defid Xi,name=Xi;
110 defid Pi,name=Pi;
111 defid Sigma,name=Sigma;
112 defid Upsilon,name=Upsilon;
113 defid Phi,name=Phi;
114 defid Psi,name=Psi;
115 defid Omega,name=Omega;

```

For the variables names I use, as operators, define how they appear in the \LaTeX , and also define that their arguments appear as subscripts.

```

116 defindex w(down);
117 defindex u(down);
118 defindex s(down);
119 defindex ff(down);

```

First use these for the specified dynamical system, later use them for the normal form equations.

```

120 defid ff,name="\dot u";

```

The ‘Ornstein–Uhlenbeck’ operator is to translate into a \LaTeX command, see the preamble, that typesets the convolution in a reasonable manner. The definition of the \LaTeX command is a bit dodgy as convolutions of convolutions are not printed in the correct order; however, convolutions commute so it does not matter.

```

121 defid ou,name="\ou";
122 defindex ou(arg,arg,arg);

```

Write the \LaTeX dynamical system Because of the way *rlfi* works, to get good quality output to the \LaTeX document, I need to write the algebraic expressions to a file, then read them back in again. While being read back in,

I send the output to the \LaTeX file. In this convoluted way I avoid extraneous output lines polluting the \LaTeX .

Temporarily use these arrays for the right-hand sides of the dynamical system.

```
123 clear ff;
124 array ff(nu);
```

Write expressions to the file `scratchfile.red` for later reading. Prepend the expressions with an instruction to write a heading, and surround the heading with anti-math mode to cancel the math environment that `rlfi` puts in.

```
125 out "scratchfile.red"$
126 write "off echo;"$ % do not understand why needed in 2021??
127 write "write ""\"
128 \paragraph{Specified dynamical system}
129 \("$";
130 for i:=1:nu do write "ff(",i,"):=urhs_(",i,",1);";
131 write "end;";
132 shut "scratchfile.red";
```

Then switch on \LaTeX output before writing to file as this \LaTeX file is to be input from the main \LaTeX file and hence does not need a header. The header here gets sent to the ‘terminal’ instead. Then write to `slowReportSys.tex` the expressions we stored in `scratchfile.red` as nice \LaTeX .

```
133 write "Ignore the following five lines of LaTeX"$
134 on latex$
135 out "slowReportSys.tex"$
136 in "scratchfile.red"$
137 shut "slowReportSys.tex"$
138 off latex$
```

4 Delayed write of text info

Because it is messy to interleave \LaTeX and plain output, I delay writing anything much in plain text until here.

Write the delayed warning message about equations.

```

139 write "
140 ===== Inspect these printed equations. Results
141 are rubbish if not appropriate for little small."$

```

Write the plain text versions of the dynamical system.

```

142 write "no. of variables ",nu:=nu;
143 for i:=1:nu do write "du(",i,")/dt = ",urhs_(i,1);

```

5 Represent the noise

The ‘noises’ w depend upon time. But we find it useful to discriminate upon the notionally fast time fluctuations of a noise/non-autonomous process, and the notionally ordinary time variations of the dynamic variables u_i . Thus introduce a notionally fast time variable tt , which depends upon the ordinary time t . Equivalently, view tt , a sort of ‘partial t ’, as representing variations in time independent of those in the variables u_i .

```

144 depend w,tt;
145 depend tt,t,ttf;

```

In the construction, convolutions of the noise arise, both backwards over history and also forwards in time to anticipate the noise ([Roberts 2008, 2022](#)). For any non-zero parameter μ , define the ‘Ornstein–Uhlenbeck’ convolution

$$e^{\mu t} \star \phi = \begin{cases} \int_{-\infty}^t \exp[\mu(t - \tau)] \phi(\tau) d\tau, & \mu < 0, \\ \int_t^{+\infty} \exp[\mu(t - \tau)] \phi(\tau) d\tau, & \mu > 0, \end{cases} \quad (3)$$

so that the convolution is always with a bounded exponential. Five useful

properties of this convolution are

$$e^{\mu t} \star 1 = \frac{1}{|\mu|}, \quad (4)$$

$$\frac{d}{dt} e^{\mu t} \star \phi = -\operatorname{sgn} \mu \phi + \mu e^{\mu t} \star \phi, \quad (5)$$

$$E[e^{\mu t} \star \phi] = e^{\mu t} \star E[\phi], \quad (6)$$

$$E[(e^{\mu t} \star \phi)^2] = \frac{1}{2|\mu|}, \quad (7)$$

$$e^{\mu t} \star e^{\nu t} \star = \begin{cases} \frac{1}{|\mu - \nu|} [e^{\mu t} \star + e^{\nu t} \star], & \mu \nu < 0, \\ \frac{-\operatorname{sgn} \mu}{\mu - \nu} [e^{\mu t} \star - e^{\nu t} \star], & \mu \nu > 0 \text{ \& } \mu \neq \nu. \end{cases} \quad (8)$$

Also remember that although with $\mu < 0$ the convolution $e^{\mu t} \star$ integrates over the past, with $\mu > 0$ the convolution $e^{\mu t} \star$ integrates into the future—both over a time scale of order $1/|\mu|$.

The operator `ou(f,tt,mu)` represents the convolution $e^{\mu t} \star f$ as defined by (3): called `ou` because it is an Ornstein–Uhlenbeck process when f is a stochastic white noise. The operator `ou` is ‘linear’ over fast time `tt` as the convolution only arises from solving PDEs in the operator $\partial_t - \mu$. Code its derivative (5) and its action upon autonomous terms (4):

```
146 clear ou; operator ou; linear ou;
147 let { df(ou(~f,tt,~mu),t)=>-sign(mu)*f+mu*ou(f,tt,mu)
148      , ou(1,tt,~mu)=>1/abs(mu) when mu neq 0
```

Also code the transform (8) that successive convolutions at different rates may be transformed into several single convolutions.

```
149      , ou(ou(~r,tt,~nu),tt,~mu) =>
150      (ou(r,tt,mu)+ou(r,tt,nu))/abs(mu-nu) when mu neq 0
151      when (sign(mu)+sign(nu)=0)and(sign(mu)neq 0)
152      , ou(ou(~r,tt,~nu),tt,~mu) =>
153      -sign(mu)*(ou(r,tt,mu)-ou(r,tt,nu))/(mu-nu)
154      when (sign(mu)=sign(nu))and(mu neq nu)
155      };
```

The above properties are *critical*: they must be correct for the results to be correct. Currently, they are only coded for *real* rates μ, ν .

Second, identify the resonant parts, some of which must go into the evolution $\mathbf{gg}(i)$, and some into the transform. It depends upon the exponent of \mathbf{yz} compared to the decay rate of this mode, here \mathbf{r} .

```
156 clear reso_; operator reso_; linear reso_;
157 let { reso_(~a,yz,~r)=>1 when df(a,yz)*yz=r*a
158      , reso_(~a,yz,~r)=>0 when df(a,yz)*yz neq r*a
159      };
```

Lastly, the remaining terms get convolved at the appropriate rate to solve their respective homological equation by the operator $\mathbf{zres_}$.

```
160 depend yz,ttf;
161 clear zres_; operator zres_; linear zres_;
162 let zres_(~a,ttf,~r)=>ou(sign(df(a,yz)*yz/a-r)
163      *sub(yz=1,a),tt,df(a,yz)*yz/a-r);
```

6 Operators to solve noisy homological equation

When solving homological equations of the form $F + \xi_t = \text{Res}$ (the resonant case $\mu = 0$), we separate the terms in the right-hand side Res into those that are integrable in fast time, and hence modify the coordinate transform by changing ξ , and those that are not, and hence must remain in the evolution by changing F . the operator $\mathbf{zint_}$ extracts those parts of a term that we know are integrable; the operator $\mathbf{znon_}$ extracts those parts which are not knowably bounded integrable. With more research, more types of terms may be found to be integrable; hence what is extracted by $\mathbf{zint_}$ and what is left by $\mathbf{zint_}$ may change with more research, or in different scenarios. These transforms are not critical: changing the transforms may change intermediate computations, but as long as the iteration converges, the computer algebra results will be algebraically correct.

```
164 clear zint_; operator zint_; linear zint_;
165 clear znon_; operator znon_; linear znon_;
```

First, avoid obvious secularity.

```
166 zrules:= { zint_(w(~i),tt)=>0,      znon_(w(~i),tt)=>w(i)
167            , zint_(1,tt)=>0,          znon_(1,tt)=>1
```

```
168      , zint_(w(~i)*~r,tt)=>0, znon_(w(~i)*~r,tt)=>w(i)*r
```

Second, by (5) a convolution may be split into an integrable part, and a part in its argument which in turn may be integrable or not. Had to change μ, ν to mur, nur in order to avoid inscrutable error.

```
169      , zint_(ou(~r,tt,~mur),tt)
170      =>ou(r,tt,mur)/mur+zint_(r,tt)/abs(mur)
171      , znon_(ou(~r,tt,~mur),tt)=>znon_(r,tt)/abs(mur)
```

Third, squares of convolutions may be integrated by parts to an integrable term and a part that may have integrable or non-integrable parts.

```
172      , zint_(ou(~r,tt,~mur)^2,tt)=>ou(~r,tt,~mur)^2/(2*mur)
173      +zint_(r*ou(r,tt,mur),tt)/abs(mur)
174      , znon_(ou(~r,tt,~mur)^2,tt)=>znon_(r*ou(r,tt,mur),tt)/abs(mur)
```

Fourth, different products of convolutions may be similarly separated using integration by parts. Why is the 2nd one here not acted upon, unless double tilde is used??

```
175      , zint_(ou(~r,tt,~~mur)*ou(~s,tt,~~nur),tt)
176      =>ou(r,tt,mur)*ou(s,tt,nur)/(mur+nur)
177      +zint_(sign(mur)*r*ou(s,tt,nur)+sign(nur)*s*ou(r,tt,mur),tt)
178      /(mur+nur) when mur+nur neq 0
179      , znon_(ou(~r,tt,~~mur)*ou(~s,tt,~~nur),tt)=>
180      +znon_(sign(mur)*r*ou(s,tt,nur)+sign(nur)*s*ou(r,tt,mur),tt)
181      /(mur+nur) when mur+nur neq 0
```

However, a zero divisor arises when $\mu + \nu = 0$ in the above. Here code rules to cater for such terms by increasing the depth of convolutions over past history.

```
182      , zint_(ou(~r,tt,~~mur)*ou(~s,tt,~~nur),tt)=>
183      ou(ou(r,tt,-nur),tt,-nur)*ou(s,tt,nur)
184      +zint_(ou(ou(r,tt,-nur),tt,-nur)*s,tt)
185      when (mur+nur=0)and(sign(nur)>0)
186      , znon_(ou(~r,tt,~~mur)*ou(~s,tt,~~nur),tt)=>
187      znon_(ou(ou(r,tt,-nur),tt,-nur)*s,tt)
188      when (mur+nur=0)and(sign(nur)>0)
```

The above handles quadratic products of convolutions. Presumably, if we

seek cubic noise effects then we may need cubic products of convolutions. However, I do not proceed so far and hence terminate the separation rules.

```
189     }; % let/zrules
```

7 Initialise approximate transform

Truncate asymptotic approximation of the coordinate transform depending upon the parameter `toosmall`. Use the ‘instant evaluation’ property of a loop index to define the truncation so that Reduce omits small terms on the fly.

```
190 for j:=toosmall:toosmall do let small^j=>0;
```

Variables `u` are operators in the specification of the equations. We now want them to map to the approximation to the coordinate transform, so point them to arrays storing the normal form expressions. Need to clear the mapping to the array before exiting.

```
191 clear u_;
192 let u(~j)=>u_(j,1);
```

Express the normal form in terms of new evolving variables s_i , denoted by operators `s(i)`. The expressions for the normal form $S/ODEs$ are stored in `ff`.

```
193 clear s;
194 operator s;
195 depend s,t;
196 let df(s(~i),t)=>ff(i);
```

The first linear approximation is then $\vec{u} \approx V_s \vec{s}$ such that $\dot{s}_i \approx 0$, in `ff(i)`.

```
197 ss_:=ov_s;
198 for j:=1:ns do ss_(j,1):=s(j);
199 u_:=sbase_*ss_;
200 if trace then write u_:=u_;
201 for j:=1:ns do ff(j):=0;
```

8 Iterative updates

We iterate to a solution of the governing S/ODEs to residuals of some order of error. The number of iterations are limited by a maximum.

```

202 write "==== Starting the constructive iteration";
203 res_:=df(u_,t)+urhs_;
204 write lengthres := tp map(length(~q),res_);
205
206 for iter:=1:maxiter do begin
207   if trace then write "
208     ITERATION = ",iter,"
209     -----";

```

Compute the current residual vector.

```

210   if trace then write res_:=res_;
211   mode_:=pp_inv*res_;
212   if trace then write mode_:=mode_;

```

8.1 Fast modes

Modify the slow manifold parametrisation in view of the fast modes.

```

213   du_:=ov_;
214   for j:=1:nf do begin
215     dyi_:=zres_(mode_(j+ns,1),ttf,-feval_(j,1));
216     if trace then write dyi_:=dyi_;
217     testing:=(dyi_ where znon_(~a,~b)=>testy(a,b,-1));
218     du_(j+ns,1):=dyi_;
219   end;
220 if trace then write "finished fast modes";

```

8.2 Slow modes

Update the s_i evolution $\text{ff}(i)$ and parts of the \vec{u} parametrisation.

```

221   for j:=1:ns do begin
222     ff(j):=ff(j)+(znon_(mode_(j,1),tt) where zrrules);
223     if trace then write "ds",j,"/dt = ",ff(j);

```

```

224     testing:=(ff(j) where znon_(~a,~b)=>testy(a,b,0));
225     dxi_:= (zint_(mode_(j,1),tt) where zrules);
226     if trace then write dxi_:=dxi_;
227     du_(j,1):=dxi_;
228     testing:=(dxi_ where zint_(~a,~b)=>testy(a,b,1));
229     end;

```

Update the slow manifold parametrisation

```

230     u_:=u_+pp_*du_;
231     if trace then write u_:=u_;

```

Terminate the iteration loop once all residuals are zero, or the maximum number of iterations has been done.

```

232     res_:=--df(u_,t)+urhs_;
233     write lengthres := tp map(length(~q),res_);
234     showtime;
235     if res_=ov_ then write "Number of iterations ",
236         iter:=1000000+iter;
237     end;

```

9 Output results

Only proceed to print if terminated successfully.

```

238 if res_=ov_
239 then write "==== SUCCESS: converged to an expansion"
240 else begin
241     out "slowReport.tex"; on latex;
242     write "Failed-to-converge";
243     off latex; shut "slowReport.tex";
244     rederr "==== FAILED TO CONVERGE: I EXIT";
245 end;

```

9.1 Plain text version

Print the resultant coordinate transform: but only print to one lower power in *small* and *sigma* in order to keep output relatively small.

```

246 write "Parametrisation of time-dependent slow manifold"$
247 write "(to one order lower in both small and sigma)"$
248 if trace then write u_=u_
249 else begin u_:=sigma*small*u_;
250     write u_:=u_/small/sigma; end;

```

Print the s/ODEs of the parametrisation of the slow manifold:

```

251 write "The slow manifold S/ODEs";
252 for i:=1:ns do write "ds(",i,")/dt = ",ff(i);

```

9.2 L^AT_EX version

Include order of error to make printing more robust. But we cannot use `small^toosmall` in the following as that is set to zero (for the asymptotics), so we hard code that `small` appears as `varepsilon` ε . Further, to avoid `sigma^3` being replaced by zero, introduce `sigma_` that maps to σ .

```

253 clear order_; operator order_;
254 defid order_,name="0";
255 defindex order_(arg,arg);
256 defid sigma_,name="\sigma";

```

As before, we have to write expressions to file for later reading so they get printed without extraneous dross in the L^AT_EX source. First open up the temporary file `scratchfile.red` again.

```

257 out "scratchfile.red"$
258 write "off echo;"$ % do not understand why needed in 2021??

```

Write the time-dependent coordinate transform to file, with a heading, and with an anti-math environment to cancel the auto-math of `rlfi`. For some reason we have to keep these writes short as otherwise it generates spurious fatal blank lines in the L^AT_EX.

```

259 write "write ""\
260 \paragraph{Time dependent slow manifold parametrisation}
261 \(";";
262 for i:=1:nu do write "u_",i,":=u_(",i,
263     ",1) +order_(varepsilon^",toosmall-1+trace,",sigma_^",2+trace,");";

```

Write the resultant model differential equations to file, with a heading, and with an anti-math environment to cancel the auto-math of `rlfi`.

```
264 write "write ""\n"
265 \paragraph{Result slow manifold DEs}
266 \("";";
267 for i:=1:ns do write "ff(",i,"):=ff(",i,
268     ") +order_(varepsilon^",toosmall,",sigma_^3);";
269 write "end;";
```

Shut the temporary output file.

```
270 shut "scratchfile.red";
```

Get expressions from file and write the main L^AT_EX file. But first *redefine* how these names get printed, namely as the slow parameters time derivatives.

```
271 defindex s_(down);
272 defid ff,name="\dot s";
```

Penultimately, write the header information file that is to be included in the report via an `\input` in the modified `on latex`.

```
273 out "slowReportHdr.tex"$
274 write "\title{A slow manifold of your dynamical system}"$
275 write "\author{A. J. Roberts, University of Adelaide\\
276 \texttt{http://orcid.org/0000-0001-8930-1552}}
277 \date{\now, \today}
278 \def\ou\big(#1,#2,#3\big)%
279     {\rm e}^{\if#31\else#3\fi t}\star\left(#1\right)}
280 \def\eps{\varepsilon} \def\_{\_}
281 \maketitle
282 Generally, the lowest order, most important,
283 terms are near the end of each expression.
284 \input{slowReportSys}
285 "$
286 shut "slowReportHdr.tex"$
```

Finally write to the main L^AT_EX file so switch on `latex` after starting to write to the file. Then write expressions in `scratchfile.red` to `slowReport.tex` as nice L^AT_EX. Switch off `latex`, to get the end of the document, and finish writing.

```

287 out "slowReport.tex"$
288 on latex$
289 in "scratchfile.red"$
290 off latex$
291 shut "slowReport.tex"$

```

10 Fin

That's all folks, so end the procedure, after clearing the mapping from operators to the stored expressions.

```

292 clear u(~j);
293 return Finished_slow_manifold_of_system$
294 end$

```

11 Override some rlfi procedures

Now setup the rlfi package to write a L^AT_EX version of the output. It is all a bit tricky and underhand. We override some stuff from `rlfi.red`.¹

First, change `name` to get Big delimiters, not left-right delimiters, so L^AT_EX can break lines.

```

295 deflist('(( ( !\!b!i!g!() (!) !\!b!i!g!)) (!P!I !\!p!i! )
296      (!p!i !\!p!i! ) (!E !e) (!I !i) (e !e) (i !i)), 'name)$

```

Override the procedure that prints annoying messages about multicharacter symbols. It ends the output of one expression. This is mainly a copy from `rlfi.red` with the appropriate if-statement deleted.

```

297 symbolic procedure prinlaend;
298 <<terpri();
299   prin2t "\)\par";
300   if !*verbatim then
301       <<prin2t "\begin{verbatim}";
302       prin2t "REDUCE Input:">>;

```

¹Find it in reduce-algebra/trunk/packages/misc/rlfi.red

```

303  ncharspr!*: =0;
304  if ofl!* then linelength(car linel!*)
305    else laline!*: =cdr linel!*;
306  nochar!*: =append(nochar!*,nochar1!*);
307  nochar1!*: =nil >>$

```

Similarly, hardcode at the beginning of expression output that the mathematics is in inline mode.

```

308  symbolic procedure prinlabegin;
309  <<if !*verbatim then
310    <<terpri();
311    prin2t "\end{verbatim}">>;
312  linel!*: =linelength nil . laline!*;
313  if ofl!* then linelength(laline!* + 2)
314    else laline!*: =car linel!* - 2;
315  prin2t "\(" >>$

```

Override the procedure that outputs the L^AT_EX preamble upon the command on latex.

```

316  symbolic procedure latexon;
317  <<!*!*a2sfn:='texaeval;
318  !*raise:=nil;
319  prin2t "\documentclass[11pt,a5paper]{article}";
320  prin2t "\usepackage[a5paper,margin=13mm]{geometry}";
321  prin2t "\usepackage{parskip,time} \raggedright";
322  prin2t "\begin{document}\input{slowReportHdr}";
323  if !*verbatim then
324    <<prin2t "\begin{verbatim}";
325    prin2t "REDUCE Input:">>;
326  put('tex,'rtypefn,'(lambda(x) 'tex)) >>$

```

End the file when input to Reduce

```

327  end;

```

References

- Carr, J. (1981), *Applications of centre manifold theory*, Vol. 35 of *Applied Math. Sci.*, Springer–Verlag.
<http://books.google.com.au/books?id=93BdN7btys0C>
- Chao, X. & Roberts, A. J. (1996), ‘On the low-dimensional modelling of Stratonovich stochastic differential equations’, *Physica A* **225**, 62–80.
- Coullet, P. H. & Spiegel, E. A. (1983), ‘Amplitude equations for systems with competing instabilities’, *SIAM J. Appl. Math.* **43**, 776–821.
- Cox, S. M. & Roberts, A. J. (1991), ‘Centre manifolds of forced dynamical systems’, *J. Austral. Math. Soc. B* **32**, 401–436.
- Fateman, R. (2003), ‘Comparing the speed of programs for sparse polynomial multiplication’, *ACM SIGSAM Bulletin* **37**(1), 4–15.
<http://www.cs.berkeley.edu/~fateman/papers/fastmult.pdf>
- Haragus, M. & Iooss, G. (2011), *Local Bifurcations, Center Manifolds, and Normal Forms in Infinite-Dimensional Dynamical Systems*, Springer.
- Knobloch, H.-W. & Aulbach, B. (1982), The role of center manifolds in ordinary differential equations, in M. Gregus, ed., ‘Proceedings of the fifth Czechoslovak conference on differential equations and their applications’, BSB B.G. Teubner Verlagsgesellschaft, Leipzig, pp. 179–189.
http://dml.cz/bitstream/handle/10338.dmlcz/702285/Equadiff_05-1982-1_40.pdf
- Roberts, A. J. (1997), ‘Low-dimensional modelling of dynamics via computer algebra’, *Computer Phys. Comm.* **100**, 215–230.
- Roberts, A. J. (2008), ‘Normal form transforms separate slow and fast modes in stochastic dynamical systems’, *Physica A* **387**, 12–38.
- Roberts, A. J. (2015), *Model emergent dynamics in complex systems*, SIAM, Philadelphia.
<http://bookstore.siam.org/mm20/>

Roberts, A. J. (2022), Backwards theory supports modelling via invariant manifolds for non-autonomous dynamical systems, Technical report, [<http://arxiv.org/abs/1804.06998>].