

A general invariant manifold construction procedure, including isochrons of slow manifolds

A. J. Roberts*

Nov 2013 – April 9, 2025

Abstract

This procedure constructs a specified invariant manifold for a specified system of ordinary differential equations or delay differential equations, or partial differential equations where the spatial gradients are small. The invariant manifold may be any of a centre manifold, a slow manifold, an un/stable manifold, a sub-centre manifold, a nonlinear normal form, any spectral submanifold, or indeed a normal form coordinate transform of the entire state space. Thus the procedure may be used to analyse pitchfork bifurcations, or oscillatory Hopf bifurcations, or any more complicated superposition. In the cases when the neglected spectral modes all decay, the constructed invariant manifold supplies a faithful large time model of the dynamics of the differential equations. Further, in the case of a slow manifold, this procedure now derives vectors defining the projection onto the invariant manifold along the isochrons: this projection is needed for initial conditions, forcing, system modifications, and uncertainty quantification.

Contents

1	Introduction	2
1.1	A simple example: <code>exampleslowman()</code>	5

*School of Mathematical Sciences, University of Adelaide, South Australia 5005, AUSTRALIA. <https://profajroberts.github.io/>

1	<i>Introduction</i>	2
1.2	Header of the procedure	7
1.3	Preamble to the procedure	8
1.4	Check the dimensionality of specified system	9
1.5	Define time and optional space dependencies	10
2	Dissect the linear part	10
2.1	Normalise the adjoint eigenvectors	11
2.2	Operator to represent delays	12
2.3	Linearise at the origin	13
2.4	Eigen-check	13
2.5	Ameliorate the nonlinearity	15
2.6	Store invariant manifold eigenvalues	15
2.7	Precompute matrices for updates	16
2.8	Define operators that invoke these inverses	19
2.9	Projection onto slow manifold for PDE case	20
3	Initialise LaTeX output	21
4	Linear approximation to the invariant manifold	24
5	Iteratively construct the invariant manifold	25
6	Output text version of results	27
7	Output LaTeX version of results	29
8	Fin	31
9	Override some system procedures	31

1 Introduction

Installation Download and install the computer algebra package *Reduce* via <http://www.reduce-algebra.com> Download and unzip the folder <https://profajroberts.github.io/InvariantManifold.zip> Within the folder `InvariantManifold`, start-up *Reduce* and load the procedure by

executing the command `in_tex "invariantManifold.tex"`¹ Test your installation by then executing `exampleslowman()`; (see [Section 1.1](#)).

Execution Thereafter, construct a specified invariant manifold of a specific dynamical system by executing the following command with specific values for the input parameters. See `diverseExamples.pdf` for many examples.

```
1 invariantmanifold(spacex, odefns, evals, evecs,
2   adjvecs, toosmall);
```

Inputs As in the example of the next [Section 1.1](#), the input parameters to the procedure are the following:

- **spacex**: for ODEs provide the empty list “{}”; for PDEs with large-scale, gradual, variations in some ‘space’ variables then specify the list of space variables, for example `{x,y}` denotes that there are ‘spatial’ gradients in variables x, y , gradients that will be treated as small.
- **odefns**, a comma separated list within `mat(...)`, the RHS expressions of the ODEs/DDEs of the system, a system expressed in terms of variables `u1, u2, ...`, for time derivatives $du_1/dt, du_2/dt, \dots$;

any time delayed variables in the RHS are coded by the time-delay in parenthesis after the variable, as in the example `u1(pi/2)` to represent $u_1(t - \pi/2)$ in the DDEs;

any ‘spatial’ derivatives *must* be specified using a `diff()` operator (currently with at most three arguments)—examples are `diff(u1,x) = $\partial u_1/\partial x$` , `diff(u3,x,y) = $\partial^2 u_3/\partial x \partial y$` , `diff(u2,x,4) = $\partial^4 u_2/\partial x^4$` .

- **evals**, a comma separated list within `mat(...)`, the eigenvalues of the modes to be the basis for the invariant manifold—each eigenvalue may be complex-valued, of the form `a+b*i`;
- **evecs**, a comma separated list of vectors within `mat(...)`—each vector a comma separated list of components within `(...)`, the eigenvectors

¹This script changes many internal settings of *Reduce*, so best done only when needed.

of the modes corresponding to the given eigenvalues of the invariant manifold basis—possibly complex-valued;

- `adjvecs`, a comma separated list of vectors within `mat(...)`, often the adjoint eigenvectors of the modes corresponding to the given eigenvalues of the invariant manifold basis;
- `toosmall`, an integer giving the desired order of error in the asymptotic approximation that is constructed. The procedure embeds the specified system in a family of systems parametrised by ε , and constructs an invariant manifold, and evolution thereon, of the embedding system to the asymptotic error $\mathcal{O}(\varepsilon^{\text{toosmall}})$ (as $\varepsilon \rightarrow 0$). Often the introduced artificial ε has a useful physical meaning, but strictly you should evaluate the output at $\varepsilon = 1$ to recover results for the specified system, and then interpret the results in terms of actual ‘small’ parameters.

Outputs This procedure reports the specified system, the embedded system it actually analyses, the number of iterations taken, the invariant manifold approximation, the evolution on the invariant manifold, and optionally a basis for projecting onto the invariant manifold.

- A plain text report to the Terminal window in which `Reduce` is executing—the invariant manifold is parametrised by variables `s(1)`, `s(2)`, ..., and the dynamics by their evolution in time.
- A L^AT_EX source report written to the file `invarManReport.tex` (and `invarManReportSys.tex`)—the invariant manifold is parametrised by variables s_1, s_2, \dots , and the dynamics by their evolution in time. Generate a pdf version by executing `pdflatex invarManReport`.
- Global variable `uu` gives the constructed invariant manifold such that `coeffn(uu,e_(i,1),1)` gives the i th coordinate, u_i , of the invariant manifold as a function of `s(j)`, s_j .
- Global variable `gg` gives the evolution on the invariant manifold, such that `coeffn(gg,e_(j,1),1)` gives the time derivative of `s(j)`, \dot{s}_j .
- Global variable `zs` (optional): in the case of a slow manifold (where all

specified eigenvalues are zero), `zs` gives the normals to the isochrons at the slow manifold, such that `coeffn(zs,e_(i,j),1)` as a function of \vec{s} , is the i th component of the j th normal vector to the isochron.

One may change the appearance of the output somewhat. For example, it is often useful to execute `factor s`; before executing `invariantmanifold(...)` in order to group terms with the same powers of amplitudes/order-parameters/coarse-variables.

Background The theoretical support for the results of the analysis of this procedure is centre/stable/unstable manifold theory (e.g., Carr 1981, Haragus & Iooss 2011, Roberts 2015b), and an embryonic backwards theory (Roberts 2022). This particular procedure is developed from a coordinate-independent algorithm for constructing centre manifolds originally by Coullet & Spiegel (1983), adapted for human-efficient computer algebra by Roberts (1997), extended to invariant/inertial manifolds (Roberts 1989b, Foias et al. 1988), extended to the evolution of gradually varying patterns in space (Roberts 1988, 2015a, Bunder & Roberts 2021); and further extended to the projection of initial conditions, forcing, uncertainty, etc (Roberts 1989a, 2000).

We use the computer algebra package *Reduce* [<http://reduce-algebra.com/>] because it is both free and perhaps the fastest general purpose computer algebra system (Fateman 2003, e.g.).

1.1 A simple example: `exampleslowman()`

Execute this example by invoking the command `exampleslowman()`; The example system to analyse is specified to be (Roberts 2015b, Example 2.1)

$$\dot{u}_1 = -u_1 + u_2 - u_1^2, \quad \dot{u}_2 = u_1 - u_2 + u_2^2.$$

```
3 procedure exampleslowman;
4   invariantmanifold({},
5     mat((-u1+u2-u1^2,u1-u2+u2^2)),
6     mat((0)),
7     mat((1,1)),
```

```

8      mat((1,1)),
9      5)$

```

We seek the slow manifold so specify the eigenvalue zero. From the linearisation matrix $\begin{bmatrix} -1 & 1 \\ 1 & -1 \end{bmatrix}$ a corresponding eigenvector is $\vec{e} = (1, 1)$, and a corresponding left-eigenvector is $\vec{z} = \vec{e} = (1, 1)$, as specified. The last parameter specifies to construct the slow manifold to errors $\mathcal{O}(\varepsilon^5)$.

The procedure actually analyses the embedding system, the family of problems,

$$\dot{u}_1 = -u_1 + u_2 - \varepsilon u_1^2, \quad \dot{u}_2 = u_1 - u_2 + \varepsilon u_2^2.$$

Here the artificial parameter ε has a physical interpretation in that it counts the nonlinearity: a term in ε^p will be a $(p + 1)$ th order term in $\vec{u} = (u_1, u_2)$. Hence the specified error $\mathcal{O}(\varepsilon^5)$ is here the same as error $\mathcal{O}(|\vec{s}|^6)$.

The constructed slow manifold is, in terms of the parameter s_1 (and reverse ordering!),

$$\begin{aligned} u_1 &= 3/8\varepsilon^3 s_1^4 - 1/2\varepsilon s_1^2 + s_1 + O(\varepsilon^4), \\ u_2 &= -3/8\varepsilon^3 s_1^4 + 1/2\varepsilon s_1^2 + s_1 + O(\varepsilon^4). \end{aligned}$$

On this slow manifold the evolution is

$$\dot{s}_1 = -3/4\varepsilon^4 s_1^5 + \varepsilon^2 s_1^3 + O(\varepsilon^5) :$$

here the leading term in s_1^3 indicates the origin is unstable. To project initial conditions onto the slow manifold, or non-autonomous forcing, or modifications of the original system, or to quantify uncertainty, use the projection defined by the derived vector

$$\vec{z}_1 = \begin{bmatrix} z_{11} \\ z_{12} \end{bmatrix} = \begin{bmatrix} 3/2\varepsilon^4 s_1^4 + 3/4\varepsilon^3 s_1^3 - 1/2\varepsilon^2 s_1^2 - 1/2\varepsilon s_1 + 1/2 \\ 3/2\varepsilon^4 s_1^4 - 3/4\varepsilon^3 s_1^3 - 1/2\varepsilon^2 s_1^2 + 1/2\varepsilon s_1 + 1/2 \end{bmatrix} + O(\varepsilon^5).$$

Evaluate these at $\varepsilon = 1$ to apply to the original specified system, or alternatively just interpret ε as a way to count the order of each term.

1.2 Header of the procedure

Need a couple of things established before defining the procedure: the `rlfi` package; and operator names for the variables of the dynamical system (in case they have delays)—currently code a max of nine variables.

```
10 load_package rlfi;
11 operator u1,u2,u3,u4,u5,u6,u7,u8,u9;
```

Allow ‘space’ derivatives, to be assumed small, via this `diff` operator that has the same syntax as `df()` but we need separate name in order to delay Reduce processing.

```
12 operator diff;
```

Usually seems best to use the `gcd` option, but sometimes worse, so set here in order for a user to optionally turn off before invoking the procedure.

```
13 on gcd,ezgcd;
```

Now define the procedure as an operator so we can define procedures internally, and may be flexible with its arguments.

```
14 operator invariantmanifold;
```

Four or fewer arguments is an error. If just five arguments, then assume no ‘space’ variables.

```
15 IErrorMsg:="invariantmanifold() Should Have Six Arguments"$
16 for all a1,a2,a3,a4 let
17     invariantmanifold(a1,a2,a3,a4) = <<rederr(IErrorMsg)>>;
18 for all a1,a2,a3,a4,a5 let
19     invariantmanifold(a1,a2,a3,a4,a5) = begin
20         write "**** Warning: ",IErrorMsg;
21         invariantmanifold({},a1,a2,a3,a4,a5);
22         rederr(IErrorMsg);
23     end;
```

Define the current procedure with six arguments. [Section 8](#) marks the end of the procedure.

```

24 for all spacex, odefns, evals, evecs, adjvecs, toosmall
25     let invariantmanifold(spacex, odefns, evals, evecs,
26         adjvecs, toosmall) = begin

```

1.3 Preamble to the procedure

Operators and arrays are always global, but we can make variables and matrices local, except for matrices that need to be declared `matrix`. So, move to implement all arrays and operators to have underscores, and almost all scalars and most matrices to be declared local here.

```

27 scalar ff, evalm, ee, zz, maxiter, trace, ll, uvec,
28 reslin, ok, rhsjact, jacadj, resd, resde, resz, rhsfn,
29 pp, est, eyem, m, diffzero;

```

Write an intro message.

```

30 write "Construct invariant manifold (version 9 Apr 2025)"$

```

Transpose the defining matrices so that vectors are columns.

```

31 ff := tp odefns;
32 ee := tp evecs;
33 zz := tp adjvecs;

```

Define default parameters for the iteration: `maxiter` is the maximum number of allowed iterations. Specific problems may override this default.

```

34 maxiter:=29$

```

For optional trace printing of test cases: comment out second line when not needed.

```

35 trace:=0$
36 %trace:=1; maxiter:=9; %%%

```

The `rationalize` switch makes code much faster with complex numbers. The switch `gcd` seems to wreck convergence for `doubleHofDDE`, so leave it off. But seems OK for all other examples.


```

37 on div; off allfac; on revpri;
38 on rationalize;

```

Use e_- as basis vector for matrices and vectors. Declare it non-commutative so that multiplication does not commute.

```

39 clear e_-; operator e_-; noncom e_-;
40 factor e_-;
41 let { e_(~j,~k)*e_(~l,~p)=>0 when k neq l
42      , e_(~j,~k)*e_(~l,~p)=>e_~(j,p) when k=l
43      , e_(~j,~k)^2=>0 when j neq k
44      , e_(~j,j)^2=>e_~(j,j) };

```

Also need (once) a transpose operator: do complex conjugation explicitly when needed.

```

45 clear tpe_-; operator tpe_-; linear tpe_-;
46 let tpe_~(e_~(i,~j),e_~)=>e_~(j,i);

```

Empty the output LaTeX file in case of error.

```

47 out "invarManReport.tex";
48 write "This empty document indicates error.";
49 shut "invarManReport.tex";

```

1.4 Check the dimensionality of specified system

Extract dimension information from the parameters of the procedure: seek mD invariant manifold of an nD system.

```

50 write "total no. of variables ",
51 n:=part(length(ee),1);
52 write "no. of invariant modes ",
53 m:=part(length(ee),2);
54 if {length(evals),length(zz),length(ee),length(ff)}
55   ={{1,m},{n,m},{n,m},{n,1}}
56   then write "Input dimensions are OK"
57   else <<write "INCONSISTENT INPUT DIMENSIONS, I EXIT";

```

```
58         return>>;
```

For the moment limit to a maximum of nine components.

```
59 if n>9 then <<write "SORRY, MAX NUMBER ODEs IS 9, I EXIT";
60     return>>;
```

Need an $m \times m$ identity matrix for normalisation of the isochron projection.

```
61 eyem:=for j:=1:m sum e_(j,j)$
```

1.5 Define time and optional space dependencies

```
62 if trace then write "creating depends upon spacex";
63 for j:=1:9 do begin
64     depend mkid(u,j),t;
65     foreach z_ in spacex do depend mkid(u,j),z_;
66 end;
```

Also set rules when needed to omit `diff` terms, currently coded up to five arguments.

```
67 diffzero:={ diff(~a)=>0, diff(~a,~b)=>0, diff(~a,~b,~c)=>0
68             , diff(~a,~b,~c,~d)=>0, diff(~a,~b,~c,~d,~e)=>0 };
```

2 Dissect the linear part

Use the exponential $\exp(u) = e^u$, but not with the myriad of inbuilt properties so clear it! Do not (yet) invoke the simplification of $\exp(0)$ as I want it to label modes of no oscillation, zero eigenvalue.

```
69 clear exp; operator exp;
70 let { df(exp(~u),t) => df(u,t)*exp(u)
71       , exp(~u)*exp(~v) => exp(u+v)
72       , exp(~u)^~p => exp(p*u)
73     };
```

Also try mapping any user supplied sinusoids into this `exp()` so that we can handle harmonically forced systems. Only invoke on the supplied ODEs as delay differential equations use trig functions.

```
74 ff:=(ff where { cos(~u) => (exp(i*u)+exp(-i*u))/2
75                  , sin(~u) => (exp(i*u)-exp(-i*u))/(2*i)
76                  } );
```

Need function `conj_` to do parsimonious complex conjugation.

```
77 procedure conj_(a)$ sub(i=-i,a)$
```

Make an array of eigenvalues for simplicity (`evals` not used hereafter). Substitute `small=0` in the eigenvalues just in case someone wants to detune eigenvalues in the analysis and supply the same parameter in the eigenvalues.

```
78 clear eval_; array eval_(m);
79 for j:=1:m do eval_(j):=sub(small=0,evals(1,j));
```

2.1 Normalise the adjoint eigenvectors

When we include delay differential equations, then we need to account for the history of the eigenvector as well. Hence multiply each eigenvector by its oscillating factor, $e^{i\omega t}$, $e^{\lambda t}$, and then take the mean. This multiplication by its oscillating factor should not make any difference for non-delay equations by the natural orthogonality of left and right eigenvectors of different eigenvalues.

Note: the ‘left eigenvectors’ have to be the eigenvectors of the complex conjugate transpose, and for the complex conjugate eigenvalue. This seems best: for example, when the linear operator is $\begin{bmatrix} 0 & 1 \\ -1 & 0 \end{bmatrix}$ then the adjoint and the right eigenvectors are the same.

For oscillations and un/stable manifolds we have to cope with imaginary and with real eigenvalues. Seems to need `zz` to have negative complex conjugated frequency so store in `cexp_`—cannot remember why this appears to work!? It may only work for pure real and for pure imaginary eigenvalues??

```
80 matrix aa_(m,m),dexp_(m,m),cexp_(m,m);
81 for j:=1:m do dexp_(j,j):=exp(eval_(j)*t);
```

```

82 for j:=1:m do cexp_(j,j):=exp(-conj_(eval_(j))*t);
83 aa_:=(tp map(conj_(~b),ee*dexp_)*zz*cexp_ )$
84 if trace then write aa_:=aa_;
85 write "Normalising the left-eigenvectors:";
86 aa_:=(aa_ where {exp(0)=>1, exp(~a)=>0 when a neq 0})$
87 if trace then write aa_:=aa_;
88 if det(aa_)=0 then << write
89     "ORTHOGONALITY ERROR IN EIGENVECTORS; I EXIT";
90     return>>;
91 zz:=zz*aa_^(-1);

```

2.2 Operator to represent delays

Introduce an operator to represent delay factors more conveniently for analysis. The `exp` rule probably only works for pure imaginary modes!?

```

92 clear d_; operator d_; linear d_;
93 let { d_(~a~p,t,~dt)=>d_(a,t,dt)~p
94     , d_(~a~b,t,~dt)=>d_(a,t,dt)*d_(b,t,dt)
95     , d_(exp(~a),t,~dt)=>exp(a)
96         *sub(t=-dt,cos(-i*a)+i*sin(-i*a))
97     , df(d_(~a,t,~dt),~b)=>d_(df(a,b),t,dt)
98     , d_(~a,t,0)=>a
99     , d_(d_(~a,t,~dta),t,~dtb)=>d_(a,t,dta+dtb)
100 };

```

Now rewrite the (delay) factors in terms of this operator. For the moment limit to a maximum of nine ODEs.

```

101 if trace then write "setting somerules";
102 somerules:={} $
103 somerules:=(u1(~dt)=d_(u1,t,dt)).somerules$
104 somerules:=(u2(~dt)=d_(u2,t,dt)).somerules$
105 somerules:=(u3(~dt)=d_(u3,t,dt)).somerules$
106 somerules:=(u4(~dt)=d_(u4,t,dt)).somerules$
107 somerules:=(u5(~dt)=d_(u5,t,dt)).somerules$

```

```

108 somerules:=(u6(~dt)=d_(u6,t,dt)).somerules$
109 somerules:=(u7(~dt)=d_(u7,t,dt)).somerules$
110 somerules:=(u8(~dt)=d_(u8,t,dt)).somerules$
111 somerules:=(u9(~dt)=d_(u9,t,dt)).somerules$
112 ff:=(ff where somerules)$

```

2.3 Linearise at the origin

Assume the equilibrium is at the origin. Find the linear operator at the equilibrium. Include `small=0` as we notionally adjoin it in the list of variables. Do not need to here make any non-zero forcing `small` at the equilibrium as it gets multiplied by `small` later. (For some reason using `mkid(u,k)=>0` does not resolve the `mkid`, but `mkid(u,k)=0` does; however, not clear if it is a problem.)

```

113 ll:=ee*(tp ee)*0; %zero nxn matrix
114 uzero:=(for k:=1:n collect (mkid(u,k)=0))$
115 equilibrium:=(small=0).uzero$
116 for j:=1:n do for k:=1:n do begin
117   ll(j,k):=df(ff(j,1),mkid(u,k));
118   ll(j,k):=( sub(equilibrium,ll(j,k)) where diffzero );
119 end;
120 write "Find the linear operator is";
121 write ll:=ll;

```

We need a vector of unknowns for a little while.

```

122 uvec:=0*ff; %nx1 zero matrix
123 for j:=1:n do uvec(j,1):=mkid(u,j);

```

2.4 Eigen-check

Variable `aa_` appears here as the diagonal matrix of eigenvalues. Check that the eigenvalues and eigenvectors are specified correctly.

```

124 write "Check invariant subspace linearisation ";
125 for j:=1:m do for k:=1:m do aa_(j,k):=0;

```

```

126 for j:=1:m do aa_(j,j):=eval_(j);
127 % following maybe only for pure centre modes??
128 reslin:=(ll*(ee*dexp_)-(ee*dexp_)*aa_
129     where exp(~a)*d_(1,t,~dt)
130     =>sub(t=-dt,cos(-i*a)+i*sin(-i*a))*exp(a) )$
131 if trace then write reslin:=reslin;
132 ok:=1$
133 for j:=1:n do for k:=1:m do
134     ok:=if reslin(j,k)=0 then ok else 0$
135 if ok then write "Linearisation is OK";

```

Try to find a correction of the linear operator that is ‘close’. Multiply by the adjoint eigenvectors and then average over time: operator $\mathcal{L}_{\text{new}} := \mathcal{L} - \mathcal{L}_{\text{adj}}$ should now have zero residual. Lastly, correspondingly adjust the ODEs, since `lladj` does not involve delays we do not need delay operator transforms in the product.

```

136 if not ok then for iter:=1:2 do begin
137 write "WARNING: I NEED TO ADJUST LINEAR OPERATOR";
138 lladj:=reslin*tp map(conj_(~b),zz*cexp_);
139 write
140 lladj:=(lladj where {exp(0)=>1, exp(~a)=>0 when a neq 0});
141 write
142 ll:=ll-lladj;
143 % following maybe only for pure centre modes??
144 write
145 reslin:=(ll*(ee*dexp_)-(ee*dexp_)*aa_
146     where exp(~a)*d_(1,t,~dt)
147     => sub(t=-dt,cos(-i*a)+i*sin(-i*a))*exp(a) );
148 ok:=1$
149 for j:=1:n do for k:=1:m do
150     ok:=if reslin(j,k)=0 then ok else 0$
151 if ok then iter:=iter+1000;
152 end;
153 if not ok then << write
154     "OOPS, INCONSISTENT EIGENVALUES, EIGENVECTORS AND OPERATOR.

```

```

155     EMAIL ME; I EXIT";
156     return >>;

```

2.5 Ameliorate the nonlinearity

Anything not in the linear operator gets multiplied by `small` to be treated as small in the analysis.

```

157 somerules:=for j:=1:n collect
158   (d_(1,t,~dt)*mkid(u,j)=d_(mkid(u,j),t,dt))$
159 ll0_:=(ll*uvec where somerules);
160 ffnondiff:=(ff where diffzero);
161 procedure scalediff(z_);
162   (z_ where {diff(~a,~b)=>small*df(a,b)
163             , diff(~a,~b,~n)=>small^n*df(a,b,n) when fixp(n)
164             , diff(~a,~b,~c)=>small^2*df(a,b,c) when not fixp(c)
165             });
166 ffdiff:=map(scalediff(~z),ff-ffnondiff);
167 ff:=small*ffnondiff
168   +(1-small)*ll0_
169   +ffdiff;

```

Any constant term in the equations `ff` has to be multiplied by `exp(0)`.

```

170 ff:=ff+(exp(0)-1)*sub(uzero,ff)$

```

From the matrix versions of the equations, create algebraic form using the matrix basis.

```

171 rhsfn:=for i:=1:n sum e_(i,1)*ff(i,1)$
172 if trace then write "rhsfn=",rhsfn;

```

Also, create the algebraic form of the jacobian transpose using the matrix basis: take the conjugate later when used.

```

173 rhsjact:=for i:=1:n sum for j:=1:n sum
174   e_(j,i)*df(ff(i,1),mkid(u,j));

```

2.6 Store invariant manifold eigenvalues

Extract all the eigenvalues in the invariant manifold, and the set of all the corresponding modes in the invariant manifold variables. The slow modes are accounted for as having zero eigenvalue. Remember the eigenvalue set is not in the ‘correct’ order. Array `modes_` stores the set of indices of all the modes of a given eigenvalue.

```

175 clear eval_s,modes_;
176 array eval_s(m),modes_(m);
177 neval:=0$ eval_set:={}
178 for j:=1:m do if not(eval_(j) member eval_set) then begin
179   neval:=neval+1;
180   eval_s(neval):=eval_(j);
181   eval_set:=eval_(j).eval_set;
182   modes_(neval):=for k:=j:m join
183     if eval_(j)=eval_(k) then {k} else {};
184 end;
```

Set a flag for the case of a slow manifold when all eigenvalues are zero, as then we compute the isochron projection. The next challenge is to get this isochron code working for the case of non-slow invariant manifolds.

```

185 itisSlowMan_:=if eval_set={0} then 1 else 0$
186 if trace then write itisSlowMan_:=itisSlowMan_;
```

Put in the non-singular general case as the zero entry of the arrays.

```

187 eval_s(0):=geneval$
188 modes_(0):={}

```

2.7 Precompute matrices for updates

Precompute matrices to solve for updates for each of the critical eigenvalues, and the general case $k = 0$. The matrix

$$\mathbf{11zz}_- = \begin{bmatrix} -\mathcal{L} + \partial_t & \mathcal{E}_0 \\ \mathcal{Z}_0^\dagger & 0 \end{bmatrix}$$

and then put its inverse in place. Subsequently, extract the blocks for the generalised inverses and solvability condition into `l_invs` and `g_invs`.

```

189 matrix llzz_(n+m,n+m);
190 clear l_invs,g_invs,l1_invs,g1_invs,l2_invs,g2_invs;
191 array l_invs(neval), g_invs(neval), l1_invs(neval),
192       g1_invs(neval), l2_invs(neval), g2_invs(neval);
193 clear sp_; operator sp_; linear sp_;
194 for k_:=0:neval do begin
195   if trace then write "ITERATION ",k_;

```

Code the operator $\mathcal{L}\hat{v}$ where the delay is to only act on the oscillation part.

```

196   for ii:=1:n do for jj:=1:n do llzz_(ii,jj):=(
197     -sub(small=0,ll(ii,jj)) where d_(1,t,~dt)
198     => cos(i*eval_s(k_)*dt)+i*sin(i*eval_s(k_)*dt));

```

Code the operator $\partial\hat{v}/\partial t$ where it only acts on the oscillation part.

```

199   for j:=1:n do llzz_(j,j):=eval_s(k_)+llzz_(j,j);

```

Now code the part leading to the solvability condition which arises from allowing the (oscillation) amplitude to evolve. Use operator `sp_` to extract the delay parts that subtly affect the updates of the evolution.

```

200   for j:=1:length(modes_(k_)) do
201     for ii:=1:n do llzz_(ii,n+j):=ee(ii,part(modes_(k_),j))
202       +(for jj:=1:n sum
203         sp_(ll(ii,jj)*ee(jj,part(modes_(k_),j)),d_)
204         where { sp_(1,d_)=>0
205               , sp_(d_(1,t,~dt),d_)=>dt*(
206                 cos(i*eval_s(k_)*dt)+i*sin(i*eval_s(k_)*dt))
207               });

```

Force the updates to be orthogonal to the left-eigenvectors in the complex conjugate transpose adjoint.

```

208   for ii:=1:length(modes_(k_)) do for j:=1:n do
209     llzz_(n+ii,j):=conj_(zz(j,part(modes_(k_),ii)));

```

```

210   if trace then write
211     "finished Force the updates to be orthogonal";

```

Set the bottom-right corner of the matrix to zero.

```

212   for i:=1:length(modes_(k_)) do
213     for j:=1:m do llzz_(n+i,n+j):=0;

```

Add some trivial rows and columns to make the matrix up to the same size for all eigenvalues.

```

214   for i:=length(modes_(k_))+1:m do begin
215     for j:=1:n+i-1 do llzz_(n+i,j):=llzz_(j,n+i):=0;
216     llzz_(n+i,n+i):=1;
217   end;
218   if trace then write
219     "finished Add some trivial rows and columns";

```

Invert the matrix and unpack into arrays ready for use by the inversion operators.

```

220   if trace then write llzz_:=llzz_;
221   llzz_:=llzz_^(-1);
222   if trace then write llzz_:=llzz_;
223   l_invs(k_):=for i:=1:n sum for j:=1:n sum e_(i,j)*llzz_(i,j);
224   g_invs(k_):=for i:=1:length(modes_(k_)) sum
225     for j:=1:n sum e_(part(modes_(k_),i),j)*llzz_(i+n,j);
226   if trace then write "finished Invert the matrix and unpack";

```

Unpack the conjugate transpose for inverse operators used for the isochrons. A difference here is that the orthogonality condition is non-trivial (in the slow manifold we assumed amplitudes were exactly orthogonal to the left-eigenvectors), so we need to remember more parts of the inverse of the matrix.

```

227   l1_invs(k_) := for ii:=1:n sum for j:=1:n sum
228     e_(ii,j)*conj_(llzz_(j,ii));
229   l2_invs(k_) := for ii:=1:n sum
230     for j:=1:length(modes_(k_)) sum

```

```

231         e_(ii,part(modes_(k_),j))*conj_(llzz_(j+n,ii));
232   g1_invs(k_) := for ii:=1:length(modes_(k_)) sum
233     for j:=1:n sum
234       e_(part(modes_(k_),ii),j)*conj_(llzz_(j,ii+n));
235   g2_invs(k_) := for ii:=1:length(modes_(k_)) sum
236     for j:=1:length(modes_(k_)) sum
237       e_(part(modes_(k_),ii),part(modes_(k_),j))
238       *conj_(llzz_(j+n,ii+n));
239   if trace then write
240     "finished Unpack the conjugate transpose";
241 end;

```

2.8 Define operators that invoke these inverses

Decompose residuals into parts, and operate on each. First for the invariant manifold. But making $e_$ non-commutative means that it does not get factored out of these linear operators: must post-multiply by $e_$ because the linear inverse is a premultiply.

```

242 clear l_inv; operator l_inv; linear l_inv;
243 let l_inv(e_(~j,~k)*exp(~a),exp)=>l_invproc(a/t)*e_(j,k);
244 procedure l_invproc(a);
245   if a member eval_set
246   then << k_:=0;
247     repeat k_:=k_+1 until a=eval_s(k_);
248     l_invs(k_)*exp(a*t) >>
249   else sub(geneval=a,l_invs(0))*exp(a*t)$

```

Second for the evolution on the invariant manifold.

```

250 clear g_inv; operator g_inv; linear g_inv;
251 let g_inv(e_(~j,~k)*exp(~a),exp)=>ginv_proc(a/t)*e_(j,k);
252 procedure ginv_proc(a);
253   if a member eval_set
254   then << k_:=0;
255     repeat k_:=k_+1 until a=eval_s(k_);

```

```

256     g_invs(k_) >>
257     else sub(geneval=a,g_invs(0))$

```

Copy and adjust the above for the projection. But first define the generic procedure.

```

258 procedure inv_proc(a,invs);
259   if a member eval_set
260   then << k_:=0;
261       repeat k_:=k_+1 until a=eval_s(k_);
262       invs(k_)*exp(a*t) >>
263   else sub(geneval=a,invs(0))*exp(a*t)$

```

Then define operators that we use to update the projection.

```

264 clear l1_inv; operator l1_inv; linear l1_inv;
265 clear l2_inv; operator l2_inv; linear l2_inv;
266 clear g1_inv; operator g1_inv; linear g1_inv;
267 clear g2_inv; operator g2_inv; linear g2_inv;
268 let { l1_inv(e_(~j,~k)*exp(~a),exp)
269       => inv_proc(a/t,l1_invs)*e_(j,k)
270       , l2_inv(e_(~j,~k)*exp(~a),exp)
271       => inv_proc(a/t,l2_invs)*e_(j,k)
272       , g1_inv(e_(~j,~k)*exp(~a),exp)
273       => inv_proc(a/t,g1_invs)*e_(j,k)
274       , g2_inv(e_(~j,~k)*exp(~a),exp)
275       => inv_proc(a/t,g2_invs)*e_(j,k)
276   };

```

2.9 Projection onto slow manifold for PDE case

Now that we have finished with `diff`, repurpose it for the bare differentiation operator needed for the projection vectors. This seems to work for 1-D space PDEs, but needs checking.

```

277 if spacex neq {} then
278 let { df(~a,~x,~b)=>diff(x) when a=b

```

```

279      , df(~a,~x,~n,~b)=>diff(x)^n when a=b and fixp(n)
280      , df(~a,~x,~y,~b)=>diff(x)*diff(y) when a=b and not fixp(y)
281      , df(~a,~x,~n,~y,~b)=>diff(x)^n*diff(y)
282          when a=b and fixp(n) and not fixp(y)
283      , df(~a,~x,~y,~n,~b)=>diff(x)*diff(y)^n
284          when a=b and not fixp(y) and fixp(n)
285      , df(~a,~x,~y,~z,~b)=>diff(x)*diff(y)*diff(z)
286          when a=b and not fixp(y) and not fixp(z)
287  };

```

3 Initialise LaTeX output

Define the Greek alphabet with `small` as well.

```

288 defid small,name="\eps";%varepsilon;
289 defid alpha,name=alpha;
290 defid beta,name=beta;
291 defid gamma,name=gamma;
292 defid delta,name=delta;
293 defid epsilon,name=epsilon;
294 defid varepsilon,name=varepsilon;
295 defid zeta,name=zeta;
296 defid eta,name=eta;
297 defid theta,name=theta;
298 defid vartheta,name=vartheta;
299 defid iota,name=iota;
300 defid kappa,name=kappa;
301 defid lambda,name=lambda;
302 defid mu,name=mu;
303 defid nu,name=nu;
304 defid xi,name=xi;
305 defid pi,name=pi;
306 defid varpi,name=varpi;
307 defid rho,name=rho;
308 defid varrho,name=varrho;

```

```
309 defid sigma,name=sigma;
310 defid varsigma,name=varsigma;
311 defid tau,name=tau;
312 defid upsilon,name=upsilon;
313 defid phi,name=phi;
314 defid varphi,name=varphi;
315 defid chi,name=chi;
316 defid psi,name=psi;
317 defid omega,name=omega;
318 defid Gamma,name=Gamma;
319 defid Delta,name=Delta;
320 defid Theta,name=Theta;
321 defid Lambda,name=Lambda;
322 defid Xi,name=Xi;
323 defid Pi,name=Pi;
324 defid Sigma,name=Sigma;
325 defid Upsilon,name=Upsilon;
326 defid Phi,name=Phi;
327 defid Psi,name=Psi;
328 defid Omega,name=Omega;
```

For the variables names I use, as operators, define how they appear in the \LaTeX , and also define that their arguments appear as subscripts.

```
329 defindex diff(down);
330 defid diff,name="\diff";
331 defindex e_(down,down);
332 defid e_,name="e";
333 defindex d_(arg,down,down);
334 defid d_,name="D";
335 defindex u(down);
336 defid u1,name="u_1";
337 defid u2,name="u_2";
338 defid u3,name="u_3";
339 defid u4,name="u_4";
340 defid u5,name="u_5";
```

```

341 defid u6,name="u_6";
342 defid u7,name="u_7";
343 defid u8,name="u_8";
344 defid u9,name="u_9";
345 defindex s(down);
346 defid exp,name="\exp";
347 defindex exp(arg);

```

Can we write the system? Not in matrices apparently. So define a dummy array `tmp_` that we use to get the correct symbol typeset.

```

348 clear tmp_,tmp_s,tmp_z;
349 array tmp_(n),tmp_s(m),tmp_z(m);
350 defindex tmp_(down);
351 defindex tmp_s(down);
352 defindex tmp_z(down);
353 defid tmp_,name="\dot u";
354 defid tmp_s,name="\vec e";
355 defid tmp_z,name="\vec z";
356 rhs_:=rhsfn$
357 for k:=1:m do tmp_s(k):=
358     {for j:=1:n collect ee(j,k),exp(eval_(k)*t)};
359 for k:=1:m do tmp_z(k):=
360     {for j:=1:n collect zz(j,k),exp(eval_(k)*t)};

```

We have to be shiftty here because `rlfi` does not work inside a loop: so write the commands to a file, and then input the file. Unclear why `off echo` needed after 2021.

```

361 out "scratchfile.red";
362 write "off echo;"$
363 write "write ""\"
364 \IfFileExists{cmsys.red}{\verbatiminput{cmsys.red}}{}
365 \paragraph{The specified dynamical system}
366 \("";";
367 for j:=1:n do write "tmp_(",j,"):=
368     coeffn(rhs_,e_(",j,",1),1);";

```

```

369 write "write ""\
370 \paragraph{Invariant subspace basis vectors}
371 \("";";
372 for j:=1:m do write "tmp_s(",j,"):=tmp_s(",j,");";
373 for j:=1:m do write "tmp_z(",j,"):=tmp_z(",j,");";
374 write "end;";
375 shut "scratchfile.red";

```

Now print the dynamical system to the LaTeX sub-file.

```

376 write "Ignore the following 13 lines of LaTeX"$
377 on latex$
378 out "invarManReportSys.tex"$
379 in "scratchfile.red"$
380 shut "invarManReportSys.tex"$
381 off latex$

```

4 Linear approximation to the invariant manifold

But first, write out the possibly adjusted nonlinear right-hand side function. According to the manual, this will append to the earlier output to the file.

```

382 write "Analyse ODE/DDE system du/dt = ",ff;

```

Parametrise the invariant manifold in terms of these amplitudes. For this substitution to work, **gg** *cannot* be declared scalar as then it gets replaced by zero here and throughout. Let **gg** be global so a user can access the time derivative expressions afterwards, similarly for **uu** the constructed invariant manifold.

```

383 clear gg;
384 clear s; operator s; depend s,t;
385 foreach z_ in spacex do depend s,z_;
386 let df(s(~j),t)=>coeffn(gg,e_(j,1),1);

```

Invoke the following procedure to substitute whatever the current approximation is into (nonlinear) expressions.


```

387 procedure manifold_(uu,n);
388     for j:=1:n collect mkid(u,j)=coeffn(uu,e_(j,1),1)$

```

The linear approximation to the invariant manifold must be the following corresponding to the eigenvalues down the diagonal (even if zero). The amplitudes s_j are slowly evolving as they are either slow modes, or the complex amplitudes of oscillating modes.

```

389 uu:=for j:=1:m sum s(j)*exp(eval_(j)*t)
390   *(for k:=1:n sum e_(k,1)*ee(k,j));
391 gg:=0;
392 if trace then write uu:=uu;

```

For some temporary trace printing, where for simplicity `small` is replaced by `s`.

```

393 procedure matify_(a,m,n)$
394   begin matrix z_(m,n);
395     for i:=1:m do for j:=1:n do z_(i,j):=coeffn(a,e_(i,j),1);
396     return (z_ where {exp(0)=>1,small=>s});
397   end$

```

For the isochron may need to do something different with eigenvalues, but this should work as the inner product is complex conjugate transpose. The `pp` matrix is proposed to place the projection residuals in the range of the isochron.

```

398 zs:=for j:=1:m sum exp(eval_(j)*t)
399   *(for k:=1:n sum e_(k,j)*zz(k,j))$
400 pp:=0$

```

5 Iteratively construct the invariant manifold

But first establish the Taylor series in any delay factors of slow amplitudes.

```

401 let d_(s(~k),t,~dt)=>s(k)+(for n:=1:toosmall sum
402     (-dt)^n*df(s(k),t,n)/factorial(n));

```

Truncate expansions to specified order of error (via loop index trick).

```
403 for j:=toosmall:toosmall do let small^j=>0;
```

Iteratively construct the invariant manifold.

```
404 write "Start iterative construction of invariant manifold";
405 for iter:=1:maxiter do begin
406   if trace then write "
407   ITERATION = ",iter,"
408   -----";
```

Compute residual vector (matrix) of the dynamical system [Roberts \(1997\)](#).

```
409 resde:=-df(uu,t)+sub(manifold_(uu,n),rhsfn);
410 if trace then write "resde=",matify_(resde,n,1);
```

Get the local directions of the coordinate system on the curving manifold:
store transpose as $m \times n$ matrix.

```
411 est:=tpe_(for j:=1:m sum df(uu,s(j))*e_(1,j),e_);
412 est:=conj_(est);
413 if trace then write "est=",matify_(est,m,n);
```

Compute residual matrix for the isochron projection [Roberts \(1989a, 2000\)](#).
But for the moment, only do it if the `eval_set` is for slow manifolds.

```
414 if itisSlowMan_ then begin
415   jacadj:=conj_(sub(manifold_(uu,n),rhsjact));
416   if trace then write "jacadj=",matify_(jacadj,n,n);
417   resd:=df(zs,t)+jacadj*zs+zs*pp;
418   if trace then write "resd=",matify_(resd,n,m);
```

Compute residual of the normalisation of the projection.

```
419   resz:=est*zs-eyem*exp(0);
420   if trace then write "resz=",matify_(resz,m,m);
421 end else resd:=resz:=0; % for when not slow manifold
```

Write lengths of residuals as a trace print (remember that the expression 0 has length one).

```
422 write lengthRes:=map(length(~a),{resde,resd,resz});
```

Solve for updates—all the hard work is already encoded in the operators.

```
423 uu:=uu+l_inv(resde,exp);
424 gg:=gg+g_inv(resde,exp);
425 if trace then write "gg=",matify_(gg,m,1);
426 if trace then write "uu=",matify_(uu,n,1);
```

Now update the isochron projection, with normalisation.

```
427 if itisSlowMan_ then begin
428 zs:=zs+l1_inv(resd,exp)-l2_inv(resz,exp);
429 pp:=pp-g1_inv(resd,exp)+youshouldnotseethis*g2_inv(resz,exp);
430 if trace then write "zs=",matify_(zs,n,m);
431 if trace then write "pp=",matify_(pp,m,m);
432 end;
```

Terminate the iteration loop once residuals are zero.

```
433 %showtime;
434 if {resde,resd,resz}={0,0,0} then write iter:=iter+10000;
435 end;
436 showtime;
```

Only proceed to print if terminated successfully.

```
437 if {resde,resd,resz}={0,0,0}
438   then write "SUCCESS: converged to an expansion"
439   else <<write "FAILED TO CONVERGE; I EXIT";
440   return; >>;
```

6 Output text version of results

Once construction is finished, simplify `exp(0)`.

```
441 let exp(0)=>1;
```

Invoking `switch complex` improves some of the output of the complex numbers, but wrecks other parts of the output. Best left off.

Write text results.

```
442 write "The invariant manifold is (to one order lower)";
443 for j:=1:n do write "u",j," = ",
444   coeffn(small*uu,e_(j,1),1)/small;
445 write "The evolution of the real/complex amplitudes";
446 for j:=1:m do write "ds(",j,")/dt = ",
447   coeffn(gg,e_(j,1),1);
```

Optionally write the projection vectors.

```
448 if itisSlowMan_ then begin write "
449 The normals to the isochrons at the slow manifold.
450 Use these vectors: to project initial conditions
451 onto the slow manifold; to project non-autonomous
452 forcing onto the slow evolution; to predict the
453 consequences of modifying the original system; in
454 uncertainty quantification to quantify effects on
455 the model of uncertainties in the original system.";
456   for j:=1:m do write "z",j," = ",
457     for i:=1:n collect coeffn(zs,e_(i,j),1);
458 end;
```

Write text results numerically evaluated when expressions are long.

```
459 if length(gg)>30 then begin
460 on rounded; print_precision 4$
461 write "Numerically, the invariant manifold is (to one order lower)";
462 for j:=1:n do write "u",j," = ",
463   coeffn(small*uu,e_(j,1),1)/small;
464 write "Numerically, the evolution of the real/complex amplitudes";
465 for j:=1:m do write "ds(",j,")/dt = ",
466   coeffn(gg,e_(j,1),1);
467 if itisSlowMan_ then begin
468   write "Numerically, normals to isochrons at slow manifold.";
```

```

469   for j:=1:m do write "z",j," = ",
470       for i:=1:n collect coeffn(zs,e_(i,j),1);
471 end;
472 off rounded;
473 end;

```

7 Output LaTeX version of results

Change the printing of temporary arrays.

```

474 clear tmp_zz; array tmp_zz(m,n);
475 defid tmp_,name="u";
476 defid tmp_s,name="\dot s";
477 defid tmp_z,name="\vec z";
478 operator zs_;%(m,n);
479 defid zs_,name="z";
480 defindex zs_(down,down);

```

Gather complicated result

```

481 for k:=1:m do for j:=1:n do
482     tmp_zz(k,j):=(1*coeffn(zs,e_(j,k),1));

```

Include order of error to make printing more robust. But we cannot use `small^toosmall` in the following as that is set to zero (for the asymptotics), so we hard code that `small` appears as `varepsilon` ε .

```

483 clear order_; operator order_;
484 defid order_,name="0";
485 defindex order_(arg);

```

Write to a file the commands needed to write the LaTeX expressions. Write the invariant manifold to one order lower than computed.

```

486 out "scratchfile.red";
487 write "off echo;$ % do not understand why needed in 2021??
488 write "write ""\
489 \paragraph{The invariant manifold}

```

```

490 These give the location of the invariant manifold in
491 terms of parameters~\((s\_j)\).
492 \("";";
493 for j:=1:n do write "tmp_(",j,"):=coeffn(small*uu,e_(",j,
494      ",1),1)/small +order_(varepsilon^",toosmall-1,");";
495 if length(gg)>30 then begin
496 write "on rounded; print_precision 4$"$
497 for j:=1:n do write "tmp_(",j,"):=coeffn(small*uu,e_(",j,
498      ",1),1)/small +order_(varepsilon^",toosmall-1,");";
499 write "off rounded;"$
500 end;

```

Write the commands to write the ODEs on the invariant manifold.

```

501 write "write ""\
502 \paragraph{Invariant manifold ODEs}
503 The system evolves on the invariant manifold such
504 that the parameters evolve according to these ODEs.
505 \("";";
506 for j:=1:m do write "tmp_s(",j,"):=1*coeffn(gg,e_(",j
507      ",1),1)+order_(varepsilon^",toosmall,");";
508 if length(gg)>30 then begin
509 write "on rounded; print_precision 4$"$
510 for j:=1:m do write "tmp_s(",j,"):=1*coeffn(gg,e_(",j
511      ",1),1)+order_(varepsilon^",toosmall,");";
512 write "off rounded;"$
513 end;

```

Optionally write the commands to write the projection vectors on the slow manifold.

```

514 if itisSlowMan_ then begin
515   write "write ""\
516   \paragraph{Normals to isochrons at the slow manifold}
517 Use these vectors: to project initial conditions
518 onto the slow manifold; to project non-autonomous
519 forcing onto the slow evolution; to predict the

```

```

520 consequences of modifying the original system; in
521 uncertainty quantification to quantify effects on
522 the model of uncertainties in the original system.
523 The normal vector \(\vec z_j:=(z_{j1},\ldots,z_{jn})\)
524 \("";";
525   for i:=1:m do for j:=1:n do
526     write "zs_(",i,",",j,"):=tmp_zz(",i,",",j
527         ,")+order_(varepsilon^",toosmall,")";";
528 end;%if itisSlowMan_

```

Finish the scratchfile.

```

529 write ";end;";
530 shut "scratchfile.red";

```

Execute the scratchfile with the required commands, with output to the main invariant manifold LaTeX file. `revpri` makes no difference to the \LaTeX output!

```

531 out "invarManReport.tex"$
532 on latex$
533 in "scratchfile.red"$
534 off latex$
535 shut "invarManReport.tex"$

```

8 *Fin*

That's all folks, so end the procedure.

```

536 return Finished_constructing_invariant_manifold_of_system$
537 end$

```

9 Override some system procedures

Bad luck if these interfere with anything else a user might try to do afterwards!

First define how various tokens get printed.

```
538 %load_package rlfi; %must be loaded early
539 deflist('(( ( !\b!i!g!() (!) !\b!i!g!)) (!P!I !\p!i! )
540          (!p!i !\p!i! ) (!E !e) (!I !i) (e !e) (i !i)), 'name)$
```

Override the procedure that prints annoying messages about multicharacter symbols. It ends the output of one expression. This is just a copy from `rlfi.red` with the appropriate if-statement deleted. While interfering, hardcode that the mathematics is in inline mode.

```
541 symbolic procedure prinlaend;
542 <<terpri();
543   prin2t "\)\par";
544   if !*verbatim then
545       <<prin2t "\begin{verbatim}";
546       prin2t "REDUCE Input:">>;
547   ncharspr!*=0;
548   if ofl!* then linelength(car linel!*)
549       else laline!*=cdr linel!*;
550   nochar!*=append(nochar!*,nochar1!*);
551   nochar1!*=nil >>$
552   %
```

Similarly, hardcode at the beginning of expression output that the mathematics is in inline mode.

```
553 symbolic procedure prinlabegin;
554 <<if !*verbatim then
555     <<terpri();
556     prin2t "\end{verbatim}">>;
557   linel!*=linelength nil . laline!*;
558   if ofl!* then linelength(laline!* + 2)
559       else laline!*=car linel!* - 2;
560   prin2 "\(" >>$
```

Override the procedure that outputs the L^AT_EX preamble upon the command `on latex`. Presumably modified from that in `rlfi.red`. Use it to write a

decent header that we use for one master file.

```

561 symbolic procedure latexon;
562 <<!!*a2sfn:='texaeval;
563   !*raise:=nil;
564   prin2t "\documentclass[11pt,a5paper]{article}";
565   prin2t "\usepackage[a5paper,margin=13mm]{geometry}";
566   prin2t "\usepackage{parskip,time,verbatim} \def\_{\_}";
567   prin2t "\raggedright \def\eps{\varepsilon}";
568   prin2t "\def\diff{\partial\makebox[0pt]{\ $ \cdot $}}";
569   prin2t "\title{Invariant manifold of your dynamical system}";
570   prin2t "\author{A. J. Roberts, University of Adelaide\\}";
571   prin2t "\texttt{http://orcid.org/0000-0001-8930-1552}}";
572   prin2t "\date{\now, \today}";
573   prin2t "\begin{document}";
574   prin2t "\maketitle";
575   prin2t "Throughout and generally: the lowest order, most";
576   prin2t "important, terms are near the end of each expression.";
577   prin2t "\par\leftskip=2em \parindent=-2em";
578   prin2t "\input{invarManReportSys}";
579   if !*verbatim then
580       <<prin2t "\begin{verbatim}";
581       prin2t "REDUCE Input:">>;
582   put('tex,'rtypefn,'(lambda(x) 'tex)) >>$

```

End the file when read by Reduce

```
583 end;
```

References

Bunder, J. E. & Roberts, A. J. (2021), 'Nonlinear emergent macroscale PDEs, with error bound, for nonlinear microscale systems', *Springer Nature Applied Sciences* **3**(703), 1–28.

Carr, J. (1981), *Applications of centre manifold theory*, Vol. 35 of *Applied*

- Math. Sci.*, Springer–Verlag.
<http://books.google.com.au/books?id=93BdN7btysoc>
- Coullet, P. H. & Spiegel, E. A. (1983), ‘Amplitude equations for systems with competing instabilities’, *SIAM J. Appl. Math.* **43**, 776–821.
- Fateman, R. (2003), ‘Comparing the speed of programs for sparse polynomial multiplication’, *ACM SIGSAM Bulletin* **37**(1), 4–15.
<http://www.cs.berkeley.edu/~fateman/papers/fastmult.pdf>
- Foias, C., Jolly, M. S., Kevrekidis, I. G., Sell, G. R. & Titi, E. S. (1988), ‘On the computation of inertial manifolds’, *Physics Letters A* **131**, 433–436.
- Haragus, M. & Iooss, G. (2011), *Local Bifurcations, Center Manifolds, and Normal Forms in Infinite-Dimensional Dynamical Systems*, Springer.
- Roberts, A. J. (1988), ‘The application of centre manifold theory to the evolution of systems which vary slowly in space’, *J. Austral. Math. Soc. B* **29**, 480–500.
- Roberts, A. J. (1989a), ‘Appropriate initial conditions for asymptotic descriptions of the long term evolution of dynamical systems’, *J. Austral. Math. Soc. B* **31**, 48–75.
- Roberts, A. J. (1989b), ‘The utility of an invariant manifold description of the evolution of a dynamical system’, *SIAM J. Math. Anal.* **20**, 1447–1458.
- Roberts, A. J. (1997), ‘Low-dimensional modelling of dynamics via computer algebra’, *Computer Phys. Comm.* **100**, 215–230.
- Roberts, A. J. (2000), ‘Computer algebra derives correct initial conditions for low-dimensional dynamical models’, *Computer Phys. Comm.* **126**(3), 187–206.
- Roberts, A. J. (2015a), ‘Macroscale, slowly varying, models emerge from the microscale dynamics in long thin domains’, *IMA Journal of Applied Mathematics* **80**(5), 1492–1518.

Roberts, A. J. (2015*b*), *Model emergent dynamics in complex systems*, SIAM, Philadelphia.

<https://epubs.siam.org/doi/10.1137/1.9781611973563>

Roberts, A. J. (2022), Backwards theory supports modelling via invariant manifolds for non-autonomous dynamical systems, Technical report, [<http://arxiv.org/abs/1804.06998>].